

# \_Greetings, Earthling!\_

Would you like to visit incredible, distant galaxies?

Do you know how to survive on a grim, desolate moon?

Did you ever meet aliens from far-off planets?

Can you escape from the asteroid of doom?

With this book and your home computer you'll be able to do all these things — and a whole lot more!

Just be sure to read our SECRETS OF THE STARS before you begin!

# SECRETS OF THE STARS: The mission support team answers your questions

You may be tempted to skip this section and go right on to the programs. You know what we say to that? **DON'T!!!**Running BASIC programs is fun. So is typing them in. But there are times when programming can make you drift off into space. In this section, we'll give you lots of tips to help keep you on course.

## Will these programs run on my computer?

If you have an IBM Personal Computer or IBM PCjr, the answer is YES! All you have to do is type in the Program Listing. You shouldn't need to make any changes at all.

If you have any of the following computers, the answer is still YES!

Apple II
Apple II Plus
Apple IIe
Atari 400, 800 or XL Series
Coleco ADAM
Commodore 64 or VIC-20
Radio Shack TRS-80 Color Computer
Texas Instruments 99/4A

With these computers, you'll usually have to change the Program Listing slightly. Most of the time only a line or two will be different.

Just be sure to read the "If You Have..." section that appears after every Program Listing. It always tells you exactly what changes you'll need to make for your computer. (The ADAM needs special changes. Keep reading to learn how to make them.)

It's a good idea to read the "If You Have . . ." information before you start typing in your program. If this is your own copy of Computer Space Adventures, you may want to use a pen or pencil to mark the Program Listing with the changes you'll need.

There's more information about each of these computers at the end of this section. Be sure to read it before starting out!

If you have any other computer, the answer is still YES! But you may have to experiment until you find the exact changes your machine needs to run each program. That should be simple. We've tried hard to use only the parts of BASIC that are available on most machines.

### Will my program run as soon as I type it in?

If a program runs as soon as you type it in, you're an absolutely amazing typist — and a very lucky one! Everybody makes little typing mistakes — even us experts! But in a computer program, little mistakes can cause BIG trouble.

If a program doesn't seem to run right, or if you get an error message — something like "Syntax Error" or "SN?" or "You Dummy!" — you'll have to go hunting for your mistakes. Programmers call them bugs. Remember, you put them there, and your program won't run right until you squish out every last one! The official word for that is "debugging."

## How do I get the bugs out of my program?

When you type in a program, you must type each line exactly as it appears in the listing. If you miss even one space or one comma, your program probably won't run the way it should.

First, check your program line by line against the listing in the book. It's a good idea to LIST only a few lines at a time, using commands like LIST 100–150. That way, the whole LIST won't scroll up your screen before you get a chance to read it.

You'll probably slap your head when you discover the idiotic goofs you made. With some computers, you'll have to type the whole line in again to make a correction. With others, you can just type in the changes. Learn which way your computer works: It'll save you lots of typing.

Second, be sure you've made the changes your particular machine needs. Always look at the "If You Have..." section for your particular computer.

If your program still doesn't run, don't give up! There are plenty of things you may have done wrong. How do we know? Because we do them all the time ourselves! Here are some things to watch out for:

### Rotten Leftovers

When you start typing in a program, be sure to type NEW and hit the ENTER (or RETURN) key. If you don't, you may find leftover lines from an old program popping up in your new one. It can be a real mess to get them out!

## The Curse of the Missing Line Number

It's easy to skip a line by accident when you're typing a long program. Remember: Every program in this book includes a REM statement on line 10 with the name of the program. The next line in every program is line 100, and the lines count up by tens from there, like this:

10 REM USELESS PROGRAM

100 INPUT X

110 INPUT Y

120 PRINT Z

The only line numbers that won't end in  $\emptyset$  will be lines you add from the "If You Have . . ." section. And the only time the lines will skip — say from 240 to 260 — is when the "If You Have . . ." section tells you to remove a line for your machine.

## One Out of Two

Some things have to come in pairs, or your computer will get terribly upset. What kinds of things? Mainly quotation marks ("") and left and right parentheses (()). Leaving one out is just about guaranteed to get you some sort of error message. That's why it's important to check a statement like

very carefully. If you're not positive, count the parentheses and quotation marks on your *other* personal computer — your fingers!

#### A Letter's Not a Number

Your computer is very, very fussy about letters and numbers. Remember, a one (1) is not a small L (1) or a capital i (1), even though they may look

almost the same. And as far as the computer is concerned, a capital letter o (0) is not the same as the number zero (0).

In our Program Listings, you'll always find a slash through the number zero (0) to keep you from getting confused. But your fingers may do something your brain never had in mind!

Here are some program lines that will cause your computer to scratch its head and give up. We stole them from our own wastebaskets. Each line contains just one error. Can you spot the goofs?

100 GOT0 240 200 1F I=2 THEN 500 300 N=J 400 R=22I 500 P=416

## A Space Isn't Nothing!

Especially in graphics programs — programs that draw pictures — spaces are terribly important. If the Program Listing shows a space, be sure you don't leave it out. For example, in

```
100 F$="MURBLEHEAD"
110 PRINT F$; " IS A JERK!"
```

if you leave out the space between the first quotation mark and the word IS in line 110, you'll get this:

#### MURBLEHEADIS A JERK!

You can probably guess what prints out if you leave the space in.
But don't add spaces where there are none in the listing. That may
cause problems that are even worse!

#### Semicolons and Commas

Semicolons and commas have special jobs in BASIC programs. Semicolons let you PRINT more than one thing on each line. In PRINT statements, commas do the same thing a little differently, but the programs in this book hardly ever use them that way.

In DATA statements, commas separate each item of data. If you leave one out or put an extra one in, it will confuse your computer. If you get a message like "Out of Data" when everything in your program looks right, chances are you left out a comma — or perhaps a whole DATA statement.

#### Mysterious Letters

Sometimes the letters and names used for variables in a program can get confusing. It's easy to type an E when you meant to type a D. They're right next to each other on the keyboard. So watch out! One mistake like that can keep an almost perfect program from running at all.

## A Real Live Bug?

If you've done everything you can think of, and you're positive we've missed a bug in our own program, please write and describe the problem. We'll try to come up with a solution!

## Why did you include the Sample Runs?

The Sample Runs are here for two reasons. First, they give you an idea of what the programs will do. That way, you can decide whether or not you want to type in a particular program.

The Sample Runs are also here to help you make sure your program is running the way it should. But remember: There are some things you can do on a computer that you can't do on a printed page. Many of these programs create pictures that move across your screen. There's no way we can show that here.

And some of the programs do different things every time you run them. So when you run your program, it may not do exactly what the Sample Run does — but it should come close.

## How come the program listings are in ALL CAPITALS?

If you've already peeked at the Program Listings, you'll notice they're entirely in capital letters. That's because some computers only have capitals. If your machine is one of them, you won't have any problems with any of these programs.

But if you've got a computer that can give you both capitals and lower-case letters, you may run into some problems when you RUN these programs. Why? Let's say we ask a question that has a yes or no answer. You type in "yes" — all lowercase. But the program will be looking for "YES" — all capitals.

The solution? It's easy. Just press the key called CAPS LOCK or ALPHA LOCK or just plain LOCK on your computer. Do it when you type in your programs and when you run them. That way, all the letters will appear in CAPITALS!

## What's an ENTER key? My computer doesn't have one!

One of the strangest and most important keys on the keyboard is the one you use to INPUT data. The strange part is that computer companies

can't agree on what to call it. Some call it the ENTER key. Some call it the RETURN key. And some put a weird symbol on it and no lettering at all!

It's always in roughly the same place — to the right of the letters on the keyboard. Our Program Listings always call it the ENTER key. If you've got a RETURN key, you'll need to type that word into your PRINT statements wherever we've used the word ENTER.

That brings up another rule we've used in this book. Whenever a program asks you to answer a question or make a choice, you have to hit the ENTER (or RETURN) key before anything will happen. That means you can use the BACKSPACE key to correct any mistakes before you ENTER your answer into the computer.

If you're a good programmer, you'll also notice that we've tried hard to keep you from entering an answer the computer isn't expecting. If we ask for a number between 1 and 4, you won't be able to enter 5 — or 0!

## My machine doesn't have a BREAK key. Or does it?

The BREAK key is something else that's different on every machine. Sometimes it's just a key marked BREAK. Sometimes it's really two keys — CTRL and BREAK or some other combination that you have to hold down at the same time. We'll tell you exactly which combination works for your computer in a second. Be sure you know how to use it!

Why? Well, some of our programs will run forever unless you BREAK out of them — or turn off your machine! Sometimes a typing error will make the same thing happen when it's not supposed to. And when you're debugging a program, you may only want to run a small part of it and then quit. It's the same with computers as it is with cars — knowing how to get your machine to stop when you want it to is a really lucky BREAK.

# How can I keep my programs from disappearing when I turn off the computer?

Many of the programs in this book will take you only a few minutes to type in. The long ones may take a few hours. And once you turn your computer off, all your hard work will disappear.

But if you have a cassette recorder or disk drive, you can SAVE your programs and RUN them again and again. SAVEs are different on almost every machine. The instruction manual that came with your computer or disk drive should tell you how to SAVE your work — and LOAD it back into your computer again.

You don't have to wait until a program is working to SAVE it. Especially with longer programs, you should SAVE now and then as you type the program in. That way, you'll be able to LOAD everything you've done right back into your machine if your dog or cat or little sister accidentally pulls the plug — or if you've had enough of programming for one day and decide to go to the movies.

If you've got a friend with the same kind of computer as yours, you can each type in different programs and SAVE them on disk or cassette. Then you can swap your disks or cassettes. It's another great way to save yourself some typing!

#### I've got a printer. What about me?

You're really in luck! With a little thought, you'll be able to print out your very own personalized program runs. Do computer-makers agree on the way to get information to your printer? Not any more than they agree on the name of the ENTER (RETURN?) key!

With some machines, it's easy to print out your program runs. All you do is type something like PR#1, and everything you see on the screen will magically turn up on your printer, too!

With other machines it's a little trickier. You may have to change PRINT statements to LPRINT to get your printer to notice them. The best advice we can give is to check the manuals for your computer.

But remember this: If you try to run a program and absolutely *nothing* happens, it's a good bet that you tried to send something to your printer — and your printer wasn't turned on! On most machines, that's an excellent way to stop a program cold!

# I can think of a better way to do some of the things you did. Why did you do them your way?

There are usually lots of different ways to get your computer to do the same thing. If you run this program:

```
10 PRINT "JESKLER LOVES PRUNES"
20 PRINT "JESKLER LOVES PRUNES"
```

you'll get exactly the same result as:

```
10 FOR A=1 TO 2
20 PRINT "JESKLER LOVES PRUNES"
30 NEXT A
```

We can think of at least ten other ways to get the same two lines printed out on the screen.

Which way is best? It all depends. If we've done something that looks silly to you, it may be because we've tried to get it to work with machines that can't do some things yours can.

We've tried to write our programs so they'll run on as many machines as possible. If we've left out your favorite BASIC command or done something in

a roundabout way, we apologize. But if we've done something *really* stupid, let us know about it!

# Will I mess up my computer if I make changes in your programs?

No! In fact, we *hope* you'll try to improve on these programs — and use them as a starting point for your own. But it's probably a good idea to get them running in their original versions first.

## How do I run a program once I've typed it in?

Just type RUN and hit the ENTER or RETURN key!

## What else do I have to read in this chapter?

Just the section on your own computer. After that — happy programming!

## What if I have an IBM Personal Computer or PCjr?

You're in luck! These programs will run on your machine without any changes at all!

You should be sure to start off in BASIC by typing the command KEY OFF (and hitting the ENTER key). If you have a color monitor or TV set, you should next type WIDTH 40 (and the ENTER key again). You can choose the colors that you see on the screen by giving COLOR commands before you begin — or by writing them into the beginning of the program. The manual that comes with your computer will give you the details.

Remember to set the CAPS LOCK key so that you only get CAPITAL letters. To BREAK out of a program on a PC, hold down the CTRL key and press the BREAK key (it's also called SCROLL LOCK). To BREAK out on a PCjr, press the Fn key and then the letter B (for break). The PC's ENTER key has no lettering on it — just a funny arrow with a bent tail.

## What if I have an Apple II Computer?

Many of these programs will run on your machine without any changes at all. In some programs, you'll have to remove one RANDOMIZE statement and change the way the program clears the screen. The "If You Have..." section after each Program Listing will tell you exactly what to do.

Be sure you run these programs in Applesoft BASIC — Apple's Integer BASIC just won't work. When you need to BREAK out of a program, hold down the CONTROL key and press the letter C. Your machine has a RETURN key instead of an ENTER key. If you have an Apple IIe, you should set the CAPS LOCK key so that you only get CAPITAL letters.

### What if I have an Atari Computer?

Atari BASIC is very unusual I t doesn't allow string arrays like N\$(3). It makes you DIMension all your string variables before you can use them. It requires a special way of entering string commands. It lacks features such as the TAB function.

Most of these programs will need simple changes and additions. The "If You Have . . " section after each Program Listing will tell you exactly what to do.

Your machine has a BRE AK key to let you break out of a program. It has a RETURN key instead of an ENTER key. And you can change the colors that appear on your screen—check your manual for the details.

#### What if I have a Coleco ADAM?

Many of these programs will run on your machine without any changes at all. In some programs, you'll have to remove one RANDOMIZE statement and change the way the program clears the screen. Because the version of BASIC in ADAM is almost exactly like the one for the Apple, you'll usually be able to follow the directions for Apple computers in the "If You Have..." section after each Program Listing.

But since the ADAM does not display as many characters across the screen as Apples do, you will have to make changes to some programs—especially the ones that make pictures. The following steps will help you get the programs running on your ADAM:

- 1. Enter the program, making the changes needed for the Apple Computer.
- 2. Now look at the changes for the TRS-80 Color Computer. Don't change any of the lines you already changed for the Apple. Don't change any lines that include the words CLS, LINE INPUT, RND or SOUND. Make all other changes needed for the Color Computer.

When you need to BREAK out of a program, hold down the CONTROL key and press the letter C. Your machine has a RETURN key instead of an ENTER key. And you should set the LOCK key so that you only get CAPITAL letters.

#### What if I have a Commodore 64 or VIC-20?

Many of these programs will run on your machine without any changes at all. In some programs, you'll have to remove a RANDOMIZE statement. In some, you'll also need to change one number in RND statements and change the way the program clears the screen. The "If You Have..." section after each Program Listing will tell you exactly what to do.

Your machine has a RETURN key instead of an ENTER key. To BREAK out of a program, press the RUN/STOP key. You may want to change the col-

ors that appear on your screen. Your computer's manual will tell you exactly how to do it.

Since the VIC prints fewer characters (letters and numbers) on the screen than most other home computers, you may want to adjust the PRINT statements so that words don't "wrap" around the ends of lines. The easiest way is to divide one PRINT statement into two. If you do this, be sure the first PRINT statement keeps its original line number.

### What if I have a Texas Instruments 99/4A Computer?

TI BASIC is unusual in many ways. It uses special characters instead of AND and OR. It uses the term SEG\$ instead of the more common MID\$. It's very fussy about what you can put after the word THEN.

Most of these programs will need simple changes and additions. The "If You Have . . . " section after each Program Listing will tell you exactly what to do

Your machine has an ENTER key. To BREAK out of a program, hold down the FCTN key and press the number 4 key. You should set the ALPHA LOCK key so that you only get CAPITAL letters. You may want to change the colors that appear on your screen. Your computer's manual will tell you how to do it.

## What if I have a TRS-80 Color Computer?

Many of these programs will run on your machine without any changes at all. In some, you'll have to take out one RANDOMIZE statement and change the way RND works. The "If You Have . . ." section after each Program Listing will tell you exactly what to do.

Your machine has an ENTER key and a BREAK key. You may want to change the colors that appear on your screen. Your computer's manual will tell you how to do it.

### NOW, ON TO THE FUN!

# Saturn Mission Control

It's not every day that a space probe you designed passes Saturn. But now it's approaching the ringed planet. If you scan the messages carefully, you may be able to see the rings!

Samp	le	R	un
~ ~F			

YOUR SCREEN IS NOW PICKING
UP TRANSMISSIONS FROM A
SATELLITE PASSING THE PLANET
SATURN. IF YOU STUDY THE
PICTURE CAREFULLY, YOU MAY
BE ABLE TO SEE THE FAMOUS
PLANETARY RINGS AMID ALL
THE SPACE STATIC.

HIT THE ENTER KEY TO FIND OUT?

RECEIVING	TRANSMITTED	DATA	NOW	389 (3)
**-	-**-*	k	*-	*
	*-*			
			*-*	
	*-*-			
	*- * *- *			
<u></u>				*
**	**	*	*	**-
		*-	*	
	*** -**			
				*
			*	*
			* **	*
			**	<b>*</b>

(continued)

```
----*----
_____
------
______
.____*___
______
-----
______
------
 -----*-----
-----
______
______
---*------*-----
*-----
------
_ *---------
------
------------
```

Break

# □ Program Listing

```
10 REM SATURN MISSION CONTROL
 100 PRINT
 110 PRINT "YOUR SCREEN IS NOW PICKING"
 120 PRINT "UP TRANSMISSIONS FROM A"
 130 PRINT "SATELLITE PASSING THE PLANET"
 140 PRINT "SATURN. IF YOU STUDY THE"
 150 PRINT "PICTURE CAREFULLY, YOU MAY"
 160 PRINT "BE ABLE TO SEE THE FAMOUS"
 170 PRINT "PLANETARY RINGS AMID ALL"
 180 PRINT "THE SPACE STATIC."
 190 PRINT
 200 PRINT "HIT THE ENTER KEY TO FIND OUT";
 210 INPUT X$
 220 PRINT
 230 PRINT "RECEIVING TRANSMITTED DATA NOW..."
240 FOR B=1 TO 25
 250 PRINT
 260 FOR LAG=1 TO 200
 270 NEXT LAG
 280 NEXT B
 290 FOR Y=1 TO 40
 300 FOR Z=1 TO 40
310 FOR A=1 TO Y
 320 PRINT "-";
 330 NEXT A
340 PRINT "*";
 350 FOR A=1 TO Z
 360 PRINT "-";
 370 NEXT A
 380 PRINT "*";
 390 NEXT Z
 400 NEXT Y
 410 PRINT "END OF SATELLITE TRANSMISSION"
```

## $\square$ If You Have . . .

#### APPLE II

Change: 200 PRINT "HIT THE RETURN KEY TO FIND OUT";

#### ATARI

Add: 95 DIM X\$(1)

Change: 200 PRINT "HIT THE RETURN KEY TO FIND OUT";

290 FOR Y=1 TO 36

300 FOR 2=1 TO 36 (continued)

# COMMODORE 64

Change: 200 PRINT "HIT THE RETURN KEY TO FIND OUT";

#### **COMMODORE VIC-20**

Change: 200 PRINT "HIT THE RETURN KEY TO FIND OUT"
290 FOR Y=1 TO 22
300 FOR Z=1 TO 22

#### TEXAS INSTRUMENTS 99/4A

Change: 200 PRINT "HIT THE ENTER KEY TO FIND OUT";

230 PRINT "RECIVING TRANSMITTED DATA NOW..."

290 FOR Y=1 TO 28 300 FOR Z=1 TO 28

#### TRS-80 COLOR COMPUTER

Change: 290 FOR Y=1 TO 31

300 FOR Z=1 TO 31


# You Sure Look Funny!\_

It's amazing how many different shapes computers can take! But it's even more amazing to discover how different the people on other planets can be! Ready for an introduction to a friendly alien?

# ☐ Sample Run

EXCUSE ME--WHAT'S YOUR NAME? MARY TUDOR

IT'S FUNNY, MARY TUDOR. ON MY PLANET, NOCILIS, YOU LOOK JUST LIKE THE COMPUTERS WE ALL USE. DO YOU WANT TO SEE WHAT THE PEOPLE WHO LIVE ON NOCILIS LOOK LIKE? YES



HERE'S WHAT I LOOK LIKE--A HANDSOME DEVIL, NO?

# □ Program Listing

```
10 REM YOU SURE LOOK FUNNY!
100 PRINT "EXCUSE ME--WHAT'S YOUR NAME";
110 INPUT N$
120 IF N$="" THEN 100
130 PRINT
140 PRINT "IT'S FUNNY,"; N$; ". ON MY PLANET,"
150 PRINT "NOCILIS, YOU LOOK JUST LIKE THE"
160 PRINT "COMPUTERS WE ALL USE. DO YOU WANT" 170 PRINT "TO SEE WHAT THE PEOPLE WHO LIVE ON"
180 PRINT "NOCILIS LOOK LIKE";
190 INPUT Y$
200 IF MID$(Y$,1,1) <>"N" THEN 240
210 PRINT
220 PRINT "NO?? WELL, TOO BAD."
230 END
240 PRINT CHR$(12)
250 S$=CHR$(32)
260 PRINT TAB (5);
270 FOR A=1 TO 20
280 PRINT "=";
290 NEXT A
300 PRINT TAB(5);"I";
310 FOR B=1 TO 18
320 PRINT S$;
330 NEXT B
340 PRINT "I"
350 FOR C=1 TO 10
360 PRINT TAB(5);"I";S$;
370 FOR D=1 TO 16
380 PRINT "#";
390 NEXT D
400 PRINT S$; "I"
410 NEXT C
420 PRINT TAB(5); "I"; S$;
430 FOR E=1 TO 17
440 PRINT S$;
450 NEXT E
460 PRINT "I"
470 FOR F=1 TO 28
480 PRINT "=";
490 NEXT F
500 PRINT
510 PRINT "I";
520 FOR G=1 TO 26
530 PRINT S$;
540 NEXT G
550 PRINT "I"
560 PRINT "I";
570 FOR H=1 TO 9
580 PRINT S$;
590 NEXT H
```

```
600 PRINT S$; "======"; S$; "======="; S$; "I"
610 PRINT "I";
620 FOR H=1 TO 9
630 PRINT S$:
640 NEXT H
650 PRINT S$; "(--O--) "; S$; "(--O--) "; S$; "I"
660 PRINT "I"; S$; S$; "////"; S$; S$; S$;
670 PRINT "======";S$;"======";S$;"I"
680 PRINT "I";
690 FOR K=1 TO 26
700 PRINT S$:
710 NEXT K
720 PRINT "I"
730 FOR L=1 TO 28
740 PRINT "=";
75Ø NEXT L
760 PRINT
770 PRINT TAB(3); "HERE'S WHAT I LOOK LIKE--"
780 PRINT TAB(3); "A HANDSOME DEVIL, NO?"
```

## $\square$ If You Have.

#### APPLE II

Add: 295 PRINT

Change: 240 HOME

#### **ATARI**

Add: 95 DIM N\$ (40),Y\$(3),S\$(1) 295 PRINT

785 END

795 FOR AA=1 TO IDX

805 PRINT " ";: NEXT AA: RETURN

Change: 200 IF Y\$(1,1) <>"N" THEN 240

240 PRINT CHR\$(125)

260 IDX=4:GOSUB 795

300 IDX=4:GOSUB 795:PRINT "I"; 360 IDX=4:GOSUB 795:PRINT "I";S\$;

420 IDX=4:GOSUB 795: PRINT "I"; S\$;

770 IDX=2:GOSUB 795:PRINT "HERE'S WHAT I LOOK LIKE--"

780 IDX=2:GOSUB 795:PRINT "A HANDSOME DEVIL, NO?"

#### **COMMODORE 64**

Add: 295 PRINT

(continued) Change: 240 PRINT CHR\$(147)

#### **COMMODORE VIC-20**

Add: 295 PRINT

Change: 100 PRINT "EXCUSE ME--WHAT'S YOUR NAME"

180 PRINT "NOCILIS LOOK LIKE"

240 PRINT CHR\$(147)

270 FOR A=1 TO 12

310 FOR B=1 TO 10

350 FOR C=1 TO 5

370 FOR D=1 TO 8

430 FOR E=1 TO 9

470 FOR F=1 TO 21

520 FOR G=1 TO 19 570 FOR H=1 TO 6

600 PRINT S\$; "=====";S\$; "=====";S\$; "I"

620 FOR H=1 TO 6

650 PRINT S\$;"(-O-)";S\$;"(-O-)";S\$;"I"
660 PRINT "I";S\$;"///";S\$;S\$;
670 PRINT "=====";S\$;"=====";S\$;"I"

690 FOR K=1 TO 19

730 FOR L=1 TO 21

#### **TEXAS INSTRUMENTS 99/4A**

Change: 160 PRINT "COMPUTERS WE ALL USE. DO YOU WANT"

170 PRINT "TO SEE WHAT THE PEOPLE WHO LIVE ON"

200 IF SEG\$(Y\$,1,1)<>"N" THEN 240

240 CALL CLEAR

350 FOR C=1 TO 8

#### TRS-80 COLOR COMPUTER

Delete: 420,430,440,450,460,510,520,530,540,550,680,690,

700,710,720

Add: 295 PRINT

Change: 240 CLS

350 FOR C=1 TO 5

# Alien Eyes

Exploring a distant galaxy, you encounter friendly aliens. Use your computer to talk with them in your language — and theirs!

1	* or %
	Sample Run
	IN A GALAXY CALLED OCULA, ALIENS CAN HAVE ANYWHERE FROM 3 TO 10 EYES
	HERE COME SOME ALIENS NOW! HOW MANY EYES DOES EACH ONE HAVE? 9
	ON THIS PLANET, CALLED RFFONWM, THE CREATURES CALL THEMSELVES RISDOXFS.
	GIVE THE RISDOXFS A WELCOMING MESSAGE: MESSAGE? HI THERE RISDOXFS
	HHHHHHHHIIIIIIII TTTTTTTHHHHHHHHHHEE EEEEEERRRRRRREEEEEEEE RRRRRRRRIIII IIIISSSSSSSSDDDDDDDDDOOOOOOOOXXXXXXX XXFFFFFFFFSSSSSSSS
	HIT "ENTER" TO CONVERSE FURTHER OR ENTER "Q" TO QUIT AND RETURN HOME OR ENTER "P" TO GO TO ANOTHER PLANET?
	MESSAGE? NICE GLASSES YOU HAVE THERE!!
× ·	NNNNNNNNIIIIIIIIIICCCCCCCCEEEEEEE GGGGGGGGLLLLLLLLAAAAAAAASSSSSSSSSSSSSSS
167	

HIT "ENTER" TO CONVERSE FURTHER
OR ENTER "Q" TO QUIT AND RETURN HOME
OR ENTER "P" TO GO TO ANOTHER PLANET? Q

# □ Program Listing

```
10 REM ALIEN EYES
100 Q$=CHR$(34)
110 NAM$=""
120 PLANETS=""
130 FOR A=1 TO 5
140 PRINT
150 NEXT A
160 PRINT "IN A GALAXY CALLED OCULA, ALIENS CAN"
170 PRINT "HAVE ANYWHERE FROM 3 TO 10 EYES..."
180 PRINT
190 PRINT "HERE COME SOME ALIENS NOW!"
200 PRINT "HOW MANY EYES DOES EACH ONE HAVE";
210 INPUT EYES
220 IF EYES<3 THEN 180
230 IF EYES>10 THEN 180
240 PRINT
250 RANDOMIZE (EYES)
260 FOR B=1 TO 7
270 PLANET$=PLANET$+CHR$(INT(RND(1)*26)+65)
280 NAM=NAM+CHR=(INT(RND(1) *26) +65)
290 NEXT B
300 PRINT "ON THIS PLANET, CALLED "; PLANETS; ", THE"
310 PRINT "CREATURES CALL THEMSELVES "; NAM$; "S."
320 PRINT
330 PRINT "GIVE THE "; NAM$; "S A WELCOMING MESSAGE: "
340 PRINT "MESSAGE";
350 INPUT MESSAGE$
360 IF MESSAGE$="" THEN 340
37Ø PRINT
380 PRINT
390 GOSUB 660
400 PRINT
410 K=0
420 FOR D=1 TO LEN(MESSAGE$)
430 M$=MID$(MESSAGE$,D,1)
440 IF MS=" " THEN 500
450 FOR E=1 TO EYES
460 PRINT M$;
470 GOSUB 720
480 NEXT E
490 GOTO 520
500 PRINT ";
510 GOSUB 720
520 NEXT D
530 PRINT
540 PRINT
550 GOSUB 660
560 PRINT
570 PRINT "HIT ";Q$; 'ENTER";Q$;" TO CONVERSE FURTHER"
580 PRINT "OR ENTER ";Q$;"Q";Q$;" TO QUIT AND RETURN HOME"
590 PRINT "OR ENTER ";Q$;"P";Q$;" TO GO TO ANOTHER PLANET";
600 INPUT D$
```

```
610 IF MID$(D$,1,1) = P THEN 110
620 IF MID$(D$,1,1) = "Q" THEN END
630 PRINT
640 GOTO 340
650 REM --- LINE DRAWING SUBROUTINE ---
660 FOR C=1 TO 38
670 PRINT "=";
680 NEXT C
690 PRINT "="
700 RETURN
710 REM --- LINE BREAK SUBROUTINE
720 K=K+1
730 IF K<39 THEN 760
740 K=0
750 PRINT
760 RETURN
```

## $\square$ If You Have . . .

#### **APPLE II**

Delete: 250

Change: 570 PRINT "HIT ";Q\$; "RETURN";Q\$; " TO CONVERSE

FURTHER"

#### **ATARI**

Delete: 250

Add: 95 DIM Q\$(1), NAM\$(7), PLANET\$(7), D\$(1),

MESSAGE\$(25), M\$(1) 605 IF D\$="" THEN 640

Change: 270 PLANET\$(B) = CHR\$(INT(RND(0) \*26) +65)

280 NAM\$(B) = CHR\$(INT(RND( $\emptyset$ ) \*26)+65)

430 M\$=MESSAGE\$(D,D)

570 PRINT "HIT ";Q\$; "RETURN";Q\$; " TO CONVERSE

FURTHER"

610 IF D\$(1,1) = "P" THEN 110

620 IF  $D$(1,1) = Q^n$  THEN END

660 FOR C=1 TO 36

730 IF K<37 THEN 760

(continued)

#### **COMMODORE 64**

Change: 250 REM

270 PLANET\$=PLANET\$+CHR\$(INT(RND(0)\*26)+65)

280 NAM\$=NAM\$+CHR\$(INT(RND(@)\*26)+65)

570 PRINT "HIT ";Q\$; "RETURN";Q\$; " TO CONVERSE

FURTHER"

#### **COMMODORE VIC-20**

Change: 200 PRINT "HOW MANY EYES DOES EACH ONE HAVE"

250 REM

270 PLANET\$=PLANET\$+CHR\$(INT(RND( $\emptyset$ ) \*26)+65)

280 NAM=NAM=+CHR=(INT(RND(0) \*26)+65)

340 PRINT "MESSAGE"

570 PRINT "HIT ";Q\$; "RETURN";Q\$; " TO CONVERSE

FURTHER"

590 PRINT "OR ENTER ";Q\$; "P";Q\$; " TO GO TO

ANOTHER PLANET"

660 FOR C=1 TO 21

#### **TEXAS INSTRUMENTS 99/4A**

Add: 785 END

Change: 160 PRINT "IN A GALAXY CALLED OCULA, ALIENS CAN"

170 PRINT "HAVE ANYWHERE FROM 3 TO 10 EYES..."

250 RANDOMIZE

270 PLANET\$=PLANET\$&CHR\$(INT(RND\*26)+65)

280 NAM\$=NAM\$&CHR\$(INT(RND\*26)+65)

430 M\$=SEG\$(MESSAGE\$,D,1)

610 IF SEG\$(D\$,1,1)="P" THEN 110

620 IF SEG\$(D\$,1,1)="Q" THEN 785

660 FOR C=1 TO 27

#### TRS-80 COLOR COMPUTER

Delete: 250

Change: 270 PLANET\$=PLANET\$+CHR\$(RND(26)+64)

280 NAM\$=NAM\$+CHR\$(RND(26)+64)

660 FOR C=1 TO 30 730 IF K<31 THEN 760

# **Historic Moon Walk**

You knew you'd be chosen for a mission someday. Now someday is here. Run this program to see your world-famous walk on the moon!

# $\Box$ Sample Run

YOU ARE THE YOUNGEST ASTRONAUT EVER TO REACH THE MOON. THE COMPUTER WILL HELP YOU LEAVE YOUR FOOTPRINTS ON ITS SURFACE. HIT THE ENTER KEY TO MAKE YOUR LANDING?

YOU ARE LANDING YOU ARE LANDING

YOU CAN HEAR THE CHEERS OVER THE RADIO AS YOU TELL THE CREW BACK ON EARTH THAT YOU'VE LANDED. NOW IT'S TIME TO LEAVE YOUR FOOTPRINTS ON THE SURFACE FOREVER. HIT THE ENTER KEY FOR EACH STEP

```
?
```

Break

# □ Program Listing

```
10 REM HISTORIC MOONWALK
100 PRINT
110 PRINT "YOU ARE THE YOUNGEST ASTRONAUT"
120 PRINT "EVER TO REACH THE MOON. THE"
130 PRINT "COMPUTER WILL HELP YOU LEAVE"
140 PRINT "YOUR FOOTPRINTS ON ITS SURFACE."
160 PRINT
170 PRINT "HIT THE ENTER KEY"
180 PRINT "TO MAKE YOUR LANDING";
190 INPUT X$
200 FOR A=25 TO 1 STEP -1
210 PRINT TAB(A); "YOU ARE LANDING"
220 NEXT A
230 FOR B=1 TO 25
240 PRINT "."; TAB(B+14); "."
250 NEXT B
260 PRINT "YOU CAN HEAR THE CHEERS OVER THE RADIO"
270 PRINT "AS YOU TELL THE CREW BACK ON EARTH THAT"
280 PRINT "YOU'VE LANDED. NOW IT'S TIME TO LEAVE"
290 PRINT "YOUR FOOTPRINTS ON THE SURFACE FOREVER."
300 PRINT "HIT THE ENTER KEY FOR EACH STEP"
310 INPUT S$
320 FOR C=1 TO 25
330 PRINT
340 NEXT C
350 GOTO 370
360 INPUT LT$
370 FOR D=1 TO 3
380 PRINT TAB(5); "**"
390 NEXT D
400 PRINT
410 INPUT RT$
420 FOR E=1 TO 3
430 PRINT TAB(9); "**"
440 NEXT E
450 GOTO 360
```

## $\square$ If You Have $\dots$

#### APPLE II

Change: 170 PRINT "HIT THE RETURN KEY"
300 PRINT "HIT THE RETURN KEY FOR EACH STEP"

(continued)

#### ATARI -

Add: 95 DIM X\$(1),S\$(1),LT\$(2),RT\$(2)

465 FOR AA=1 TO IDX

475 PRINT " ";:NEXT AA:RETURN

Change: 170 PRINT "HIT THE RETURN KEY"

200 FOR A=21 TO 1 STEP -1

210 IDX=A:GOSUB 465:PRINT "YOU ARE LANDING"

230 FOR B=1 TO 21

240 PRINT ".";: IDX=(B+10):GOSUB 465:PRINT "."

300 PRINT "HIT THE RETURN KEY FOR EACH STEP"

380 IDX=4:GOSUB 465:PRINT "\*\*"

430 IDX=8:GOSUB 465:PRINT "\*\*"

#### COMMODORE 64

Change: 170 PRINT "HIT THE RETURN KEY"

200 FOR A=23 TO 1 STEP -1

230 FOR B=1 TO 23

300 PRINT "HIT THE RETURN KEY FOR EACH STEP"

#### **COMMODORE VIC-20**

Change: 170 PRINT "HIT THE RETURN KEY"

200 FOR A=5 TO 1 STEP -1

230 FOR B=1 TO 5

300 PRINT "HIT THE RETURN KEY FOR EACH STEP"

#### **TEXAS INSTRUMENTS 99/4A**

Change: 110 PRINT "YOU ARE THE YOUNGEST ASTRO- NAUT"

140 PRINT "YOUR FOOTPRINTS ON ITS SUR- FACE"

200 FOR A=14 TO 1 STEP -1

230 FOR B=1 TO 14

240 PRINT "."; TAB(B+12);"."
280 PRINT "YOU'VE LANDED. NOW IT'S TIME TO LEAVE"

290 PRINT "YOUR FOOTPRINTS ON THE SUR- FACE

FOREVER."

300 PRINT "HIT THE ENTER KEY FOR EACH STEP"

#### TRS-80 COLOR COMPUTER

Change: 200 FOR A=16 TO 1 STEP -1

230 FOR B=1 TO 16

# Streaking Comet

Want to discover a comet no other human has ever seen before?

Type in this program and it just may happen!

# ☐ Sample Run

A NEW COMET HAS BEEN DISCOVERED! HIT THE ENTER KEY TO SEE IT ARCH ACROSS THE SKY!?

=====\*\* =====\*\*

====== \*\*

=====\*\*

=====\*\*

=====\*\* =====\*\*

# ☐ Program Listing

10 REM STREAKING COMET

100 PRINT "A NEW COMET HAS BEEN DISCOVERED!"

110 PRINT "HIT THE ENTER KEY TO SEE IT ARCH"

120 PRINT "ACROSS THE SKY!";

130 INPUT X\$

140 FOR A=1 TO 7

150 FOR B=8-A TO 1 STEP -1

160 PRINT

(continued)

```
170 NEXT B
180 T=A
190 GOSUB 340
200 GOSUB 340
210 PRINT CHR$(12)
220 NEXT A
230 FOR C=1 TO 21
240 FOR D=1 TO C
250 PRINT
260 NEXT D
 270 T=10+C
280 GOSUB 340
 290 GOSUB 340
300 PRINT CHR$(12)
310 NEXT C
320 END
330 REM --- IMAGE ---
340 PRINT TAB(T);
350 FOR P=1 TO 6
360 PRINT CHR$(61);
370 NEXT P
380 PRINT CHR$(42); CHR$(42)
390 RETURN
```

## If You Have.

#### **APPLE II**

Change: 110 PRINT "HIT THE RETURN KEY TO SEE IT ARCH"

210 HOME 300 HOME

#### ATARI

Add: 95 DIM X\$(1)

395 FOR AA=1 TO IDX
397 PRINT " ";:NEXT AA:RETURN

Change: 110 PRINT "HIT THE RETURN KEY TO SEE IT ARCH"

210 PRINT CHR\$(125) 300 PRINT CHR\$(125) 340 IDX=T:GOSUB 395

#### **COMMODORE 64**

Change: 118 PRINT "HIT THE RETURN KEY TO SEE IT ARCH"

210 PRINT CHR\$(147) 300 PRINT CHR\$(147)

### **COMMODORE VIC-20**

Change: 110 PRINT "HIT THE RETURN KEY TO SEE IT ARCH"

210 PRINT CHR\$(147) 230 FOR C=1 TO 8

270 T=6+C

300 PRINT CHR\$(147)

#### TEXAS INSTRUMENTS 99/4A

Change: 210 CALL CLEAR

300 CALL CLEAR

### TRS-80 COLOR COMPUTER

Change: 210 CLS

230 FOR C=1 TO 16

270 T=6+C 300 CLS

# **Laser Defender**

You live on a nice, peaceful planet, but your neighbors just won't leave you alone. Time to get out your laser beams and defend yourself!

# $\square$ Sample Run

HORRIBLE WINGED ALIENS HAVE BEEN SIGHTED ON THE HORIZON, AND THEY'RE FLYING TOWARD US. THE ONLY WAY TO STOP THEM IS TO SHOOT THOUSANDS OF LASER ARROWS DIRECTLY OVERHEAD!

HIT THE ENTER KEY TO FIRE?

A ! ! ! ! ! !

A ! ! H H ! ! !

A

NICE SHOOTING, ACE! UH-OH--HERE COME SOME MORE!

HIT THE ENTER KEY TO FIRE?

WHERE'D YOU LEARN TO SHOOT LIKE THAT? UH-OH--HERE COME SOME MORE!

HIT THE ENTER KEY TO FIRE?

Break

# □ Program Listing

```
10 REM LASER DEFENDER
100 PRINT
110 PRINT "HORRIBLE WINGED ALIENS HAVE"
120 PRINT "BEEN SIGHTED ON THE HORIZON,"
130 PRINT "AND THEY'RE FLYING TOWARD US."
140 PRINT "THE ONLY WAY TO STOP THEM IS"
150 PRINT "TO SHOOT THOUSANDS OF LASER"
160 PRINT "ARROWS DIRECTLY OVERHEAD!"
170 K=0
180 PRINT
190 PRINT "HIT THE ENTER KEY TO FIRE";
200 INPUT XS
210 FOR W≃1 TO 25
220 PRINT
230 NEXT W
240 R=INT(RND(1)*38)+1
250 IF R=OLDR THEN 240
260 PRINT TAB(R); "A"
270 FOR A=1 TO 4
280 PRINT TAB(R);"1"
290 NEXT A
300 PRINT TAB(R); "H"
310 OLDR=R
320 K=K+1
330 IF K>38 THEN K=0
340 IF K<>R AND K<>R-1 THEN 670
350 PRINT CHR$(7)
360 PRINT CHR$(12)
370 FOR N=1 TO 20
380 P=INT(RND(1)*9)+1
390 O=INT(RND(1)*7)+1
400 S = INT(RND(1) *9) +1
410 T = INT(RND(1)*8)+1
420 U=INT(RND(1)*7)+1
430 W=39-(P+Q+S+T+U)
440 FOR G=1 TO N
450 PRINT
460 NEXT G
470 PRINT TAB(P); CHR$(33+P);
480 PRINT TAB (Q+P); CHR$(33+Q);
490 PRINT TAB (Q+P+S); CHR(33+S);
500 PRINT TAB (Q+P+S+T); CHR$ (33+T);
510 PRINT TAB (Q+P+S+T+U); CHR$ (33+U);
520 PRINT TAB (Q+P+S+T+U+W); CHR$ (33+W)
530 PRINT CHR$ (12)
540 NEXT N
550 ON (RND(1) *4)+1 GOTO 560,580,600,620,640
560 PRINT "NICE SHOOTING, ACE!"
570 GOTO 650
580 PRINT "YOU BLASTED THEM OUT OF THE SKY!"
590 GOTO 650
600 PRINT "WHERE'D YOU LEARN TO SHOOT LIKE THAT?"
```

```
610 GOTO 650
620 PRINT "SCRATCH ANOTHER NASTY ALIEN!"
630 GOTO 650
640 PRINT "BULLSEYE! GOT THEM!"
650 PRINT "UH-OH--HERE COME SOME MORE!"
660 GOTO 170
670 GOTO 240
```

# $\square$ If You Have $\dots$

#### **APPLE II**

Change: 190 PRINT "HIT THE RETURN KEY TO FIRE"; 360 HOME 530 HOME

#### **ATARI**

Add: 95 DIM X\$(1) 685 FOR AA=1 TO IDX 695 PRINT " ";:NEXT AA:RETURN Change: 190 PRINT "HIT THE RETURN KEY TO FIRE"; 240 R=INT(RND(0)\*38)+1 260 IDX=R:GOSUB 685:PRINT "A" . 280 GOSUB 685:PRINT "1" 300 GOSUB 685:PRINT "!" 350 PRINT CHR\$(253) 360 PRINT CHR\$(125) 380 P = INT(RND(0) \* 9) + 1390 Q=INT(RND(0)\*7)+1 400 S = INT(RND(0)\*9)+1410 T=INT(RND(0)\*8)+1 420 U=INT(RND(0) \*7)+1 470 IDX=(P):GOSUB 685:PRINT CHR\$(33+P); 480 IDX=(Q+P):GOSUB 685:PRINT CHR\$(33+Q); 490 IDX=(Q+P+S):GOSUB 685:PRINT CHR\$(33+S); 500 IDX = (Q+P+S+T) : GOSUB 685 : PRINT CHR\$(33+T);510 IDX=(Q+P+S+T+U):GOSUB 685:PRINT CHR\$(33+U); 520 IDX=(Q+P+S+T+U+W):GOSUB 685:PRINT CHR\$(33+W); 530 PRINT CHR\$(125) 550 ON (RND(0)\*4)+1 GOTO 560,580,600,620,640

#### **COMMODORE 64**

Delete: 350

Change: 190 PRINT "HIT THE RETURN KEY TO FIRE";

240 R=INT(RND(0)\*38)+1 360 PRINT CHR\$(147) 380 P=INT(RND(0)\*9)+1 390 Q=INT(RND(0)\*7)+1

(continued)

```
400 \text{ S= INT (RND (0) * 9) +1}
410 T=INT(RND(0) *8) +1
420 U=INT(RND(0)*7)+1
530 PRINT CHR$(147)
550 ON (RND(0)*4)+1 GOTO 560,580,600,620,640
```

#### **COMMODORE VIC-20**

Delete: 350

Change: 190 PRINT "HIT THE RETURN KEY TO FIRE"

240 R=INT(RND(0) \*38) +1 330 IF K>21 THEN  $K=\emptyset$ 360 PRINT CHR\$(147) 380 P = INT(RND(0) \* 9) + 13 90 Q=INT(RND(0)  $\star$ 7)+1 400 S = INT(RND(0) \* 9) + 1410 T=INT(RND(0)\*8)+1420 U=INT(RND(0)\*7)+1430 W=20-(P+Q+S+T+U)530 PRINT CHR\$(147)

550 ON (RND(0)\*4)+1 GOTO 560,580,600,620,640

#### TEXAS INSTRUMENTS 99/4A

Add: 332 K=0

Change: 240 R=INT(RND\*38)+1 330 IF K<29 THEN 340 340 IF (K<>R)\*(K<>R-1) THEN 670 350 CALL SOUND (100,440,2) 360 CALL CLEAR 380 P = INT(RND\*7) + 1

390 Q = INT(RND\*5) + 1400 S=INT(RND\*6)+1410 T=INT(RND\*5)+1420 U = INT(RND\*5) + 1

530 CALL CLEAR

550 ON (RND\*4)+1 GOTO 560,580,600,620,640

#### TRS-80 COLOR COMPUTER

Change: 240 R= RND (38) 350 SOUND 173,3 360 CLS

380 P=RND(9)

390 Q = RND(7)400. S=RND (9)

410 T=RND(8)

420 U = RND(7)530 CLS

550 ON RND(5) GOTO 560,580,600,620,640

# Scary Black Holes

There's no telling what you'll see on an intergalactic flight. It's just possible that on this one you'll spot some of the famous black holes in space.

	Samp	le Run
--	------	--------

LADIES AND GENTLEMEN, THIS IS YOUR AUTOMATIC COMPUTER PILOT SPEAKING. FLIGHT #8923 TO RIGEL AND CASSIOPEIA IS NOW APPROACHING THE REGION OF BLACK HOLES. IF YOU LOOK OUT YOUR SPACEPORTS, YOU JUST MAY SEE SOME.

HIT THE ENTER KEY TO OPEN THE PORT?

******	******	*****
******	*****	*****
******	**	*****
******	**	*****
*****	**	*****
******	**	*****
*****	**	******
******	*****	*****
******	*******	*****
******	******	*****
******		**
******	****	**
****	****	= :# <b>**</b>
******	****	**
******	*****	**
*****	*****	**
******	*******	*****
******	********	*****
******	******	*****
****	******	*****
****	******	*****
****	******	*****
****	******	*****
****	******	*****
*****	*****	*****
******		
*****	,	*****
******		****
*****		****
*****	***	****
******		****
*****		****
*****		****
*****		
*****		
*****		
******	*****	*****

Break

## ☐ Program Listing

```
10 REM SCARY BLACK HOLES
100 PRINT "LADIES AND GENTLEMEN, THIS IS YOUR"
110 PRINT "AUTOMATIC COMPUTER PILOT SPEAKING."
120 PRINT "FLIGHT #8923 TO RIGEL AND CASSIOPEIA"
130 PRINT "IS NOW APPROACHING THE REGION OF"
140 PRINT "BLACK HOLES. IF YOU LOOK OUT YOUR"
150 PRINT "SPACEPORTS, YOU JUST MAY SEE SOME."
160 PRINT
170 PRINT "HIT THE ENTER KEY TO OPEN THE PORT";
180 INPUT J$
190 PRINT CHR$(12)
200 GOTO 350
210 A=INT (RND(1) *30) +1
220 B=INT (RND(1) *(30-A))+8
230 FOR H=1 TO B/2
240 FOR C=1 TO A
250 PRINT "*";
260 NEXT C
270 FOR T=1 TO B+1
280 PRINT " ";
290 NEXT T
300 \text{ FOR C=1 TO } 37-(A+B)
310 PRINT "*";
320 NEXT C
330 PRINT "*"
340 NEXT H
350 FOR S=1 TO 5
360 FOR D=1 TO 38
370 PRINT "*";
380 NEXT D
390 PRINT "*"
400 NEXT S
410 GOTO 210
```

### $\square$ If You Have .

#### APPLE II

Add: 225 B=B-1

295 IF(A+B) > 36 THEN 330

Change: 170 PRINT "HIT THE RETURN KEY TO OPEN THE PORT";

190 HOME

#### **ATARI**

Add: 95 DIM J\$(1)

305 IF A+B>34 THEN 330

Change: 170 PRINT "HIT THE RETURN KEY TO OPEN THE PORT";

190 PRINT CHR\$(125)

210 A=INT(RND(0)  $\star$  28) +1

220 B=INT (RND (0) \* (28-A))+8

300 FOR C=1 TO 35-(A+B)

360 FOR D=1 TO 36

### **COMMODORE 64**

Add: 295 IF (A+B) =37 THEN 330

Change: 170 PRINT "HIT THE RETURN KEY TO OPEN THE PORT";

190 PRINT CHR\$(147)

210 A=INT(RND(0) \*30)+1

220 B=INT(RND(0)  $\star$ (30-A))+8

#### **COMMODORE VIC-20**

Change: 170 PRINT "HIT THE RETURN KEY TO OPEN THE PORT"

190 PRINT CHR\$(147)

210 A=INT(RND(0) \*15) +1

220 B=INT(RND(0)  $\star$ (11-A))+5

270 FOR T=1 TO B+2

300 FOR C=1 TO 18-(A+B)

360 FOR D=1 TO 20

### **TEXAS INSTRUMENTS 99/4A**

Change: 100 PRINT "LADIES AND GENTLEMEN, THIS IS YOUR"

110 PRINT "AUTOMATIC COMPUTER PILOT SPEAKING."

120 PRINT "FLIGHT #8923 TO RIGEL AND CASSIOPEIA"

130 PRINT "IS NOW APPROACHING THE REGION OF"

140 PRINT "BLACK HOLES. IF YOU LOOK OUT YOUR"

170 PRINT "HIT THE ENTER KEY TO OPEN THE PORT"

190 CALL CLEAR

210 A=INT(RND\*30)+1

220 B=INT (RND\* (30-A)) +8

300 FOR C=1 TO 26-(A+B)

360 FOR D=1 TO 27

### TRS-80COLOR COMPUTER

Change: 190 CLS

210 A=RND (26)

220 B=RND (26-A)

300 FOR C=1 TO 29-(A+B)

360 FOR D=1 TO 30

# **Mysterious Moon Maze**

On the surface of some moons, you'll find giant, hollow crystals. They're exciting to explore — if you don't get lost inside. But that's exactly what you've done. Find your way out, or you'll never see home again!

		200
1 1	Sample R	
$\Box$		W.

WHAT IS THE NAME OF YOUR SHINY NEW SPACE SHIP? USS PRETZEL

YOU ARE ON A NEWLY DISCOVERED MOON OF THE PLANET SATURN. IT IS MADE OUT OF STACKS OF HOLLOW CUBES. WHILE EXPLORING YOU HAVE GOTTEN LOST IN THE MAZE OF THESE CUBES. EACH ONE HAS SIX SURFACES, AND THERE IS A SMALL DOOR IN EACH SURFACE. HOW FAST CAN YOU ESCAPE FROM THIS TREACHEROUS MAZE OF CUBES?

HOW MANY CUBES ARE ON THIS MOON (2-20)? 4

-----

YOU ARE IN CUBE NUMBER 1

1--UP 2--DOWN

3--NORTH 4--SOUTH

5--EAST 6--WEST

WHICH DOOR DO YOU WANT TO TRY? 4

\*\*\* RIGHT!! NOW ON TO CUBE 2 \*\*\*

------

YOU ARE IN CUBE NUMBER 2

1 -- UP 2 -- DOWN

3--NORTH 4--SOUTH

5--EAST 6--WEST

WHICH DOOR DO YOU WANT TO TRY? 4

\*\*\* RIGHT!! NOW ON TO CUBE 3 \*\*\* YOU ARE IN CUBE NUMBER 3 1--UP 2--DOWN 3--NORTH 4--SOUTH 5--EAST 6--WEST WHICH DOOR DO YOU WANT TO TRY? 5. SORRY! DOOR 5 WON'T OPEN! NOW TRY ANOTHER ONE... YOU ARE IN CUBE NUMBER 3 1--UP 2--DOWN 3--NORTH 4--SOUTH 5~~EAST 6--WEST WHICH DOOR DO YOU WANT TO TRY? 6 \*\*\* RIGHT!! NOW ON TO CUBE 4 \*\*\* \_\_\_\_\_\_\_ YOU ARE IN CUBE NUMBER 4 1--UP 2--DOWN 3--NORTH 4--SOUTH 5--EAST 6--WEST WHICH DOOR DO YOU WANT TO TRY? 6 YOU DID IT!! YOU ESCAPED!! THERE WERE 4 CUBES, AND IT TOOK YOU ONLY 5 TRIES IITI OD OT WANT TO TRY AGAIN (YES OR NO)? NO

## □ Program Listing

```
10 REM MYSTERIOUS MOON MAZE
100 PRINT "WHAT IS THE NAME OF YOUR SHINY NEW"
110 PRINT "SPACE SHIP";
120 INPUT V$
130 IF LEN(V$) < 2 THEN 100
140 RANDOMIZE ASC(MID$(V$,1,1)) *ASC(MID$(V$,2,1))
150 PRINT
160 PRINT "YOU ARE ON A NEWLY DISCOVERED MOON OF"
170 PRINT "THE PLANET SATURN. IT IS MADE OUT OF"
180 PRINT "STACKS OF HOLLOW CUBES. WHILE"
190 PRINT "EXPLORING YOU HAVE GOTTEN LOST IN THE"
200 PRINT "MAZE OF THESE CUBES.
                                  EACH ONE HAS SIX"
210 PRINT "SURFACES, AND THERE IS A SMALL DOOR IN"
220 PRINT "EACH SURFACE.
                          HOW FAST CAN YOU ESCAPE"
230 PRINT "FROM THIS TREACHEROUS MAZE OF CUBES?"
240 PRINT
250 GOSUB 760
260 PRINT "HOW MANY CUBES ARE ON THIS MOON (2-20)";
270 INPUT C
280 IF C<2 THEN 260
290 IF C>20 THEN 260
300 GOSUB 760
310 K=K+1
320 R=INT(RND(1) *6)+1
330 IF R=X THEN 320
340 PRINT
350 PRINT "YOU ARE IN CUBE NUMBER "; K
360 PRINT
370 PRINT TAB(5); "1--UP"; TAB(16); "2--DOWN"
380 PRINT TAB(5); "3--NORTH"; TAB(16); "4--SOUTH"
390 PRINT TAB(5); "5--EAST"; TAB(16); "6--WEST"
400 PRINT
410 PRINT "WHICH DOOR DO YOU WANT TO TRY";
420 INPUT G
430 M=M+1
440 IF G>6 THEN 340
450 IF G<1 THEN 340
4.60 IF G=R THEN 540
470 GOSUB 760
480 PRINT TAB(5); "SORRY! DOOR ";G; " WON'T OPEN!"
490 IF G=X THEN PRINT TAB(5); "YOU JUST CAME FROM THERE"
500 IF G=X THEN PRINT TAB(5); "AND IT LOCKED BEHIND YOU!"
510 PRINT TAB(5); "NOW TRY ANOTHER ONE..."
520 GQSUB 760
530 GOTO 350
540 PRINT
550 IF INT(R/2) = R/2 THEN X=R-1
560 IF INT(R/2) <> R/2 THEN X=R+1
570 IF K=C THEN 620
580 GOSUB 760
590 PRINT TAB(3);"*** RIGHT!! NOW ON TO CUBE ";K+1;"
```

```
600 GOSUB 760
610 GOTO 310
620 PRINT TAB(5); "YOU DID IT!! YOU ESCAPED!!"
630 PRINT TAB(5); "THERE WERE "; K; " CUBES, AND"
640 PRINT TAB(5); "IT TOOK YOU ";
650 IF M<C*3 THEN PRINT "ONLY ";
660 PRINT M;" TRIES"
670 PRINT TAB(5); "TO DO IT!!"
680 GOSUB 760
690 PRINT "WANT TO TRY AGAIN (YES OR NO)";
700 INPUT Y$
710 IF MID$(Y$,1,1) = "N" THEN END
720 K=0
730 M=0
740 GOTO 250
750 REM --- LINE SUBROUTINE ---
760 PRINT
770 FOR L=1 TO 39
780 PRINT "=";
790 NEXT L
800 PRINT
810 PRINT
820 RETURN
```

### ☐ If You Have...

### APPLEII

Delete: 140

### **ATARI**

```
Delete: 140
```

Add: 95 DIM V\$(40),Y\$(3)

391 IDX=3:GOSUB 835:PRINT "6--WEST"
835 FOR AA=1 TO IDX
845 PRINT " ";:NEXT AA:RETURN

Change: 320 R=INT(RND(0)\*6)+1
370 IDX=4:GOSUB 835:PRINT "1--UP";:IDX=5:
GOSUB 835:PRINT "2--DOWN"
380 IDX=4:GOSUB 835:PRINT "3--NORTH";
390 IDX=4:GOSUB 835:PRINT "5--EAST";
480 IDX=4:GOSUB 835:PRINT "SORRY! DOOR ";G;
" WON'T OPEN!" (continued)

381 IDX=2:GOSUB 835:PRINT "4--SOUTH"

- 490 IF G=X THEN GOSUB 835: PRINT "YOU JUST CAME FROM THERE"
- 500 IF G=X THEN GOSUB 835: PRINT "AND IT LOCKED BEHIND YOU!"
- 510 GOSUB 835:PRINT "NOW TRY ANOTHER ONE..."
- 590 IDX=2:GOSUB 835:PRINT "\*\*\* RIGHT!! NOW ON TO CUBE "; K+1; "\*\*\*"
- 620 IDX=4:GOSUB 835:PRINT "YOU DID IT!! YOU ESCAPED!!"
- 630 GOSUB 835: PRINT "THESE WERE "; K; " CUBES, AND"
- 640 GOSUB 835: PRINT "IT TOOK YOU ";
- 670 GOSUB 835:PRINT "TO DO IT!!"
- 710 IF Y\$(1,1) = "N" THEN END
- 770 FOR L=1 TO 37

### **COMMODORE 64**

Delete: 140

Change: 320 R=INT(RND(0) \*6)+1

### **COMMODORE VIC-20**

Delete: 140

Change: 110 PRINT "SPACE SHIP"

260 PRINT "HOW MANY CUBES ARE ON THIS MOON (2-20)"

320 R=INT(RND( $\emptyset$ ) \*6)+1

370 PRINT TAB(3);"1--UP";TAB(13);"2--DOWN"

380 PRINT TAB(3); "3--NORTH"; TAB(13); "4--SOUTH"

390 PRINT TAB(3); "5--EAST"; TAB(13); "6--WEST"

410 PRINT "WHICH DOOR DO YOU WANT TO TRY"

690 PRINT "WANT TO TRY AGAIN (YES OR NO)"

770 FOR L=1 TO 21

#### **TEXAS INSTRUMENTS 99/4A**

Add: 492 PRINT TAB(5); "YOU JUST CAME FROM THERE"

552 X=R-1

562 X=R+1

652 PRINT "ONLY ";

845 END

Change: 100 PRINT "WHAT IS THE NAME OF YOUR SHINY NEW"

140 RANDOMIZE

170 PRINT "THE PLANET SATURN. IT IS MADE OUT OF"

180 PRINT "STACKS OF HOLLOW CUBES. WHILE"

190 PRINT "EXPLORING YOU HAVE GOTTEN LOST IN THE" 200 PRINT "MAZE OF THESE CUBES. EACH ONE HAS SIX"

210 PRINT "SURFACES, AND THERE IS A SMALL DOOR IN"
230 PRINT "FROM THIS TREACHEROUS MAZE OF CUBES?"

260 PRINT "HOW MANY CUBES ARE ON THIS MOON (2-20)";

```
320 R=INT(RND*6)+1
410 PRINT "WHICH DOOR DO YOU WANT TO TRY";
490 IF G<>X THEN 510
500 PRINT TAB(5); "AND IT LOCKED BEHIND YOU!"
550 IF INT(R/2)<>R/2 THEN 560
560 IF INT(R/2)=R/2 THEN 570
650 IF M>=C*3 THEN 660
710 IF SEG$(Y$,1,1)="N" THEN 845
770 FOR L=1 TO 27
```

### TRS-80 COLOR COMPUTER

Delete: 140,240

Change: 160 PRINT "YOU ARE ON A NEWLY DISCOVERED MOON OF ";
170 PRINT "THE PLANET SATURN. IT IS MADE OUT OF ";
180 PRINT "STACKS OF HOLLOW CUBES. WHILE ";
190 PRINT "EXPLORING YOU HAVE GOTTEN LOST IN THE ";
200 PRINT "MAZE OF THESE CUBES. EACH ONE HAS SIX ";
320 R=RND(6)
590 PRINT "RIGHT!! NOW ON TO CUBE "; K+1
770 FOR L=1 TO 31

# What a View!

When you're captain of a great star cruiser, you're busy every minute.

But it's important to take time out to enjoy the view.


Run this program and you'll see a spectacular one!

### $\square$ Sample Run

WHAT IS YOUR LAST NAME? MUMBLEGRUBBER

OK, CAPTAIN MUMBLEGRUBBER, YOU ARE ON THE BRIDGE OF THE STARSHIP USS KEYBOARD II. YOU'VE JUST SPENT THE NIGHT BATTLING EVIL ALIENS AND YOU'RE READY FOR A SHOWER (NOT A METEOR SHOWER, THE OTHER KIND) AND A GOOD LONG REST. YOU DESERVE IT!


THIS ELEVATOR WILL TAKE YOU DOWN TO THE OFFICER'S CABIN LEVEL. WE ARE TRAVELING IN A BEAUTIFUL QUADRANT OF THE UNIVERSE, SO WHY NOT LOOK OUT THE WINDOW AND ENJOY THE VIEW AS YOU DESCEND--HIT THE ENTER KEY TO START?







\*\* ONLY 1000 MORE FLOORS TO GO \*\*





Break

# ☐ Program Listing

```
10 REM WHAT A VIEW!
100 PRINT "WHAT IS YOUR LAST NAME";
110 INPUT L$
120 IF L$="" THEN 100
130 PRINT
140 PRINT "OK, CAPTAIN ";L$;", YOU ARE ON"
150 PRINT "THE BRIDGE OF THE STARSHIP USS"
160 PRINT "KEYBOARD II. YOU'VE JUST SPENT"
170 PRINT "THE NIGHT BATTLING EVIL ALIENS"
180 PRINT "AND YOU'RE READY FOR A SHOWER"
```

```
190 PRINT "(NOT A METEOR SHOWER, THE OTHER"
200 PRINT "KIND) AND A GOOD LONG REST.
210 PRINT "DESERVE IT!"
220 PRINT
230 PRINT
          "THIS ELEVATOR WILL TAKE YOU DOWN TO"
240 PRINT
          "THE OFFICER'S CABIN LEVEL.
          "TRAVELING IN A BEAUTIFUL QUADRANT OF"
250 PRINT
260 PRINT "THE UNIVERSE, SO WHY NOT LOOK OUT THE"
270 PRINT "WINDOW AND ENJOY THE VIEW AS YOU"
280 PRINT "DESCEND--HIT THE ENTER KEY TO START";
290 INPUT F$
300 PRINT
310 K = 0
320 L=1000
330 A = INT(RND(1) *29) +1
340 B=INT(RND(1)*(37-A))+1
350 IF A>=B THEN 330
360 PRINT TAB(A+1);
370 FOR T=1 TO B+1
380 PRINT "=";
390 NEXT T
400 FOR S=1 TO B
410 IF INT(RND(1) *3) <>2 THEN 470
420 \cdot Z = INT(RND(1) * (B-A) + 1)
430 IF Z=OLDZ THEN Z=Z+1
440 OLDZ≠Z
450 PRINT TAB(A+1);"!";TAB(A+1+Z);"*";TAB(A+1+B);"!"
.460 GOTO 480
470 PRINT TAB(A+1); "!"; TAB(A+1+B); "!"
480 NEXT S
490 PRINT TAB (A+1);
500 FOR R=1 TO B+1
510 PRINT "=";
520 NEXT R
530 FOR M=1 TO 5
540 PRINT
550 NEXT M
560 K=K+1
570 IF K<>4 THEN 670
580 PRINT TAB(2); "** ONLY "; L; " MORE FLOORS TO GO **"
590 FOR N=1 TO 4
600 PRINT
610 NEXT N
620 L=L~4
630 IF L>0 THEN 660
640 PRINT TAB(2); "WELL, IT'S MORNING-TIME TO GET UP!"
650 END
660 K=0
670 GOTO 330
```

### $\square$ If You Have . . .

### APPLE II

Add: 395 PRINT

Change: 280 PRINT "DESCEND--HIT THE RETURN KEY TO START";

### **ATARI**

Add: 95 DIM L\$(40),F\$(1)

395 PRINT

455 IDX=(B-Z-1):GOSUB 685:PRINT "!"

685 FOR AA=1 TO IDX

695 PRINT ";:NEXT AA:RETURN

Change: 280 PRINT "DESCEND--HIT THE RETURN KEY TO START";

330 A = INT(RND(0) \*29) +1

340 B=INT(RND(0) \*(34-A))+1

350 IF A>B THEN 330

360 IDX = (A+1) : GOSUB 685

370 FOR T=1 TO B+2

410 IF INT(RND(0)\*3)<>2 THEN 470

420 Z=INT(RND(0)\*(B-A)+1)

450 IDX=(A+1):GOSUB 685:PRINT "!";:IDX=2:GOSUB 685: PRINT "\*";

470 IDX=(A+1):GOSUB 685:PRINT "1";:IDX=B:GOSUB 685:
PRINT "!"

490 IDX=(A+1):GOSUB 685

500 FOR R=1 TO B+2

580 IDX=2:GOSUB 685:PRINT "\*\* ONLY ";L; "MORE FLOORS TO GO \*\*"

640 IDX=2:GOSUB 685:PRINT "WELL, IT'S MORNING--TIME TO GET UP!"

#### **COMMODORE 64**

Add: 395 PRINT

Change: 280 PRINT "DESCEND--HIT RETURN KEY TO START";

330 A=INT(RND(0) \*29)+1

340 B=INT(RND(0) \*(37-A))+1

410 IF INT(RND(0) \*3) <> 2 THEN 470

420 Z=INT(RND(0) \* (B-A)+1)

### **COMMODORE VIC-20**

Add: 395 PRINT

225 INPUT TXT\$

Change: 100 PRINT "WHAT IS YOUR LAST NAME"

220 PRINT "HIT RETURN TO CONTINUE" (continued)

```
280 PRINT "DESCEND--HIT RETURN KEY TO START"

330 A=INT(RND(0)*25)+1

340 B=INT(RND(0)*(20-A))+1

370 FOR T=1 TO B

410 IF INT(RND(0)*3)<>2 THEN 470

420 Z=INT(RND(0)*(B-A)+1)

450 PRINT TAB(A+1);"!";TAB(A+Z);"*";TAB(A+B);"!"

470 PRINT TAB(A+1);"!";TAB(A+B);"!"

500 FOR R=1 TO B
```

### **TEXAS INSTRUMENTS 99/4A**

Add: 222 PRINT "HIT ENTER TO CONTINUE"

224 INPUT D\$

432 Z=Z+1

Change: 150 PRINT "THE BRIDGE OF THE STARSHIP USS"

170 PRINT "THE NIGHT BATTLING EVIL ALIENS"

190 PRINT "(NOT A METEOR SHOWER, THE OTHER"

250 PRINT "TRAVELING IN A BEAUTIFUL QUADRANT OF"

260 PRINT "THE UNIVERSE, SO WHY NOT LOOK OUT THE"

280 PRINT "DESCEND---HIT THE ENTER KEY TO START";

320 L=200

330 A=INT(RND\*19)+1

340 B=INT(RND\*(27-A))+1

410 IF INT(RND\*3)<>2 THEN 470

420 Z=INT(RND\*(B-A)+1)

430 IF Z<>OLDZ THEN 440

### TRS-80 COLOR COMPUTER

Add: 395 PRINT

```
140 PRINT "OK, CAPTAIN ";L$; ", YOU ARE ON ";
Change:
        150 PRINT "THE BRIDGE OF THE STARSHIP USS ";
        160 PRINT "KEYBOARD II. YOU'VE JUST SPENT ";
        170 PRINT "THE NIGHT BATTLING EVIL ALIENS ";
        180 PRINT "AND YOU'RE READY FOR A SHOWER ";
        190 PRINT "(NOT A METEOR SHOWER, THE OTHER ";
        200 PRINT "KIND) AND A GOOD LONG REST. YOU ";
        210 PRINT "DESERVE IT! ";
        230 PRINT "THIS ELEVATOR WILL TAKE YOU DOWN TO ";
        240 PRINT "THE OFFICER'S CABIN LEVEL. WE ARE ";
        250 PRINT "TRAVELING IN A BEAUTIFUL QUADRANT OF ";
        260 PRINT "THE UNIVERSE, SO WHY NOT LOOK OUT THE ";
        270 PRINT "WINDOW AND ENJOY THE VIEW AS YOU ":
        330 A=RND(25)
        340 B=RND(27-A)
        410 IF RND(3)<>2 THEN 470
       420 Z=RND (B-A)
```

# Deep Space Sheriff

You're looking for an intergalactic desperado — and it's looking for you!

Find it on a far-off moon and put it back behind bars.

Just don't let it and you first!

	Just don't let it ind you hist!	E
	va va	₩
	Sample Run	
sa s	YOU ARE THE SHERIFF OF THIS PART OF THE GALAXY. YOU ARE CHASING AN ESCAPED ALIEN DESPERADO. YOU ARE THE ONLY TWO BEINGS ON A LONELY MOON THAT HAS TEN HUGE CRATERS.	
	IF YOU FIND THE ALIEN BEFORE IT FINDS YOU, YOU'LL BE ABLE TO TAKE IT BACK TO PRISON ON EARTH. BUT IF IT FINDS YOU FIRST, IT WILL EAT YOU.	
19	HAPPY HUNTING!	200
15 B	WHAT'S YOUR NAME, SHERIFF? WYATT GARP	2 N A
	HOW MANY DESPERADOS HAVE YOU CAUGHT? 5	H
×	WHICH CRATER DO YOU WANT TO LOOK IN? 3	
	YOU ARE HUNTING IN CRATER 3 THE ALIEN IS IN CRATER 6NOT HERE!	
26		10 March 20
	THE ALIEN JUST LOOKED FOR YOU IN CRATER 10BUT HE MIGHT HAVE GONE SOMEWHERE ELSE BY NOW.	
ii.	WHICH CRATER DO YOU WANT TO LOOK IN? 7	2 0 1 2
		8 5
Si se	YOU ARE HUNTING IN CRATER 7 THE ALIEN IS IN CRATER 5NOT HERE!	
		a * * · · ·
	THE ALIEN JUST LOOKED FOR YOU IN CRATER 1-BUT HE MIGHT HAVE GONE SOMEWHERE ELSE BY NOW.	

WHICH CRATER DO YOU WANT TO LOOK IN? 2

YOU ARE HUNTING IN CRATER 2
THE ALIEN IS IN CRATER 5--NOT HERE!

THE ALIEN JUST LOOKED FOR YOU IN CRATER 6-BUT HE MIGHT HAVE GONE SOMEWHERE ELSE BY NOW.

WHICH CRATER DO YOU WANT TO LOOK IN? 9

YOU ARE HUNTING IN CRATER 9
GOOD WORK, SHERIFF WYATT GARP!
YOU FOUND THE LITTLE GREEN THING-WRAP IT UP AND TAKE IT BACK HOME!

ANOTHER ALIEN'S ON THE LOOSE! DO YOU WANT TO TRY AND FIND IT? NO

## □ Program Listing

10 REM DEEP SPACE SHERIFF 100 PRINT 110 PRINT "YOU ARE THE SHERIFF OF THIS PART OF" 120 PRINT "THE GALAXY. YOU ARE CHASING AN" 130 PRINT "ESCAPED ALIEN DESPERADO. YOU ARE" 140 PRINT "THE ONLY TWO BEINGS ON A LONELY MOON" 150 PRINT "THAT HAS TEN HUGE CRATERS." 160 PRINT 170 PRINT "IF YOU FIND THE ALIEN BEFORE IT" 180 PRINT "FINDS YOU, YOU'LL BE ABLE TO TAKE IT" 190 PRINT "BACK TO PRISON ON EARTH. BUT IF IT" 200 PRINT "FINDS YOU FIRST, IT WILL EAT YOU." 210 PRINT 220 PRINT TAB(7); "HAPPY HUNTING!" 230 PRINT 240 PRINT "WHAT'S YOUR NAME, SHERIFF"; 250 INPUT N\$ 260 IF LEN(N\$) < 2 THEN 240

280 PRINT "HOW MANY DESPERADOS HAVE YOU CAUGHT";

- 290 INPUT R 300 IF R=0 THEN 280
- 310 RANDOMIZE R

270 GOSUB 760

- 320 C=INT(RND(1)\*10)+1
- 330 A=INT(RND(1)\*10)+1
- 340 IF C=A THEN 320
- 350 PRINT
- 360 GOTO 510

```
370 GOSUB 600
380 PRINT "YOU ARE HUNTING IN CRATER "; C
390 IF C=A THEN 650
400 PRINT "THE ALIEN IS IN CRATER "; A; " -- NOT HERE!"
410 GOSUB 600
420 GOSUB 760
430 GOSUB 600
440 A=INT(RND(1)*10)+1
450 IF A=C THEN 710
460 PRINT "THE ALIEN JUST LOOKED FOR YOU"
470 PRINT "IN CRATER "; A; " -- BUT HE MIGHT HAVE"
480 PRINT "GONE SOMEWHERE ELSE BY NOW."
490 PRINT
500 A = INT(RND(1) * 10) + 1
510 PRINT "WHICH CRATER DO YOU WANT TO LOOK IN";
520 INPUT C$
530 C=VAL(C$)
540 IF C>0 AND C<11 THEN 570
550 PRINT "THERE'S NO CRATER "; C; " ON THIS MOON!"
560 GOTO 510
570 GOSUB 600
580 GOSUB 760
590 GOTO 370
600 \text{ FOR B=1 TO } 38
610 PRINT "-";
620 NEXT B
630 PRINT "-"
640 RETURN
650 PRINT "GOOD WORK, SHERIFF "; N$; "!"
660 PRINT "YOU FOUND THE LITTLE GREEN THING--"
670 PRINT "WRAP IT UP AND TAKE IT BACK HOME!"
680 GOSUB 600
690 GOTO 800
700 END
710 PRINT "THE ALIEN SNUCK UP ON YOU FROM BEHIND!"
720 PRINT "SHERIFF "; N$; ", "; " YOU'RE FINGER-LICKIN' GOOD!"
730 GOSUB 600
740 GOTO 800
750 END
760 FOR S=1 TO 3
770 PRINT
780 NEXT S
790 RETURN
800 GOSUB 760
810 PRINT "ANOTHER ALIEN'S ON THE LOOSE!"
820 PRINT "DO YOU WANT TO TRY AND FIND IT";
830 INPUT A$
840 IF MID$(A$,1,1) = "N" THEN END
850 GOSUB 760
860 GOTO 320
```

### $\square$ If You Have . .

### APPLE II

Delete: 310

Change: 340 IF A=C THEN 320

390 IF A=C THEN 650

### **ATARI**

Delete: 310

Add: 95 DIM N\$(40), C\$(2), A\$(3)

Change: 220 PRINT ";"HAPPY HUNTING!"

320 C=INT(RND(0)\*10)+1

330 A = INT(RND(0) \*10) +1

440 A=INT(RND(0)\*10)+1 500 A=INT(RND(0)\*10)+1

600 FOR B=1 TO 36

840 IF  $A$(1,1) = N^{n}$  THEN END

### **COMMODORE 64**

Change: 310 REM

 $320^{\circ} C = INT(RND(0) * 10) + 1$ 

330 A = INT(RND(0) \*10) +1

440 A=INT(RND( $\emptyset$ ) \*1 $\emptyset$ )+1

500 A = INT(RND(0) \*10) +1

#### **COMMODORE VIC-20**

Delete: 210,230,310

Change: 240 PRINT "WHAT'S YOUR NAME, SHERIFF"

280 PRINT "HOW MANY DESPERADOS HAVE YOU CAUGHT"

320 C=INT(RND(0) \*10) +1

330 A=INT(RND(0)\*10)+1

440 A=INT(RND(0)\*10)+1

500 A = INT(RND(0) \* 10) + 1

510 PRINT "WHICH CRATER DO YOU WANT TO LOOK IN"

600 FOR B=1 TO 21

820 PRINT "DO YOU WANT TO TRY AND FIND IT"

### **TEXAS INSTRUMENTS 99/4A**

Add: 522 IF C\$="" THEN 510

885 END

Change: 130 PRINT "ESCAPED ALIEN DESPERADO. YOU ARE"

140 PRINT "THE ONLY TWO BEINGS ON A LONELY MOON"

190 PRINT "BACK TO PRISON ON EARTH. BUT IF IT"

310 RANDOMIZE

320 C=INT(RND\*10)+1 330 A=INT(RND\*10)+1

440 A=INT(RND\*10)+1 500 A=INT(RND\*10)+1

540 IF (C>0)\*(C<11) THEN 570

600 FOR B=1 TO 27

660 PRINT "YOU FOUND THE LITTLE GREEN THING--"

840 IF SEG\$(A\$,1,1)="N" THEN 885

### TRS-80 COLOR COMPUTER

Delete: 310,770,780

Change: 110 PRINT "YOU ARE THE SHERIFF OF THIS PART ";

120 PRINT "THE GALAXY. YOU ARE CHASING AN ";

130 PRINT "ESCAPED ALIEN DESPERADO. YOU ARE ";

140 PRINT "THE ONLY TWO BEINGS ON A LONELY MOON ";

150 PRINT "THAT HAS TEN HUGE CRATERS. ";

170 PRINT "IF YOU FIND THE ALIEN BEFORE IT ";

180 PRINT "FINDS YOU, YOU'LL BE ABLE TO TAKE IT ";

190 PRINT "BACK TO PRISON ON EARTH. BUT IF IT ";

200 PRINT "FINDS YOU FIRST, IT WILL EAT YOU. ";

320 C=RND(10)

330 A=RND(10)

440 A=RND(10)

470 PRINT "IN CRATER "; A; " -- BUT HE MIGHT HAVE ";

500 A=RND(10)

600 FOR I=1 TO 30

760 REM

# Planet of the Shapes

When you live on the Planet of the Shapes, it's a compliment to be called square. And there are a lot of folks who don't mind their pointy heads a bit.

Type in this program and take a visit.

### □ Sample Run

ON THE PLANET OF THE SHAPES, THE PEOPLE HAVE GEOMETRIC FIGURES. WHILE YOU MIGHT THINK THEY LOOK FUNNY, THEY THINK YOU LOOK JUST HILARIOUS!

THE SHAPE KIDS ARE GETTING VERY RESTLESS BECAUSE SCHOOL WILL BE OUT SOON AND THEY'LL GET TO GO HOME TO THEIR TRIANGULAR CATS AND SQUARE DOGS. YOU CAN GET THEM OUT EARLY BY RINGING THE SCHOOL BELL. JUST HIT THE ENTER KEY, AND SCHOOL WILL BE OUT?

```
HERE THEY COME!!!
```





\*\*\*\*\*\*\* \*\*\*\*\*\*\*\* \*\*\*\*\*\*\*\*\*

WHEEEE!

\*\*\*\*\*\*\*\*\*\* \*\*\*\*\*\*\*\* \*\*\*\*

HOORAY!

\*\*\*\*\*\*\* \*\*\*\*\*\*\* \*\*\*\*\*\*

> \* \*\*\* \*\*\*\* \*\*\*\*\*

\*\*\*\*\*\*\* \*\*\*\*\*\*\* \*\*\*\*\*

HOORAY!

\*\*\*\*\*\*\*\*\*\* \*\*\*\*\*\*\*\*\* \*\*\*\*\*\*

OH BOY!

\*\*\*\*\*\*\*\* \*\*\*\*\*\*\*\* \*\*\*\*\*\*\*\*\* \*\*\*\*\*\*\*

OH BOY!

Break

## □ Program Listing

10 REM PLANET OF THE SHAPES

100 PRINT "ON THE PLANET OF THE SHAPES, THE"

110 PRINT "PEOPLE HAVE GEOMETRIC FIGURES."

120 PRINT "WHILE YOU MIGHT THINK THEY LOOK"

130 PRINT "FUNNY, THEY THINK YOU LOOK JUST"

140 PRINT "HILARIOUS!"

150 PRINT

160 PRINT "THE SHAPE KIDS ARE GETTING VERY"

```
170 PRINT "RESTLESS BECAUSE SCHOOL WILL BE"
180 PRINT "OUT, SOON AND THEY'LL GET TO GO"
190 PRINT "HOME TO THEIR TRIANGULAR CATS AND"
200 PRINT "SQUARE DOGS. YOU CAN GET THEM OUT"
210 PRINT "EARLY BY RINGING THE SCHOOL BELL."
220 PRINT "JUST HIT THE ENTER KEY, AND SCHOOL"
230 PRINT "WILL BE OUT";
240 INPUT H$
250 PRINT CHR$(7)
260 PRINT
270 FOR G=1 TO 25
280 PRINT TAB(7); "HERE THEY COME!!!
290 NEXT G
300 FOR V=1 TO 25
310 PRINT
320 NEXT V
330 A=INT(RND(1)*30)+1
340 B=INT(RND(1) * (30-A))+8
350 REM --- BOX ---
360 FOR H=1 TO B/2
370 PRINT TAB(A);
380 FOR T=1 TO B+1
390 PRINT "*":
400 NEXT T
410 NEXT H
420 GOSUB 810
430 REM --- PARALLELOGRAM LEFT -
440 IF A<B+2 THEN 530
450 FOR H=1 TO B/2
460 PRINT TAB(A-H-8);
.470 FOR T=1 TO B+1
480 PRINT "*";
490 NEXT T
500 NEXT H
510 GOSUB 810
520 REM --- TRIANGLE UP
530 IF A<B/2 THEN 330
540 FOR C=1 TO B STEP 2
550 PRINT TAB (A-C/2);
560 FOR D=1 TO C
570 PRINT "*";
580 NEXT D
590 NEXT C
600 GOSUB 810
610 REM --- PARALLELOGRAM RIGHT -
620 FOR H=1 TO B/2
630 PRINT TAB (A+H-8);
640 FOR T=1 TO B+1
650 PRINT "*";
660 NEXT T
670 NEXT H
680 GOSUB 810
690 REM --- TRIANGLE UP --
700 IF A<B/2 THEN 330
```

```
710 IF B/2=4 OR B/2=5 OR B/2=6 OR B/2=7 OR B/2=8 THEN 330
720 IF B/2=9 OR B/2=10 OR B/2=11 THEN 330
730 FOR C=B TO 1 STEP -2
740 PRINT TAB(A~C/2);
750 FOR D=1 TO C
760 PRINT "*";
770 NEXT D
780 NEXT C
790 GOSUB 810
800 GOTO 330
810 FOR S=1 TO 5
820 PRINT
830 NEXT S
840 ON INT(RND(1)*6)+1 GOTO 850,870,890,910,910,910
850 PRINT TAB(10); "HOORAY!"
860 GOTO 910
870 PRINT TAB(10); "OH BOY!"
880 GOTO 910
890 PRINT TAB(10); "WHEEEE!"
900 GOTO 910
910 FOR Y=1 TO 4
920 PRINT
930 NEXT Y
940 RETURN
```

### ☐ If You Have .

### APPLEII

Add: 345 B=B-1 405 PRINT 495 PRINT 585 PRINT 615 IF A<8 THEN 700 617 IF (A+B/2+B)>39 THEN 700 665 PRINT 775 PRINT

Change: 220 PRINT "JUST HIT THE RETURN KEY, AND SCHOOL" 440 IF A<=B+3 THEN 530 530 IF A<=B/2 THEN 330 630 PRINT TAB(A+H); 700 IF A<=B/2+1 THEN 330

### **ATARI**

Add: 95 DIM H\$(1) 405 PRINT 495 PRINT 585 PRINT 665 PRINT

775 PRINT

955 FOR AA=1 TO IDX

965 PRINT " ";: NEXT AA: RETURN

Change: 220 PRINT "JUST HIT THE RETURN KEY, AND SCHOOL"

250 PRINT CHR\$ (253)

280 IDX=6:GOSUB 955:PRINT "HERE THEY COME!!!"

330 A = INT(RND(0) \*28) + 1

340 B=INT(RND(0)\*(28-A))+8

370 IDX=A:GOSUB 955

460 IDX=(A-H-8):GOSUB 955

550 IDX=(A-C/2):GOSUB 955

630 IDX=(A+H-8):GOSUB 955

740 IDX=(A-C/2):GOSUB 955

840 ON INT(RND(0) \*6)+1 GOTO 850,870,890,910,910,910

850 IDX=10:PRINT "HOORAY!"

870 IDX=10:PRINT "OH BOY!"

890 IDX=10:PRINT "WHEEEE!"

#### **COMMODORE 64**

Delete: 250

Add: 405 PRINT

495 PRINT 585 PRINT

665 PRINT 775 PRINT

Change: 220 PRINT "JUST HIT THE RETURN KEY, AND SCHOOL"

330 A=INT(RND(0)\*30)+1

340 B=INT(RND(0) \* (30-A)) +8

840 ON INT(RND(0) \*6)+1 GOTO 850,870,890,910,910,910

### **COMMODORE VIC-20**

Delete: 250

Add: 155 INPUT TP\$

405 PRINT

495 PRINT

585 PRINT

665 PRINT

775 PRINT

Change: 150 PRINT "HIT THE RETURN KEY FOR MORE!"

220 PRINT "JUST HIT THE RETURN KEY, AND SCHOOL"

230 PRINT "WILL BE OUT"

280 PRINT TAB(4); "HERE THEY COME!!!"

330 A = INT(RND(0) \*11) +1

340 B=INT(RND(0)\*(15-A))+4

460 PRINT TAB(A-H-4);

```
630 PRINT TAB(A+H-4);
840 ON INT(RND(0)*6)+1 GOTO 850,870,890,910,910,910
```

### TEXAS INSTRUMENTS 99/4A

Delete: 140

110 PRINT "PEOPLE HAVE GEOMETRIC Change: FIGURES"

120 PRINT "WHILE YOU MIGHT THINK THEY LOOK"

130 PRINT "FUNNY, THEY THINK YOU LOOK JUST HILARIOUS.

160 PRINT "THE SHAPE KIDS ARE GETTING VERY"

190 PRINT "HOME TO THEIR TRIANGULAR" CATS AND" 200 PRINT "SQUARE DOGS. YOU CAN GET THEM OUT"

250 CALL SOUND (100,440,2)

330 A = INT(RND\*26) + 1

340 B=INT(RND\*(20-A))+8

710 IF (B/2=4)+(B/2=5)+(B/2=6)+(B/2=7)+(B/2=8) THEN 330

720 IF (B/2=9)+(B/2=10)+(B/2=11) THEN 330

840 ON INT(RND\*6)+1 GOTO 850,870,890,910,910,910

### TRS-80 COLOR COMPUTER

Add: 405 PRINT

455 IF A-H-8<=0 THEN 510

495 PRINT

585 PRINT

625 IF A+H-8 < = 0 THEN 680

665 PRINT

775 PRINT

100 PRINT "ON THE PLANET OF THE SHAPES, THE Change:

110 PRINT "PEOPLE HAVE GEOMETRIC FIGURES. ";

120 PRINT "WHILE YOU MIGHT THINK THEY LOOK ";

130 PRINT "FUNNY, THEY THINK YOU LOOK JUST ";

160 PRINT "THE SHAPE KIDS ARE GETTING VERY

170 PRINT "RESTLESS BECAUSE SCHOOL WILL BE ";

180 PRINT "OUT SOON AND THEY'LL GET TO GO "; 190 PRINT "HOME TO THEIR TRIANGULAR CATS AND "

200 PRINT "SQUARE DOGS. YOU CAN GET THEM OUT ";

210 PRINT "EARLY BY RINGING THE SCHOOL BELL.

220 PRINT "JUST HIT THE ENTER KEY, AND SCHOOL ";

250 SOUND 173,3

330 A=RND(30)

340 B=RND(30)

840 ON RND(6) GOTO 850,870,890,910,910,910

# **Asteroid Alert**

You can't be too careful in the asteroid beit. Thousands of tiny planetoids can put dents in your spacecraft unless you're careful. But your ship is no match for the biggest ones. That's why every astronaut venturing into the asteroid zone must pass this test.

### ☐ Sample Run

THERE ARE MORE THAN 30,000 ASTEROIDS IN THE SOLAR SYSTEM.

EVEN THE SMALL ONES CAN PACK A REAL WALLOP IF YOUR SPACESHIP COLLIDES WITH ONE. BUT ANY SPACE TRAVELER HAD BETTER KNOW ABOUT THE TEN BIGGEST ONES!

HIT THE ENTER KEY TO CONTINUE?

- 1 ASTRAEA
- 2 CERES
- 3 FLORA
- 4 HEBE
- 5 HYGEIA
- 6 IRIS
- 7 JUNO
- 8 METIS
- 9 PALLAS
- 10 VESTA

ONE OF THE ABOVE ASTEROIDS WAS DISCOVERED IN 1848, HAS A DIAMETER OF 78 MILES, AND IS ROUGHLY 222 MILLION MILES FROM THE SUN.

ENTER THE RIGHT NUMBER? 8

#### RIGHT!

ONE OF THE ABOVE ASTEROIDS WAS DISCOVERED IN 1845, HAS A DIAMETER OF 50 MILES, AND IS ROUGHLY 239 MILLION MILES FROM THE SUN.

ENTER THE RIGHT NUMBER? 1

RIGHT!

[There's more—when you get to the end of the program, you'll see something like this:]

ONE OF THE ABOVE ASTEROIDS WAS DISCOVERED IN 1807, HAS A DIAMETER OF 243 MILES, AND IS ROUGHLY 219 MILLION MILES FROM THE SUN.

ENTER THE RIGHT NUMBER? 10

RIGHT!

IT TOOK YOU 11 TRIES TO GET THEM ALL-DO YOU WANT TO TRY AGAIN? NO

## □ Program Listing

```
10 REM ASTEROID ALERT
100 PRINT "THERE ARE MORE THAN 30,000"
110 PRINT "ASTEROIDS IN THE SOLAR SYSTEM."
120 PRINT
130 PRINT "EVEN THE SMALL ONES CAN PACK"
140 PRINT "A REAL WALLOP IF YOUR SPACESHIP"
150 PRINT "COLLIDES WITH ONE. BUT ANY"
160 PRINT "SPACE TRAVELER HAD BETTER KNOW"
170 PRINT "ABOUT THE TEN BIGGEST ONES!"
180 PRINT
190 PRINT "HIT THE ENTER KEY TO CONTINUE";
200 INPUT X$
210 PRINT
220 DIM F(10), L(10), D(10), N(10)
230 K=1
240 T=T+1
250 PRINT
260 GOSUB 680
270 PRINT
280 PRINT "ONE OF THE ABOVE ASTEROIDS WAS"
290 PRINT "DISCOVERED IN "; F(K); ", HAS A"
300 PRINT "DIAMETER OF "; D(K); " MILES, AND"
310 PRINT "IS ROUGHLY "; L(K); " MILLION MILES"
320 PRINT "FROM THE SUN."
330 PRINT
340 PRINT "ENTER THE RIGHT NUMBER";
350 INPUT Q
360 IF Q<1 OR Q>10 THEN 340
370 IF Q<>N(K) THEN 450
380 PRINT
                                                 (continued)
390 PRINT "RIGHT!"
```

```
400 PRINT
410 K=K+1
420 IF K=11 THEN 550
430 RESTORE
440 GOTO 240
450 RESTORE
460 PRINT
470 PRINT "SORRY--THE ANSWER IS NOT ";
480 FOR C=1 TO Q
490 READ A$
500 NEXT C
510 PRINT A$;"!"
520 PRINT
530 RESTORE
540 GOTO 240
550 PRINT
560 IF X=0 THEN 590
570 IF X<T THEN PRINT "YOU DID BETTER LAST TIME!"
580 IF X>T THEN PRINT "TERRIFIC! YOU DID BETTER THAN
    BEFORE!"
590 PRINT "IT TOOK YOU ";T; " TRIES TO GET THEM ALL--"
600 PRINT "DO YOU WANT TO TRY AGAIN";
610 INPUT Y$
620 IF MID$(Y$,1,1)<>"N" THEN 640
630 END
640 X=T
650 RESTORE
660 GOTO 230
670 REM --- LOOKUP SUBROUTINES
680 FOR Z=1 TO 10
690 READ A$
700 PRINT TAB(5); Z; TAB(10); A$
710 NEXT Z
720 FOR B=1 TO 10
730 READ F(B), L(B), D(B), N(B)
740 NEXT B
750 RETURN
760 DATA ASTRAEA, CERES
770 DATA FLORA, HEBE, HYGEIA
780 DATA IRIS, JUNO, METIS
790 DATA PALLAS, VESTA
800 DATA 1848,222,78,8,1845,239,50,1,1849,293,40,5
810 DATA 1804,248,118,7,1847,204,56,3,1801,257,485,2
820 DATA 1847,225,121,4,1802,257,304,9,1847,221,121,6
830 DATA 1807,219,243,10
```

### $\square$ If You Have . . .

### **APPLE II**

Change: 190 PRINT "HIT THE RETURN KEY TO CONTINUE";

### **ATARI**

Add: 95 DIM X\$(1),A\$(7),Y\$(3)

735 F(B) = FB : L(B) = LB : D(B) = DB : N(B) = NB

Change: 190 PRINT "HIT THE RETURN KEY TO CONTINUE";

620 IF Y\$(1,1) <> "N" THEN 640
700 PRINT " "; Z; " "; A\$

730 READ FB, LB, DB, NB

### **COMMODORE 64**

Change: 190 PRINT "HIT THE RETURN KEY TO CONTINUE"

### **COMMODORE VIC-20**

Change: 190 PRINT "HIT THE RETURN KEY TO CONTINUE"

330 REM

340 PRINT "ENTER THE RIGHT NUMBER" 600 PRINT "DO YOU WANT TO TRY AGAIN"

### **TEXAS INSTRUMENTS 99/4A**

Add: 572 PRINT "YOU DID BETTER LAST TIME!"

582 PRINT "TERRIFIC! YOU DID BETTER THAN BEFORE!"

Change: 110 PRINT "ASTEROIDS IN THE SOLAR SYSTEM"

140 PRINT "A REAL WALLOP IF YOUR SPACE-SHIP"

160 PRINT "SPACE TRAVELER HAD BETTER KNOW"

190 PRINT "HIT THE ENTER KEY TO CONTINUE";

280 PRINT "ONE OF THE ABOVE ASTEROIDS WAS"

360 IF (Q<1)+(Q>10) THEN 340

570 IF X>=T THEN 580 580 IF X<=T THEN 590

620 IF XC-1 THEN 590
620 IF SEG\$(Y\$,1,1)<>"N" THEN 640

### TRS-80 COLOR COMPUTER

Delete: 220,250,270,330,520

Add: 95 DIM F(10), L(10), D(10), N(10)

Change: 440 GOTO 190

540 GOTO 190

# Chinese Calendar Magic

Time is very important in outer space — and calendars are crucial. Here on Earth, the Chinese use an unusual system to count their years.

Discover it by running this program.

## ☐ Sample Run

WHAT IS THE CURRENT YEAR? 1984

THE CHINESE CALENDAR WORKS IN 12-YEAR CYCLES. EVERY YEAR IS NAMED AFTER ONE OF TWELVE DIFFERENT ANIMALS.

ENTER A YEAR, AND I'LL TELL YOU WHICH ANIMAL IT'S NAMED FOR? 1776

IN THE CHINESE CALENDAR, 1776 WAS THE YEAR OF THE MONKEY.

WANT TO TRY ANOTHER YEAR? YES

ENTER A YEAR, AND I'LL TELL YOU WHICH ANIMAL IT'S NAMED FOR? 1990

IN THE CHINESE CALENDAR, 1990 WILL BE THE YEAR OF THE HORSE.

WANT TO TRY ANOTHER YEAR? YES

ENTER A YEAR, AND I'LL TELL YOU WHICH ANIMAL IT'S NAMED FOR? 1984

IN THE CHINESE CALENDAR, 1984 IS THE YEAR OF THE RAT.

WANT TO TRY ANOTHER YEAR? NO

## ☐ Program Listing

10 REM CHINESE CALENDAR MAGIC 100 PRINT "WHAT IS THE CURRENT YEAR"; 110 INPUT C 120 IF C<1983 OR C>1990 THEN 100 130 PRINT

```
140 PRINT "THE CHINESE CALENDAR WORKS IN 12-YEAR"
150 PRINT "CYCLES. EVERY YEAR IS NAMED AFTER ONE"
160 PRINT "OF TWELVE DIFFERENT ANIMALS."
170 PRINT
180 PRINT "ENTER A YEAR, AND I'LL TELL YOU"
190 PRINT "WHICH ANIMAL IT'S NAMED FOR";
200 INPUT Y
210 N=0
220 D=Y-1972
230 IF D<0 THEN N=1
240 D=ABS(D)
250 FOR A=1 TO D STEP 12
260 IF D>12 THEN D=D-12
270 NEXT A
280 IF N=1 THEN D=12-D
290 IF D=12 THEN D=0
300 FOR A=1 TO D+1
310 READ F$
320 NEXT A
330 PRINT
340 PRINT "IN THE CHINESE CALENDAR,"
350 IF Y<C THEN 410
360 IF Y=C THEN 390
370 PRINT Y; WILL BE THE YEAR OF THE "; F$; "."
380 GOTO 420
390 PRINT Y; " IS THE YEAR OF THE "; F$; "."
400 GOTO 420
410 PRINT Y; " WAS THE YEAR OF THE "; F$; "."
420 PRINT
430 PRINT "WANT TO TRY ANOTHER YEAR";
440 INPUT Y$
450 IF MID$(Y$,1,1) = "N" THEN END
460 PRINT
470 RESTORE
480 GOTO 180
490 DATA RAT, OX, TIGER, CAT, DRAGON, SNAKE
500 DATA HORSE, SHEEP, MONKEY, ROOSTER, DOG, PIG
```

### $\Box$ If You Have . . .

**APPLE II** No Changes Required

### **ATARI**

Add: 95 DIM F\$(7),Y\$(3)

Change: 450 IF Y\$(1,1) = "N" THEN END

### **COMMODORE 64** No Changes Required

### **COMMODORE VIC-20**

Change: 100 PRINT "WHAT IS THE CURRENT YEAR"

190 PRINT "WHICH ANIMAL IT'S NAMED FOR"
430 PRINT "WANT TO TRY ANOTHER YEAR"

### TEXAS INSTRUMENTS 99/4A

Add: 232 N=1

262 D=D-12

282 D=12-D

292 D=Ø

525 END

Change:, 120 IF (C<1983)+(C>1990) THEN 100

140 PRINT "THE CHINESE CALENDAR WORKS IN 12-YEAR"

230 IF D>=0 THEN 240

260 IF D<=12 THEN 270

280 IF N<>1 THEN 290

290 IF D<>12 THEN 300

450 IF SEG\$(Y\$,1,1)="N" THEN 525

TRS-80 COLOR COMPUTER No Changes Required

# Giant Diamond Miner\_

It's not easy hunting for diamonds in space. Most of the time you find nothing but useless space rocks. Butkeep trying — you may just strike it rich!

	Sample Run
	HOW MANY DIAMONDS DO YOU THINK YOU ARE GOING TO FIND TODAY? 34512
%	YOU'VE INVESTED YOUR LIFE'S SAVINGS IN A MINE ON URANUS THAT IS SUPPOSED TO PRODUCE GIANT DIAMONDS. YOU HAVE TO DECIDE WHERE TO SEND YOUR MINERS.
21	WHICH SHAFT DO YOU WANT TO TRY (1-9)? 3
	HIT THE ENTER KEY TO START THE CONVEYOR BELT AND WE'LL SEE WHAT COMES OUT?
*	*******  ******  ******  ******  ******
-9	TOO BAD! NOTHING BUT MUDDY ICE
30	MIGHT AS WELL TRY AGAIN.
	WHICH SHAFT DO YOU WANT TO TRY (1-9)? 7
	HIT THE ENTER KEY TO START THE CONVEYOR BELT AND WE'LL SEE WHAT COMES OUT?
ş: (X	********
	******
	*******
ę.	*****************
	********

TOO BAD! NOTHING BUT PETRIFIED DONUTS...
MIGHT AS WELL TRY AGAIN.

WHICH SHAFT DO YOU WANT TO TRY (1-9)? 9

HIT THE ENTER KEY TO START THE CONVEYOR BELT AND WE'LL SEE WHAT COMES OUT?

TOO BAD! NOTHING BUT OILY GRAVEL...

MIGHT AS WELL TRY AGAIN.

WHICH SHAFT DO YOU WANT TO TRY (1-9)? 5

HIT THE ENTER KEY TO START THE CONVEYOR BELT AND WE'LL SEE WHAT COMES OUT?

HOLD ON--IT LOOKS PROMISING!!!!

******	******
******	**********
******	*******
*****	*********
*****	*********
******	*********
******	******
******	*******
**********	*******
******	***********
******	**********
*******	**********
******	*********
*****	********
******	*********
*******	**********
******	**********
*******	***********

Break

### □ Program Listing

```
10 REM GIANT DIAMOND MINER
100 PRINT
110 PRINT "HOW MANY DIAMONDS DO YOU THINK YOU"
120 PRINT "ARE GOING TO FIND TODAY";
130 INPUT R
140 RANDOMIZE R
150 PRINT CHR$(12)
160 PRINT "YOU'VE INVESTED YOUR LIFE'S SAVINGS"
170 PRINT "IN A MINE ON URANUS THAT IS SUPPOSED"
180 PRINT "TO PRODUCE GIANT DIAMONDS. YOU HAVE"
190 PRINT "TO DECIDE WHERE TO SEND YOUR MINERS."
200 J=INT(RND(1)*9)+1
210 PRINT
220 PRINT "WHICH SHAFT DO YOU WANT TO TRY (1-9)";
230 INPUT X
240 IF X<1 OR X>9 THEN 220
250 PRINT
260 PRINT "HIT THE ENTER KEY TO START THE CONVEYOR"
270 PRINT "BELT AND WE'LL SEE WHAT COMES OUT";
280 INPUT K$
290 PRINT
300 IF X<>J THEN 630
310 PRINT
320 PRINT "HOLD ON--IT LOOKS PROMISING!!!!"
330 PRINT
340 A=INT(RND(1)*19)*2-1
350 IF A<3 THEN 340
360 FOR B=1 TO A STEP 2
370 \text{ SIDE} = (40-B)/2
380 FOR L=1 TO SIDE
390 PRINT "*";
400 NEXT L
410 FOR M=1 TO B
420 PRINT " ";
430 NEXT M
440 FOR N=1 TO SIDE-1
450 PRINT "*";
460 NEXT N
470 PRINT "*"
480 NEXT B
490 FOR C=A TO 1 STEP -2
500 \text{ SIDE} = (40-C)/2
510 FOR X=1 TO SIDE
520 PRINT "*";
530 NEXT X
540 FOR Y=1 TO C
550 PRINT " ";
560 NEXT Y
570 FOR Z=1 TO SIDE-1
580 PRINT "*";
590 NEXT Z
600 PRINT "*"
```

```
610 NEXT C
620 GOTO 340
630 FOR W=1 TO 17
640 P=(INT(RND(1)*32))+1
650 Q=(INT(RND(1)*4))+1
660 FOR V=1 TO P
670 PRINT "*";
680 NEXT V
690 FOR H=1 TO Q
700 PRINT CHR$(32);
710 NEXT H
720 FOR F=1 TO 38-(P+Q)
730 PRINT "*";
740 NEXT F
750 PRINT
760 NEXT W
770 PRINT
780 PRINT "TOO BAD! NOTHING BUT ";
790 ON (RND(1) *5)+1 GOTO 800,820,840,860,880
800 PRINT "OILY GRAVEL..."
810 GOTO 890
820 PRINT "SPACE DIRT..."
830 GOTO 890
840 PRINT "MUDDY ICE..."
850 GOTO 890
860 PRINT "PETRIFIED DONUTS..."
870 GOTO 890
880 PRINT "BROKEN FOSSILS..."
890 PRINT
900 PRINT "MIGHT AS WELL TRY AGAIN."
910 GOTO 210
```

### $\square$ If You Have $\dots$

### **APPLE II**

Delete: 140

Change: 150 HOME

260 PRINT "HIT THE RETURN KEY TO START THE CONVEYOR"

#### **ATARI**

Delete: 140

Add: 95 DIM K\$(1)

Change: 150 PRINT CHR\$(125)

200 J=INT(RND(0)\*9)+1

260 PRINT "HIT THE RETURN KEY TO START THE CONVEYOR"

340 A=INT(RND(0) \*19) \*2-1

```
370 SIDE=(38-B)/2

500 SIDE=(38-C)/2

640 P=(INT(RND(0)*32))+1

650 Q=(INT(RND(0)*4))+1

790 ON (RND(0)*5)+1 GOTO 800,820,840,860,880
```

### **COMMODORE 64**

Delete: 140

Change: 150 PRINT CHR\$(147)

200 J = INT(RND(0) \* 9) + 1

260 PRINT "HIT THE RETURN KEY TO START THE CONVEYOR"

340 A=INT(RND(0)\*19)\*2-1 640 P=(INT(RND(0)\*32))+1 650 Q=(INT(RND(0)\*4))+1

790 ON (RND(0) \*5)+1 GOTO 800,820,840,860,880

#### **COMMODORE VIC-20**

Delete: 140

Change: 120 PRINT "ARE GOING TO FIND TODAY"

150 PRINT CHR\$(147) = 200 J=INT(RND(0)\*9)+1

220 PRINT "WHICH SHAFT DO YOU WANT TO TRY (1-9)"

260 PRINT "HIT THE RETURN KEY TO START THE CONVEYOR"

270 PRINT "BELT AND WE'LL SEE WHAT COMES OUT"

340 A=INT (RND(0) \*9) \*2-1

370 SIDE=(22-B)/2 500 SIDE=(22-C)/2

640 P=(INT(RND(0)\*16))+1

650 Q=(INT(RND(0) \*4))+1

720 FOR F=1 TO 19-(P+Q)

790 ON (RND(0) \*5)+1 GOTO 800,820,840,860,880

#### TEXAS INSTRUMENTS 99/4A

Change: 110 PRINT "HOW MANY DIAMONDS DO YOU THINK YOU"
140 RANDOMIZE
150 CALL CLEAR
200 J=INT(RND\*9)+1

220 PRINT "WHICH SHAFT DO YOU WANT TO TRY (1-9)";

240 IF (X<1)+(X>9) THEN 220

260 PRINT "HIT THE ENTER KEY TO START THE CONVEYOR" 270 PRINT "BELT AND WE'LL SEE WHAT COMES OUT";

340 A=INT (RND $^{*}14$ ) \*2-1

370 SIDE = (28-B)/2

500 SIDE = (28-C)/2

640 P = INT(RND\*24) + 1

650 Q = INT(RND\*4) + 1

720 FOR F=1 TO 26-(P+Q)790 ON (RND\*5) +1 GOTO 800,820,840,860,880

### TR8-80 COLOR COMPUTER

Delete: 140

Change: 150 CLS

200 J=RND(9)

340 A=RND(14)\*2-1 370 SIDE=(32-B)/2 500 SIDE = (32-C)/2640 P=RND(26)

650 Q=RND(4) 720 FOR F=1 TO 31-(P+Q) 790 ON RND(5) GOTO 800,820,840,860,880

## Visit to a Backward Planet

You've just sighted the planet Sdrawkcab. Its citizens are friendly — but they're slightly different from us humans. When you run this program, you'll be in computer contact with them!

### ☐ Sample Run

ON THE PLANET SDRAWKCAB, PEOPLE DO SOME THINGS DIFFERENTLY FROM WHAT WE'RE USED TO. HIT THE ENTER KEY TO PUT YOUR COMPUTER INTO CONTACT WITH ONE OF ITS CITIZENS?

WELCOME TO THE PLANET SDRAWKCABI

WHAT'S YOUR NAME, STRANGER? HENRY JAMES
THAT'S FUNNY--MINE IS SEMAJ YRNEH.

WHAT COUNTRY DO YOU COME FROM? CANADA REALLY? MY COUNTRY IS ADANAC.

WHAT CITY DO YOU LIVE IN? ROME INTERESTING...I LIVE IN EMOR.

WHAT IS YOUR STREET ADDRESS? 324 PARADISE STREET GEE, I LIVE AT TEERTS ESIDARAP 423.

HENRY JAMES, TYPE IN HOW YOU COUNT TO 9? 123456789

WELL, WE COUNT TO 9 LIKE THIS: 9. BUT HERE'S HOW WE COUNT ALL THE WAY UP TO 1: 987654321

WHAT'S YOUR BEST FRIEND'S NAME? WALDO EMERSON

NEAT! MY BEST FRIEND HERE IS NOSREME ODLAW. (continued)

WHAT COMPUTER ARE YOU USING? FERN III
WELL, HENRY JAMES, I'M USING MY III NREF.

WHAT'S YOUR FAVORITE FOOD? SNAILS

SNAILS?? YUCK! I LOVE SLIANS.

SORRY, HENRY JAMES, I'VE GOT TO GO.

WHAT DO YOU SAY WHEN YOU LEAVE? TA-TA

AT-AT, HENRY JAMES!

### ☐ Program Listing

360 IF AS="" THEN 340

370 PRINT 380 GOSUB 880

400 GOSUB 940

```
10 REM VISIT TO A BACKWARD PLANET
100 PRINT "ON THE PLANET SDRAWKCAB, PEOPLE"
110 PRINT "DO SOME THINGS DIFFERENTLY FROM WHAT"
120 PRINT "WE'RE USED TO. HIT THE ENTER KEY TO"
130 PRINT "PUT YOUR COMPUTER INTO CONTACT WITH"
140 PRINT "ONE OF ITS CITIZENS";
150 INPUT X$
160 GOSUB 940
170 PRINT "WELCOME TO THE PLANET SDRAWKCAB!"
180 GOSUB 940
190 PRINT "WHAT'S YOUR NAME, STRANGER";
200 INPUT NAM$
210 IF NAMS="" THEN 190
220 PRINT
230 A$=NAM$
240 GOSUB 880
250 PRINT "THAT'S FUNNY--MINE IS "; B$; "."
260 GOSUB 940
270 PRINT "WHAT COUNTRY DO YOU COME FROM";
280 INPUT A$
290 IF A$="" THEN 270
300 PRINT
310 GOSUB 880
320 PRINT "REALLY? MY COUNTRY IS ";B$;"."
330 GOSUB 940
340 PRINT "WHAT CITY DO YOU LIVE IN";
350 INPUT A$
```

390 PRINT "INTERESTING...I LIVE IN ";B\$;"."

```
410 PRINT "WHAT IS YOUR STREET ADDRESS";
420 INPUT A$
430 IF AS=" THEN 410
440 PRINT
450 GOSUB 880
460 PRINT "GEE, I LIVE AT ";B$;"."
470 GOSUB 940
480 PRINT NAMS; ", TYPE IN HOW YOU COUNT TO 9";
490 INPUT AS
    IF AS="" THEN 480
500
510 PRINT
520 GOSUB 880
530 PRINT "WELL, WE COUNT TO 9 LIKE THIS:
540 PRINT "BUT HERE'S HOW WE COUNT ALL THE WAY"
550 PRINT "UP TO 1: "; B$
560 GOSUB 940
570 PRINT "WHAT'S YOUR BEST FRIEND'S NAME";
580 INPUT A$
590 IF A$="" THEN 570
600 PRINT
610 GOSUB 880
620 PRINT "NEAT! MY BEST FRIEND HERE IS "; B$; "."
630 GOSUB 940
640 PRINT "WHAT COMPUTER ARE YOU USING";
650 INPUT A$
660 IF A$="" THEN 640
670 PRINT
680 GOSUB 880
690 PRINT "WELL, "; NAM$; ", I'M USING MY "; B$; "."
700 GOSUB 940
710 PRINT "WHAT'S YOUR FAVORITE FOOD";
720 INPUT A$
730 IF A$="" THEN 710
740 PRINT
750 GOSUB 880
760 PRINT A$; "?? YUCK! I LOVE "; B$; "."
770 GOSUB 940
780 PRINT "SORRY, "; NAM$;", I'VE GOT TO GO."
800 PRINT "WHAT DO YOU SAY WHEN YOU LEAVE";
810 INPUT A$
820 IF A$="" THEN 800
830 GOSUB 880
840 GOSUB 940
850 PRINT B$; ", "; NAM$; "!"
860 END
870 REM --- BACKWARDS SUBROUTINE
880 B$=""
890 FOR A=LEN(A$) TO 1 STEP -1
900 B$=B$+MID$(A$,A,1)
910 NEXT A
920 RETURN
930 REM --- SPACING SUBROUTINE -
940 FOR S=1 TO 5
950 PRINT
960 NEXT S
970 RETURN
```

### ☐ If You Have . . .

### **APPLE II**

Change: 120 PRINT "WE'RE USED TO. HIT THE RETURN KEY TO"

### **ATARI**

Add: 95 DIM X\$(1), NAM\$(40), A\$(40), B\$(40)

Change: 120 PRINT "WE'RE USED TO. HIT THE RETURN KEY TO"

900 B\$(LEN(A\$) +1-A) = A\$(A,A)

### **COMMODORE 64**

Change: 120 PRINT "WE'RE USED TO. HIT THE RETURN KEY TO"

#### **COMMODORE VIC-20**

Change: 120 PRINT "WE'RE USED TO. HIT THE RETURN KEY TO"

140 PRINT "ONE OF ITS CITIZENS"

190 PRINT "WHAT'S YOUR NAME, STRANGER "

270 PRINT "WHAT COUNTRY DO YOU COME FROM"

340 PRINT "WHAT CITY DO YOU LIVE IN"

410 PRINT "WHAT IS YOUR STREET ADDRESS"

480 PRINT NAMS;", TYPE IN HOW YOU COUNT TO 9"

570 PRINT "WHAT'S YOUR BEST FRIEND'S NAME"

640 PRINT "WHAT COMPUTER ARE YOU USING"

710 PRINT "WHAT'S YOUR FAVORITE FOOD"

800 PRINT "WHAT DO YOU SAY WHEN YOU LEAVE"

### **TEXAS INSTRUMENTS 99/4A**

Change: 900 B\$=B\$&SEG\$(A\$,A,1)

TRS-80 COLOR COMPUTER No Changes Required

# \_Measurement Madness\_

You probably know how tall you are in inches — but has anybody ever measured how many picas you are? The unusual measurements in this program come from Earth — but some of them do seem a bit unearthly!

### ☐ Sample Run

IN SPACE, ODD THINGS HAPPEN TO THE MEASUREMENTS WE'RE FAMILIAR WITH. YOU CANW EIGH DIFFERENT AMOUNTS ON DIFFERENT PLANETS. EVEN YOUR HEIGHT CAN VARY WHEN YOU TRAVEL NEAR THE SPEED OF LIGHT.

WE'RE ALL USED TO CERTAIN WEIGHTS AND MEASURES, SUCH AS INCHES, POUNDS, AND GALLONS. THIS PROGRAM WILL SHOW YOU SOME OF THE MORE UNUSUAL ONES.

- 1 INCHES
- 2 QUARTS
- 3 POUNDS

SELECT ONE CATEGORY? 1

HOW MANY INCHES TALL ARE YOU? 47

IF YOU ARE 47 INCHES TALL, YOU ARE

2.611111 CUBITS TALL

1.044444 ENGLISH ELLS TALL

11.75 HANDS TALL

282 PICAS TALL

3384 POINTS TALL

5.222223 SPANS TALL

5.934344 LINKS TALL

#### WANT TO TRY SOME MORE? YES

- 1 INCHES
- 2 QUARTS
- 3 POUNDS

SELECT ONE CATEGORY? 2

HOW MANY QUARTS DO YOU THINK YOU CAN DRINK IN ONE WHOLE YEAR? 543

IF YOU CAN DRINK 543 QUARTS, THEN YOU CAN ALSO SWALLOW

4344 GILLS 4.309524 BARRELS 8.619047 HOGSHEADS 17.23809 PIPES .5386905 TUNS 271.5 MAGNUMS

WANT TO TRY SOME MORE? NO

### ☐ Program Listing

```
10 REM MEASUREMENT MADNESS
100 PRINT "IN SPACE, ODD THINGS HAPPEN TO THE"
110 PRINT "MEASUREMENTS WE'RE FAMILIAR WITH."
120 PRINT "YOU CAN WEIGH DIFFERENT AMOUNTS ON"
130 PRINT "DIFFERENT PLANETS. EVEN YOUR HEIGHT"
140 PRINT "CAN VARY WHEN YOU TRAVEL NEAR THE"
150 PRINT "SPEED OF LIGHT."
160 PRINT
170 PRINT "WE'RE ALL USED TO CERTAIN WEIGHTS"
180 PRINT "AND MEASURES, SUCH AS INCHES, POUNDS,"
190 PRINT "AND GALLONS. THIS PROGRAM WILL SHOW"
200 PRINT "YOU SOME OF THE MORE UNUSUAL ONES."
210 PRINT
220 PRINT TAB(5); "1"; TAB(10); "INCHES"
230 PRINT TAB(5); "2"; TAB(10); "QUARTS"
240 PRINT TAB(5); "3"; TAB(10); "POUNDS"
250 PRINT
260 PRINT "SELECT ONE CATEGORY";
270 INPUT C
280 ON C GOTO 300,450,610
290 GOTO 260
300 PRINT
310 PRINT "HOW MANY INCHES TALL ARE YOU";
320 INPUT T
330 IF T<1 OR T>80 THEN 310
340 PRINT
350 PRINT "IF YOU ARE ";T;" INCHES TALL, YOU ARE"
360 PRINT
370 PRINT TAB(5); T/18; CUBITS TALL
380 PRINT TAB(5); T/45; " ENGLISH ELLS TALL"
390 PRINT TAB(5); T/4; " HANDS TALL"
400 PRINT TAB(5); T*6; " PICAS TALL"
410 PRINT TAB(5); T*72; POINTS TALL
420 PRINT TAB(5); T/9; " SPANS TALL"
430 PRINT TAB(5);T/7.92; LINKS TALL
440 GOTO 730 =
450 PRINT
460 PRINT "HOW MANY QUARTS DO YOU THINK YOU CAN"
470 PRINT "DRINK IN ONE WHOLE YEAR";
```

```
480 INPUT Q
490 IF Q<1 THEN 460
500 PRINT
510 PRINT "IF YOU CAN DRINK ";Q;" QUARTS, THEN"
520 PRINT "YOU CAN ALSO SWALLOW "
530 PRINT
540 PRINT TAB(5); Q*8; " GILLS"
550 PRINT TAB(5);Q/126; BARRELS"
560 PRINT TAB(5);Q/63;" HOGSHEADS"
570 PRINT TAB(5);Q/31.5;" PIPES'
580 PRINT TAB(5);Q/1008; " TUNS"
590 PRINT TAB(5);Q/2; " MAGNUMS"
600 GOTO 730
610 PRINT
620 PRINT "HOW MANY POUNDS DO YOU WEIGH";
630 INPUT P
640 IF P<1 THEN 620
650 PRINT
660 PRINT "IF YOU WEIGH "; P; " POUNDS, THEN YOU"
670 PRINT "ALSO WEIGH"
680 PRINT
690 PRINT TAB(5); P/14; " STONE"
700 PRINT TAB(5); P/220.46; " QUINTAL
710 PRINT TAB(5); P*256; " DRAMS"
720 PRINT TAB(5); P*7000; "GRAINS"
730 PRINT
740 PRINT "WANT TO TRY SOME MORE";
750 INPUT Y$
760 IF MID$(Y$,1,1)<>"N" THEN 210
```

### ☐ If You Have . . .

**APPLE II** No Changes Required

#### **ATARI**

Add: 95 DIM Y\$(3)
765 END
775 FOR AA=1 TO IDX
785 PRINT " ";:NEXT AA: RETURN

Change: 220 IDX=4:GOSUB 775:PRINT "1";:GOSUB 775:PRINT
"INCHES"

230 GOSUB 775:PRINT "2";:GOSUB 775:PRINT "QUARTS"

240 GOSUB 775:PRINT "3";:GOSUB 775:PRINT "POUNDS"

370 IDX=4:GOSUB 775:PRINT T/18; " CUBITS TALL"

380 GOSUB 775:PRINT T/45; " ENGLISH ELLS TALL"

390 GOSUB 775:PRINT T/4; " HANDS TALL"

400 GOSUB 775: PRINT T\*6; PICAS TALL 410 GOSUB 775: PRINT T\*72; POINTS TALL 420 GOSUB 775: PRINT T/9; SPANS TALL 430 GOSUB 775: PRINT T/7.92; LINKS TALL 540 IDX=4:GOSUB 775: PRINT Q\*8; GILLS 550 GOSUB 775: PRINT Q/126; BARRELS 550 GOSUB 775: PRINT Q/63; HOGSHEADS 570 GOSUB 775: PRINT Q/63; HOGSHEADS 570 GOSUB 775: PRINT Q/31.5; PIPES 580 GOSUB 775: PRINT Q/1008; TUNS 590 GOSUB 775: PRINT Q/2; MAGNUMS 690 IDX=4:GOSUB 775: PRINT P/14; STONE 700 GOSUB 775: PRINT P/220.46; QUINTAL 710 GOSUB 775: PRINT P\*256; DRAMS 770 GOSUB 775: PRINT P\*7000; GRAINS 770 GOSUB 775: PRINT P\*7000; GRAINS 7700 IF Y\$(1,1) <> NT THEN 210

### **COMMODORE 64** No Changes Required

#### **COMMODORE VIC-20**

Add: 165 PRINT "HIT RETURN KEY FOR MORE!"

167 INPUT MOR\$

Change: 260 PRINT "SELECT ONE CATEGORY"

310 PRINT "HOW MANY INCHES TALL ARE YOU"

470 PRINT "DRINK IN ONE WHOLE YEAR"

620 PRINT "HOW MANY POUNDS DO YOU WEIGH"

740 PRINT "WANT TO TRY SOME MORE"

### **TEXAS INSTRUMENTS 99/4A**

Delete: 160

Add: 195 PRINT "GALLONS. THIS PROGRAM WILL"

Change: 170 PRINT "WE'RE ALL USED TO CERTAIN"

180 PRINT "WEIGHTS AND MEASURES, SUCH"

190 PRINT "AS INCHES, POUNDS, AND"

200 PRINT "SHOW YOU SOME OF THE MORE UNUSUAL ONES."

330 IF (T<1)+(T>80) THEN 310

760 IF SEG\$(Y\$,1,1)<>"N" THEN 210

#### TRS-80 COLOR COMPUTER

Add: 95 CLS

165 INPUT " (PRESS ENTER FOR MORE) ";Y\$

168 CLS

# Space Hero

You did it! You saved your hometown from enemy aliens! And you did it with your wits, without firing a single laser bolt! Type this program in, and you can read all about it. hero!

### ☐ Sample Run

WHAT'S YOUR FIRST NAME? JOSHUA
WHAT'S YOUR LAST NAME? TREE
ENTER YOUR CITY? MEMPHIS
ENTER YOUR STATE? EGYPT
WHAT KIND OF COMPUTER DO YOU HAVE? ABACUS-64
WHAT'S THE FIRST NAME OF
ONE OF YOUR PARENTS? BEAVER
ENTER A FRIEND'S NAME? EDDIE HASKELL
ENTER AN ENEMY'S NAME? TINKERBELL
WHAT SCHOOL DO YOU ATTEND? ETONIA JUNIOR HIGH
WHO IS YOUR FAVORITE TEACHER? MR. BLISTER

#### EXTRA! EXTRA!

THE EGYPT GAZETTE

JOSHUA SAVES MEMPHIS

MEMPHIS--FLASH! JOSHUA TREE HAS SAVED THE CITY FROM A HORRIBLE FATE. THIS MORNING, EVIL ALIENS FROM A FAR-OFF PLANET SECRETLY INVADED EGYPT. JOSHUA WAS COMING HOME FROM ETONIA JUNIOR HIGH SCHOOL WHEN ONE OF THE HORRIBLE CREATURES ENTERED THE TREE HOME.

THE ALIEN INSISTED THAT NO ONE ON EARTH COULD PERFORM COMPLICATED CALCULATIONS THAT COULD MATCH THEIR OWN. BUT TREE SWITCHED ON AN ABACUS-64 COMPUTER AND SHOWED THE TERRIFYING CREATURE SOME OF THE ASTONISHING THINGS THE MACHINE COULD DO.

(HIT THE ENTER KEY FOR PAGE 2)?

JUST AS TWO OF THESE EVIL ALIENS WERE ABOUT TO EAT BEAVER TREE FOR DINNER, JOSHUA SOLVED A TOUGH MATH PROBLEM ONE OF THE ALIENS HAD GIVEN JOSHUA AS A TEST. THE ALIENS THEN FLED

IN TERROR. THEY RELEASED THE TREES, ALONG WITH EDDIE HASKELL, ONE OF JOSHUA'S CLOSE FRIENDS.

JOSHUA'S TEACHER, MR. BLISTER, PRAISED TREE'S ACTIONS, AND NOTED THAT THE ONLY ONE THE ALIENS HAD MANAGED TO EAT WAS TINKERBELL.

### ☐ Program Listing

```
10 REM SPACE HERO
100 PRINT "WHAT'S YOUR FIRST NAME";
110 INPUT F$
120 IF F$="" THEN 100
130 PRINT "WHAT'S YOUR LAST NAME";
140 INPUT L$
150 IF L$="" THEN 130
160 PRINT "ENTER YOUR CITY";
170 INPUT C$
180 IF C$="" THEN 160
190 PRINT "ENTER YOUR STATE";
200 INPUT S$
210 IF S$="" THEN 190
220 PRINT "WHAT KIND OF COMPUTER DO YOU HAVE";
230 INPUT M$
240 IF M$="" THEN 220
250 PRINT "WHAT'S THE FIRST NAME OF"
260 PRINT "ONE OF YOUR PARENTS";
270 INPUT P$
280 IF P$="" THEN 250
290 PRINT "ENTER A FRIEND'S NAME";
300 INPUT B$
310 IF B$="" THEN 290
320 PRINT "ENTER AN ENEMY'S NAME";
330 INPUT E$
340 IF E$="" THEN 320
350 PRINT "WHAT SCHOOL DO YOU ATTEND";
360 INPUT Z$
370 IF Z$="" THEN 350
380 PRINT "WHO IS YOUR FAVORITE TEACHER";
390 INPUT T$
400 IF T$="" THEN 380
410 FOR A=1 TO 24
420 PRINT
430 NEXT A
440 A$="EXTRA! EXTRA!"
450 GOSUB 990
460 GOSUB 920
470 A$="THE "+S$+" GAZETTE"
480 GOSUB 990
490 GOSUB 920
```

```
500 A$=F$+" SAVES "+C$
510 GOSUB 990
520 PRINT
530 PRINT C$; "--FLASH! "; F$; " "; L$; " HAS"
540 PRINT "SAVED THE CITY FROM A HORRIBLE FATE."
550 PRINT "THIS MORNING, EVIL ALIENS FROM A FAR-"
560 PRINT "OFF PLANET SECRETLY INVADED "; S$;"
570 PRINT F$; " WAS COMING HOME FROM "; Z$
580 PRINT "SCHOOL WHEN ONE OF THE HORRIBLE"
590 PRINT "CREATURES ENTERED THE "; L$; " HOME."
600 PRINT
610 PRINT "THE ALIEN INSISTED THAT NO ONE ON EARTH"
620 PRINT "COULD PERFORM COMPLICATED CALCULATIONS"
630 PRINT "THAT COULD MATCH THEIR OWN. BUT "; L$
640 PRINT "SWITCHED ON A "; M$; " COMPUTER AND"
650 PRINT "SHOWED THE TERRIFYING CREATURE SOME"
660 PRINT "OF THE ASTONISHING THINGS THE MACHINE"
670 PRINT "COULD DO.
680 PRINT
690 P RINT "(HIT THE ENTER KEY FOR PAGE 2)";
700 INPUT XS
710 PRINT
720 PRINT
730 PRINT
740 PRINT "JUST AS TWO OF THESE EVIL ALIENS WERE"
750 PRINT "ABOUT TO EAT "; P$; " "; L$; " FOR DINNER, "
760 PRINT F$; " SOLVED A TOUGH MATH PROBLEM"
770 PRINT "ONE OF THE ALIENS HAD GIVEN "; F$
780 PRINT "AS A TEST.
                        THE ALIENS THEN FLED"
790 PRINT "IN TERROR. THEY RELEASED THE"
800 PRINT LS:
810 IF MID$(L$,LEN(L$),1)="S" THEN PRINT "E";
820 PRINT "S, ALONG WITH "; B$; ", ONE OF"
830 PRINT F$; "'S CLOSE FRIENDS."
840 PRINT
850 PRINT F$;"'S TEACHER, ";T$;", PRAISED"
860 PRINT L$;"'S ACTIONS, AND NOTED THAT THE"
870 PRINT "ONLY ONE THE ALIENS HAD MANAGED"
880 PRINT "TO EAT WAS "; E$; "."
890 PRINT
900 END
910 REM --- LINE DRAWING SUBROUTINE ---
920 FOR Z=1 TO 38
930 PRINT "-":
940 NEXT Z
950 PRINT
960 RETURN
970 RETURN
980 REM --- CENTERING SUBROUTINE ---
990 K=19-LEN(A$)/2
1000 PRINT TAB(K); AS
1010 RETURN
1020 PRINT
```

### 🗌 If You Have . . .

### APPLE II

Change: 690 PRINT "(HIT THE RETURN KEY FOR PAGE 2)";

#### **ATARI**

Add: 91 DIM F\$(20),L\$(20),C\$(40),S\$(14),M\$(40),P\$(20)
92 DIM B\$(40),E\$(40),Z\$(40),T\$(40),A\$(30),X\$(1)
445 L=LEN(A\$)
455 PRINT
475 L=LEN(A\$)+LEN(S\$)+8
485 PRINT S\$;" GAZETTE"
505 L=LEN(F\$)+LEN(C\$)+7
515 PRINT "SAVES ";C\$
995 GOSUB 1035

1035 FOR AA=1 TO K

1045 PRINT " ";: NEXT AA: RETURN

Change: 470 A\$="THE '

500 A\$=F\$

690 PRINT "(HIT THE RETURN KEY FOR PAGE 2)"; 810 IF L\$(LEN(L\$), LEN(L\$)) = "S" THEN PRINT "E";

920 FOR Z=1 TO 36

990 K=19-L/2 1000 PRINT A\$;

### **COMMODORE 64**

Change: 690 PRINT "(HIT THE RETURN KEY FOR PAGE 2)";

#### **COMMODORE VIC-20**

Delete: 600,690,700,710,720,840,890

Add: 605 PRINT "(HIT RETURN KEY FOR PAGE 2)"

606 INPUT X\$

607 PRINT

754 PRINT "(HIT RETURN KEY FOR PAGE 3)"

755 INPUT X\$

756 PRINT

Change: 100 PRINT "WHAT'S YOUR FIRST NAME"

130 PRINT "WHAT'S YOUR LAST NAME"

160 PRINT "ENTER YOUR CITY"

190 PRINT "ENTER YOUR STATE"

220 PRINT "WHAT KIND OF COMPUTER DO YOU HAVE"

260 PRINT "ONE OF YOUR PARENTS"

290 PRINT "ENTER A FRIEND'S NAME"

320 PRINT "ENTER AN ENEMY'S NAME"

- 350 PRINT "WHAT SCHOOL DO YOU ATTEND"
- 380 PRINT "WHO IS YOUR FAVORITE TEACHER"
- 920 FOR Z=1 TO 21
- 990 K=11-LEN(A\$)/2

### T EXAS INSTRUMENTS 99/4A

- Add: 595 PRINT "HIT THE ENTER KEY FOR PAGE 2"
  - 596 INPUT X\$
  - 815 PRINT "E";
  - 835 PRINT "HIT THE ENTER KEY FOR PAGE 4"
  - 836 INPUT X\$
- Change: 470 A\$="THE "&S\$&" GAZETTE"
  - 500 A\$=F\$&" SAVES "&C\$
  - 690 PRINT "HIT THE ENTER KEY FOR PAGE 3"
  - 810 IF SEG\$(L\$,LEN(L\$),1)<>"S" THEN 810
  - 920 FOR Z=1 TO 28
  - 990 K=14-LEN(A\$)/2

### TRS-80COLOR COMPUTER

- Delete: 680,690,700,720,730
  - Add: 525 GOSUB 1105
    - 595 GOSUB 1105
    - 700 GOSUB 1105
    - 835 GOSUB 1105
    - 1105 REM -- NEXT SCREEN ROUTINE --
    - 1115 PRINT
    - 1125 PRINT "(PRESS ENTER FOR THE NEXT PAGE)";
    - 1135 INPUT X\$
    - 1145 CLS
    - 1155 RETURN
- Change: 920 FOR Z=1 TO 32
  - 990 K=16-LEN(A\$)/2

# Pyramid Planet \_\_\_Explorer\_\_\_

When you're searching for distant life forms, it pays to be patient.

Most of the time you'll find dry, dusty planets.

But now and then you'll make discoveries that will astound you.

### $\square$ Sample Run WHAT'S YOUR FIRST INITIAL? W WHAT'S YOUR LAST INITIAL? B YOU'RE AN EXPLORER WHO IS FAMOUS ALL ACROSS THE UNIVERSE FOR YOUR DISCOVERIES. FOR YEARS YOU HAVE HEARD OF A MYSTERIOUS PLANET IN A FAR-OFF GALAXY WHERE AN ANCIENT CIVILIZATION BUILT A LONG STRING OF GIANT PYRAMIDS, EACH WITH ITS OWN REFLECTING POOL. IT WOULD BE YOUR GREATEST DISCOVERY IF YOU LOCATED THESE. THERE ARE TEN PLANETS IN THIS SYSTEM. ENTER WHICH ONE YOU WANT TO CHECK (FROM 1 TO 10)? 3 NO, PLANET #3 IS TOO GREASY TO SUPPORT LIFE. TRY AGAIN \_\_\_\_\_\_ THERE ARE TEN PLANETS IN THIS SYSTEM. ENTER WHICH ONE YOU WANT TO CHECK (FROM 1 TO 10)? 8

THERE ARE TEN PLANETS IN THIS SYSTEM. ENTER WHICH ONE YOU WANT TO CHECK (FROM 1 TO 10)? 2

NO, PLANET #8 IS TOO SMELLY TO SUPPORT LIFE. TRY AGAIN

THERE ARE TEN PLANETS IN THIS SYSTEM. ENTER WHICH ONE YOU WANT TO CHECK (FROM 1 TO 10)? 7

YES...IT TOOK YOU 5 TRIES, BUT YOUR SENSORS INDICATE THIS COULD BE THE ONE. HIT THE ENTER KEY TO TURN YOUR VIEWING SCREEN ON?

> WWW WWWWW **WWWWWWW** WWWWWWWW WWWWWWWWW WWWWWWWWWW WWWWWWWWWWW WWWWWWWWWWWW WWWWWWWWWWWWW **BBBBBBBBBBBBBBBBB BBBBBBBBBBBBBBB BBBBBBBBBBBBBB BBBBBBBBBBBB BBBBBBBBBB BBBBBBBBB BBBBBBB BBBBB BBB** В W WWW WWWWW WWWWWWW **WWWWWWWW WWWWWWWWW** wwwwwwwww WHAMING MEMBANA WWWWWWWWWWWWW **BBBBBBBBBBBBBBB BBBBBBBBBBBBBB BBBBBBBBBBBB BBBBBBBBBB BBBBBBBB BBBBBBB** BBBBB **BBB** В

```
W
        WWW
       WWWWW
       BBBBB
        BBB
         В
         W
        WWW
       WWWWW
      WWWWWWW
     WWWWWWWW
     WWWWWWWWW
    WWWWWWWWWW
  WWWWWWWWWWWWWW
 WWWWWWWWWWWWW
 BBBBBBBBBBBBBBB
   BBBBBBBBBBBBB
    BBBBBBBBBBBB
     BBBBBBBBBB
     BBBBBBBB
      BBBBBBB
       BBBBB
        BBB
         В
         W
        WWW
       WWWWW
      WWWWWWW
     WWWWWWWW
     WWWWWWWWW
    WWWWWWWWWW
   wanamuninimawa
  WWWWWWWWWWWW
 WWWWWWWWWWWWWW
WWWWWWWWWWWWWW
WWWWWWWWWWWWWWWWW
BBBBBBBBBBBBBBBBBBB
 BBBBBBBBBBBBBBBBB
  BBBBBBBBBBBBBBBB
   BBBBBBBBBBBBB
    BBBBBBBBBBBB
     BBBBBBBBBB
     BBBBBBBBB
      BBBBBBB
       BBBBB
        BBB
```

В

Break

### ☐ Program Listing

```
10 REM PYRAMID PLANET EXPLORER
100 M=0
110 PRINT "WHAT'S YOUR FIRST INITIAL";
120 INPUT FIŞ
130 IF FI$="" THEN 110
140 IF LEN(FI$)>1 THEN FI$=MID$(FI$,1,1)
150 PRINT "WHAT'S YOUR LAST INITIAL";
160 INPUT LI$
170 IF LI$="" THEN 150
180 IF LEN(LI$) >1 THEN LI$=MID$(LI$,1,1)
190 RANDOMIZE ASC(FI$) *ASC(LI$)
200 Y = INT(RND(1) * 10) + 1
210 PRINT
220 PRINT "YOU'RE AN EXPLORER WHO IS FAMOUS"
230 PRINT "ALL ACROSS THE UNIVERSE FOR YOUR"
240 PRINT "DISCOVERIES. FOR YEARS YOU HAVE"
250 PRINT "HEARD OF A MYSTERIOUS PLANET IN"
260 PRINT "A FAR-OFF GALAXY WHERE AN ANCIENT"
270 PRINT "CIVILIZATION BUILT A LONG STRING"
280 PRINT "OF GIANT PYRAMIDS, EACH WITH ITS"
290 PRINT "OWN REFLECTING POOL. IT WOULD BE"
300 PRINT "YOUR GREATEST DISCOVERY IF YOU"
310 PRINT "LOCATED THESE."
320 PRINT
330 PRINT "THERE ARE TEN PLANETS IN THIS"
340 PRINT "SYSTEM. ENTER WHICH ONE YOU WANT"
350 PRINT "TO CHECK (FROM 1 TO 10)";
360 INPUT G
370 IF G<1 OR G>10 THEN 330
380 IF G=Y THEN 460
390 GOSUB 770
400 GOSUB 690
410 PRINT "NO, PLANET # ";G; " IS TOO "; W$
420 PRINT "TO SUPPORT LIFE. TRY AGAIN."
430 M=M+1
440 GOSUB 690
450 GOTO 320
460 PRINT
470 IF M=0 THEN 490
480 PRINT "YES...IT TOOK YOU "; M+1; " TRIES, BUT"
490 PRINT "YOUR SENSORS INDICATE THIS COULD BE"
500 PRINT "THE ONE. HIT THE ENTER KEY TO TURN"
510 PRINT "YOUR VIEWING SCREEN ON";
520 INPUT D$
530 A = INT(RND(1)*19)*2-1
540 IF A<3 THEN 530
550 FOR B=1 TO A STEP 2
560 PRINT TAB(20-B/2);
570 FOR F=1 TO B
580 PRINT FIS:
590 NEXT F
                                                  (continued)
600 NEXT B
```

```
610 FOR C=A TO 1 STEP -2
620 PRINT TAB (20-C/2);
630 FOR G=1 TO C
640 PRINT LIS;
650 NEXT G
660 NEXT C
670 GOTO 530
680 REM --- LINE DRAWING SUBROUTINE
690 PRINT
700 FOR L=1 TO 34
710 PRINT "=";
720 NEXT L
730 PRINT
740 PRINT
750 RETURN
760 REM --- READING SUBROUTINE
770 FOR J=1 TO INT(RND(1)*15)+1
780 READ W$
790 NEXT J
800 RESTORE
810 RETURN
820 DATA COLD, HOT, STUPID, ROCKY, AIRLESS, WATERY, DRY, BORING
830 DATA SILLY, GREASY, DIRTY, TINY, SMELLY, DUSTY, STEAMY
```

### $\square$ If You Have $\dots$

### **APPLE II**

Delete: 190

Add: 595 PRINT

655 PRINT

Change: 500 PRINT "THE ONE. HIT THE RETURN KEY TO TURN"

#### **ATARI**

Delete: 190

Add: 95 DIM FI\$(1),LI\$(1),D\$(1),W\$(7)

595 PRINT 655 PRINT

845 FOR AA=1 TO IDX

855 PRINT " ";: NEXT AA: RETURN

Change: 140 IF LEN(FI\$)>1 THEN FI\$=FI\$(1,1)

180 IF LEN(LI\$) > 1 THEN LI\$=LI\$(1,1)

200 Y = INT(RND(0) \*10) +1

500 PRINT "THE ONE. HIT THE RETURN KEY TO TURN"

530 A=INT(RND(0)\*19)\*2-1

```
560 IDX=(20-B/2):GOSUB 845
620 IDX=(20-C/2):GOSUB 845
770 FOR J=1 TO INT(RND(0) *15) +1
```

### **COMMODORE 64**

Delete: 190

Add: 595 PRINT

655 PRINT

Change: 200 Y=INT(RND(0)\*10)+1
500 PRINT "THE ONE. HIT THE RETURN KEY TO TURN"

530 A = INT(RND(0) \*19) \*2-1

770 FOR J=1 TO INT(RND(0) \*15)+1

### **CO MMODORE VIC-20**

Delete: 190

Add: 314 PRINT "HIT THE RETURN KEY FOR MORE!"

315 INPUT MOR\$

595 PRINT 655 PRINT

Change: 110 PRINT "WHAT'S YOUR FIRST INITIAL"

150 PRINT "WHAT'S YOUR LAST INITIAL"

200 Y=INT(RND(0)\*10)+1

350 PRINT "TO CHECK (FROM 1 TO 10)" 500 PRINT "THE ONE. HIT THE RETURN HIT THE RETURN KEY TO TURN"

530 A=INT((RND(0)\*9))\*2-1

560 PRINT TAB(10-B/2)

620 PRINT TAB(10-C/2)

700 FOR L=1 TO 21

#### TEXAS INSTRUMENTS 99/4A

Add: 145 FI \$=SEG\$(FI\$,1,1)

185 LI\$=SEG\$(LI\$,1,1)

Change: 140 IF LEN(FI\$) <=1 THEN 150

180 IF LEN(LI\$) <= 1 THEN 200

190 RANDOMIZE

200 Y=INT(RND\*10)+1

300 PRINT "YOUR GREATEST DISCOVERY"

310 PRINT "IF YOU LOCATED THESE."

370 IF (G<1)+(G>10) THEN 330

530 A=INT(RND\*13)\*2-1

560 PRINT TAB (13-B/2);

620 PRINT TAB (13-C/2);

700 FOR L=1 TO 28

770 FOR J=1 TO INT(RND\*15)+1

### TRS-80 COLOR COMPUTER

```
Delete: 190

Add: 215 CLS
595 PRINT
655 PRINT

Change: 200 Y=RND(10)
220 PRINT "YOU'RE AN EXPLORER WHO IS FAMOUS";
230 PRINT "ALL ACROSS THE UNIVERSE FOR YOUR ";
240 PRINT "DISCOVERIES. FOR YEARS YOU HAVE ";
270 PRINT "CIVILIZATION BUILTA LONG STRING";
280 PRINT "OF GIANT PYRAMIDS, EACH WITH ITS ";
340 PRINT "SYSTEM. ENTER WHICH ONE YOU WANT ";
530 A=RND(15) *2-1
560 PRINT TAB(16-B/2);
620 PRINT TAB(16-C/2);
700 FOR L=1 TO 32
770 FOR J=1 TO RND(15)
```

# Stranded on Mars

You and your robot vehicle have strayed too far from your spacecraft.

But your space suit's oxygen supply won't last forever.

Can you get back to your ship before it's too late?

### ☐ Sample Run

WHAT'S YOUR SPACE SHIP NAMED? YANKEE CLIPPER

YOU ARE A BRAVE SPACE EXPLORER
STRANDED ON MARS DURING A VIOLENT
DUST STORM. YOU CAN'T SEE YOUR
SPACE SHIP, BUT YOU KNOW IT IS
SOMEWHERE AHEAD. FORTUNATELY YOU
ARE RIDING IN A SPECIAL ROBOT CAR
WITH FEET THAT CAN WALK QUICKLY
ACROSS. THE ROUGH MARTIAN LANDSCAPE.
THE ONLY PROBLEM IS THAT YOU ARE
RUNNING LOW ON OXYGEN AND NEED TO
GET BACK TO THE SHIP QUICKLY.

WHICH WAY IS THE SHIP:

- 1 TO THE LEFT
- 2 TO THE RIGHT
- 3 STRAIGHT AHEAD

NOW CHOOSE ONE? 2

\*\*\*

\*\*\*\*\*\*\*\*\*\*\*

\*\*\*

\*\*

GOOD WORK! YOU GUESSED RIGHT AND MADE IT SAFELY BACK!!

### ☐ Program Listing

10 REM STRANDED ON MARS 100 PRINT "WHAT'S YOUR SPACE SHIP NAMED"; 110 INPUT NS 120 IF LEN(N\$) <2 THEN 100 130 RANDOMIZE ASC(MID\$(N\$,1,1))\*ASC(MID\$(N\$,2,1)) 140 X=0 150 Y=INT(RND(1)\*3)+1160 PRINT 170 PRINT "YOU ARE A BRAVE SPACE EXPLORER" 180 PRINT "STRANDED ON MARS DURING A VIOLENT" 190 PRINT "DUST STORM. YOU CAN'T SEE YOUR" 200 PRINT "SPACE SHIP, BUT YOU KNOW IT IS" 210 PRINT "SOMEWHERE AHEAD. FORTUNATELY YOU" 220 PRINT "ARE RIDING IN A SPECIAL ROBOT CAR" 230 PRINT "WITH FEET THAT CAN WALK QUICKLY" 240 PRINT "ACROSS THE ROUGH MARTIAN LANDSCAPE." 250 PRINT "THE ONLY PROBLEM IS THAT YOU ARE" 260 PRINT "RUNNING LOW ON OXYGEN AND NEED TO" 270 PRINT "GET BACK TO THE SHIP QUICKLY." 280 PRINT 290 PRINT "WHICH WAY IS THE SHIP:"

```
300 PRINT
310 PRINT TAB(5); "1"; TAB(8); "TO THE LEFT"
320 PRINT TAB(5); "2"; TAB(8); "TO THE RIGHT"
330 PRINT TAB(5); "3"; TAB(8); "STRAIGHT AHEAD"
340 PRINT
350 PRINT "NOW CHOOSE ONE";
360 INPUT Z
370 IF Z<1 OR Z>3 THEN 350
380 ON Z GOTO 390,430,430
390 P=29
400 Q=1
410 R=-1
420 GOTO 460
430 P≃1
440 Q=29
450 R=1
460 FOR C=1 TO 25
470 PRINT
480 NEXT C
490 FOR D=P TO Q STEP R
500 FOR E=1 TO 5
510 IF Z<3 THEN 540
520 PRINT TAB(5); "**"
530 GOTO 550
540 PRINT TAB(5+D); "**"
550 NEXT E
560 PRINT
570 FOR F=1 TO 5
580 IF Z<3 THEN 610
590 PRINT TAB(9); "**"
600 GOTO 620
610 PRINT TAB(9+D); "**"
620 NEXT F
630 NEXT D
640 PRINT
650 IF Z<>Y THEN 690
660 PRINT "GOOD WORK! YOU GUESSED RIGHT AND"
670 PRINT "MADE IT SAFELY BACK!!"
680 END
690 X=X+1
700 IF X=2 THEN 750
710 PRINT "NO--IT'S DEFINITELY NOT IN THIS"
720 PRINT "DIRECTION. YOU HAVE JUST ENOUGH"
730 PRINT "OXYGEN LEFT FOR ONE MORE GUESS."
740 GOTO 280
750 PRINT "TOO BAD -- YOUR SHIP WAS IN THE OTHER"
760 PRINT "DIRECTION. AND..."
770 PRINT "GASP...NO...MORE...OXYGEN..."
```

### $\square$ If You Have . . .

### **APPLE II**

Delete: 130

### **ATARI**

Delete: 130

Add: 95 DIM N\$(40)

775 END

785 FOR AA=1 TO IDX

795 PRINT " ";:NEXT AA:RETURN

Change: 150 Y=INT(RND(0) \*3)+1

310 IDX=4:GOSUB 785:PRINT "1";:GOSUB 785:PRINT

"TO THE LEFT"

320 GOSUB 785:PRINT "2";:GOSUB 785:PRINT "TO THE

RIGHT"

330 GOSUB 785:PRINT "3";:GOSUB 785:PRINT

"STRAIGHT AHEAD"

390 P=27

440 Q=27

520 IDX=4:GOSUB 785:PRINT "\*\*"

540 IDX=(4+D):GOSUB 785:PRINT "\*\*"

590 IDX=8:GOSUB 785:PRINT "\*\*"

610 IDX=8+D:GOSUB 785:PRINT "\*\*"

### **COMMODORE 64**

Delete: 130

Change: 150 Y=INT(RND(0)\*3)+1

390 P=28 440 Q=28

### **COMMODORE VIC-20**

Delete: 130

Add: 245 PRINT "HIT THE RETURN KEY FOR MORE!"

246 INPUT MOR\$

Change: 100 PRINT "WHAT'S YOUR SPACE SHIP NAMED"

150 Y=INT(RND(0) \*3) +1

350 PRINT "CHOOSE ONE"

390 P=14

440 Q=14

520 PRINT TAB(1);"\*\*"

540 PRINT TAB(1+D); "\*\*"

590 PRINT TAB(5); "\*\*"
610 PRINT TAB(5+D); "\*\*"

### **TEXAS INSTRUMENTS 99/4A**

Add: 285 INPUT DAM\$

Change: 130 RANDOMIZE

150 Y=INT(RND\*3)

280 PRINT "HIT ENTER TO CONTINUE"

370 IF (Z<1)+(Z>3) THEN 350

390 P=13 440 Q=13

### TRS-80 COLOR COMPUTER

Delete: 130

Add: 285 PRINT "(PRESS ENTER FOR NEXT PAGE)"

288 INPUT X\$

289 CLS

Change: 150 Y=RND(3)

170 PRINT "YOU ARE A BRAVE SPACE EXPLORER ";

180 PRINT "STRANDED ON MARS DURING A VIOLENT ";

210 PRINT "SOMEWHERE AHEAD. FORTUNATELY YOU ";

220 PRINT "ARE RIDING IN A SPECIAL ROBOT CAR ";

230 PRINT "WITH FEET THAT CAN WALK QUICKLY ";

240 PRINT "ACROSS THE ROUGH MARTIAN LANDSCAPE.

250 PRINT "THE ONLY PROBLEM IS THAT YOU ARE ";

260 PRINT "RUNNING LOW ON OXYGEN AND NEED TO ";

390 P=21

440 Q=21

720 PRINT "DIRECTION. YOU HAVE JUST ENOUGH ";

# Space Shuttle Pilot

You're piloting the first space shuttle mission beyond Earth's orbit! But your onboard computer isn't working properly. You and the machine will have to help each other or you'il end up stranded in space forever!

Sam	ple	Run

YOU ARE STRANDED ON YOUR WAY BACK FROM THE MOON BECAUSE YOUR COMPUTER HAS DEVELOPED A TRICKY PROBLEM.

IT CAN STILL NAVIGATE PERFECTLY, BUT IT HAS LOST ITS ABILITY TO PERFORM ADDITION OR SUBTRACTION.

YOU MUST HELP IT OR YOU'LL NEVER SEE THE EARTH AGAIN.

HOW MANY SPACE VOYAGES HAVE YOU MADE? 41

YOU ARE 200 THOUSAND MILES FROM HOME--THE COMPUTER NEEDS YOUR HELP TO SOLVE THIS PROBLEM:

31 - 18 =

QUICK--WHAT IS THE ANSWER? 13

RIGHT!!
THE COMPUTER THANKS YOU!

YOU ARE 175 THOUSAND MILES FROM HOME--THE COMPUTER NEEDS YOUR HELP TO SOLVE THIS PROBLEM:

36 - 16 =

QUICK--WHAT IS THE ANSWER? 20

-----

RIGHT!!
GOOD--YOU'RE ON COURSE!

YOU ARE 135 THOUSAND MILES FROM HOME--THE COMPUTER NEEDS YOUR HELP TO SOLVE THIS PROBLEM:

95 - 77 =

QUICK--WHAT IS THE ANSWER? 75

-----

#### WRONG!!

THE ANSWER IS ACTUALLY 18 YOU'RE HEADING FARTHER FROM HOME.

YOU ARE 155 THOUSAND MILES FROM HOME--THE COMPUTER NEEDS YOUR HELP TO SOLVE THIS PROBLEM:

91 - 30 =

QUICK--WHAT IS THE ANSWER? 61

RIGHT! 1 GOOD--YOU'RE ON COURSE! 

YOU ARE 110 THOUSAND MILES FROM HOME--THE COMPUTER NEEDS YOUR HELP

TO SOLVE THIS PROBLEM:

98 - 65

QUICK--WHAT IS THE ANSWER? 33

\*-----

RIGHT!! GOOD--YOU'RE ON COURSE!

YOU ARE 70 THOUSAND MILES FROM HOME--THE COMPUTER NEEDS YOUR HELP TO SOLVE THIS PROBLEM:

73 - 64 =

QUICK--WHAT IS THE ANSWER? 9

\_\_\_\_\_\_

RIGHT!!

NICE WORK--EARTH IS GETTING CLOSER!

YOU ARE 20 THOUSAND MILES FROM HOME--THE COMPUTER NEEDS YOUR HELP TO SOLVE THIS PROBLEM:

96 + 37 =

QUICK--WHAT IS THE ANSWER? 133

\_\_\_\_\_

RIGHT!!

YOU DID IT! YOU MADE IT SAFELY HOME! WANT TO TRY AGAIN (YES OR NO)? NO

### □ Program Listing

```
10 REM SPACE SHUTTLE PILOT
 100 PRINT
 110 PRINT "YOU ARE STRANDED ON YOUR WAY BACK"
 120 PRINT "FROM THE MOON BECAUSE YOUR COMPUTER"
 130 PRINT "HAS DEVELOPED A TRICKY PROBLEM."
 140 PRINT
150 PRINT "IT CAN STILL NAVIGATE PERFECTLY,"
160 PRINT "BUT IT HAS LOST ITS ABILITY TO"
170 PRINT "PERFORM ADDITION OR SUBTRACTION."
180 PRINT
190 PRINT "YOU MUST HELP IT OR YOU'LL NEVER"
 200 PRINT "SEE THE EARTH AGAIN."
 210 PRINT
 220 PRINT "HOW MANY SPACE VOYAGES HAVE YOU MADE";
230 INPUT R
 240 RANDOMIZE R
 250 PRINT
 260 D=200
 270 PRINT "YOU ARE ";D; " THOUSAND MILES FROM"
 280 PRINT "HOME--THE COMPUTER NEEDS YOUR HELP"
 290 PRINT "TO SOLVE THIS PROBLEM:"
 300 A=INT(RND(1)*100)+1
310 B=INT(RND(1) *100) +1
 320 IF A<B+1 THEN 300
 330 PRINT
 340 \text{ if } INT(RND(1) *2) = 0 \text{ THEN } 400
 350 PRINT TAB(5); A; TAB(10); "+"; TAB(13); B; TAB(20); "="
360 GOSUB 860
370 R=A+B
380 IF Z=R THEN 450
 390 GOTO 680
 400 PRINT TAB(5); A; TAB(10); "-"; TAB(13); B; TAB(20); "="
410 GOSUB 860
 420 R=A-B
430 IF Z=R THEN 450
440 GOTO 680
450 REM --- RIGHT ANSWER ---
 460 D=D-(INT(RND(1)*10)+1)*5
470 GOSUB 920
 480 PRINT TAB(16); "RIGHT!!
490 PRINT
500 IF D>0 THEN 580
 510 PRINT TAB(2); "YOU DID IT! YOU MADE IT SAFELY HOME!"
520 PRINT
530 PRINT TAB(5); "WANT TO TRY AGAIN (YES OR NO)";
540 INPUT Y$
550 IF MID(Y$,1,1) = "N" THEN END
560 GOSUB 920
570 GOTO 250
580 ON INT(RND(1)*4)+1 GOTO 590,610,630,650
590 PRINT TAB(8); "GOOD--YOU'RE ON COURSE!"
600 GOTO 660
```

```
610 PRINT TAB(5); "YOU'RE HEADING CLOSER TO HOME!"
 620 GOTO 660
 630 PRINT TAB(8); "THE COMPUTER THANKS YOU!"
 640 GOTO 660
 650 PRINT TAB(2); "NICE WORK--EARTH IS GETTING CLOSER!"
 660 GOSUB 920
 670 GOTO 270
 680 REM --- WRONG ANSWER ---
 690 D=D+(INT(RND(1)*5)+1)*5
 700 GOSUB 920
 710 PRINT
 720 PRINT TAB(16); "WRONG!!"
 730 PRINT
 740 PRINT TAB(7); "THE ANSWER IS ACTUALLY "; R
 750 PRINT
 760 ON INT(RND(1) *4) +1 GOTO 770,790,810,830
 770 PRINT TAB (4); "YOU'RE DRIFTING WELL OFF COURSE."
 780 GOTO 840
 790 PRINT TAB(4); "YOU'RE HEADING FARTHER FROM HOME."
 800 GOTO 840
 810 PRINT TAB(2); "YOU'RE RAPIDLY MOVING AWAY FROM EARTH."
 820 GOTO 840
 830 PRINT TAB(2); "YOU'D BETTER GET THE NEXT ONE RIGHT!"
 840 GOSUB 920
 850 GOTO 270
 860 PRINT
 870 PRINT "QUICK--WHAT IS THE ANSWER";
 880 INPUT Z
 890 IF Z<1 THEN 860
 900 RETURN
 910 REM --- LINE SUBROUTINE ---
920 PRINT
 930 FOR L=1 TO 39
 940 PRINT "=";
 950 NEXT L
 960 PRINT
 970 PRINT
 980 RETURN
```

### $\square$ If You Have $\dots$

#### APPLEII

Delete: 240

#### **ATARI**

Delete: 240

Add: 95 DIM Y\$(3)

355 GOSUB 995:PRINT B;:IDX=5:GOSUB 995:PRINT "="

```
405 GOSUB 995: PRINT B;:IDX=5:GOSUB 995: PRINT "="
        995 FOR AA=1 TO IDX
        1005 PRINT " ";:NEXT AA: RETURN
Change: 300 \text{ A=INT}(RND(0)*100)+1
        310 B=INT(RND(0) *100) +1
        340 IF INT(RND(0) \pm2) = 0 THEN 400
        350 IDX=4:GOSUB 995:PRINT A;:GOSUB 995:PRINT "+";
        400 IDX=4:GOSUB 995:PRINT A;:GOSUB 995:PRINT "-";
        460 D=D-(INT(RND(0)*10)+1)*5
        480 IDX=13:GOSUB 995:PRINT "RIGHT!!"
        510 IDX=2:GOSUB 995:PRINT "YOU DID IT! YOU MADE
            IT SAFELY HOME!"
        530 IDX=5:GOSUB 995:PRINT "WANT TO TRY AGAIN (YES
            OR NO)";
        550 IF Y$(1,1) = N^n THEN END
        580 ON INT(RND(0) *4)+1 GOTO 590,610,630,650
        590 IDX=8:GOSUB 995:PRINT "GOOD--YOU'RE ON COURSE!"
        610 IDX=5:GOSUB 995:PRINT "YOU'RE HEADING CLOSER.
            TO HOME!
        630 IDX=8:GOSUB 995:PRINT "THE COMPUTER THANKS YOU!"
        650 IDX=2:GOSUB 995:PRINT "NICE WORK--EARTH IS
            GETTING CLOSER!
        690 D=D+(INT(RND(0) *5) +1) *5
        720 IDX=16:GOSUB 995:PRINT "WRONG!!"
        740 IDX=7:GOSUB 995:PRINT "THE ANSWER IS ACTUALLY "; R
        760 ON INT(RND(0)*4)+1 GOTO 770,790,810,830
        770 IDX=4:GOSUB 995:PRINT "YOU'RE DRIFTING WELL OFF
            COURSE."
            IDX=4:GOSUB 995:PRINT "YOU'RE HEADING FARTHER
        790
            FROM HOME. "
        810 IDX=2:GOSUB 995:PRINT "YOU'RE RAPIDLY MOVING"
            AWAY FROM HOME."
            IDX=2:GOSUB 995: PRINT "YOU'D BETTER GET THE
```

#### **COMMODORE 64**

Delete: 240

Change: 300 A=INT(RND(0) \*100)+1

930 FOR L=1 TO 37

310 B=INT(RND(0) \*100)+1

NEXT ONE RIGHT!"

340 IF INT(RND(0)  $\pm$  2)=0 THEN 400

460 D=D-(INT(RND(0) \*10)+1) \*5

580 ON INT(RND(0) \*4)+1 GOTO 590,610,630,650

690 D=D+(INT(RND(0)\*5)+1)\*5

760 ON INT(RND(0) \*4)+1 GOTO 770,790,810,830

#### **COMMODORE VIC-20**

Delete: 240

```
Change: 220 PRINT "HOW MANY SPACE VOYAGES HAVE YOU MADE"
300 A=INT(RND(0)*100)+1
310 B=INT(RND(0)*100)+1
340 IF INT(RND(0)*2)=0 THEN 400
460 D=D-(INT(RND(0)*10)+1)*5
480 PRINT TAB(7); "RIGHT!!"
530 PRINT TAB(5); "WANT TO TRY AGAIN (YES OR NO)"
580 ON INT(RND(0)*4)+1 GOTO 590,610,630,650
690 D=D+(INT(RND(0)*5)+1)*5
760 ON INT(RND(0)*4)+1 GOTO 770,790,810,830
870 PRINT "QUICK--WHAT IS THE ANSWER"
930 FOR L=1 TO 21
```

### **TEXAS INSTRUMENTS 99/4A**

Add: 555 END

Change: 240 RANDOMIZE

300 A=INT(RND\*100)+1

310 B=INT(RND\*100)+1

340 IF INT(RND\*2)=0 THEN 400

460 D=D-(INT(RND\*10)+1)\*5

550 IF SEG\$(Y\$,1,1)<>"N" THEN 560

580 ON INT(RND\*4)+1 GOTO 590,610,630,650
690 D=D+(INT(RND\*5)+1)\*5

760 ON INT(RND\*4)+1 GOTO 770,790,810,830

### TRS-80COLOR COMPUTER

930 FOR L=1 TO 28

Delete: 240

Add: 846 PRINT "(PRESS ENTER TO CONTINUE)"
848 INPUT X\$:CLS

Change: 150 PRINT "IT CAN STILL NAVIGATE PERFECTLY, ";
170 PRINT "PERFORM ADDITION OR SUBTRACTION. ";
190 PRINT "YOU MUST HELP IT OR YOU'LL NEVER ";
300 A=RND(100)
310 B=RND(100)
340 IF RND(2)-1=0 THEN 400
460 D=D-(RND(10))\*5
580 ON RND(4) GOTO 590,610,630,650
690 D=D+RND(5)\*5
760 ON RND(4) GOTO 770,790,810,830
930 FOR L=1 TO 31

# Interplanetary Art \_\_\_Gallery\_\_\_\_

Here's the greatest art collection of the twenty-first century!

Even Rembrandt would be jealous!

			2												
	Sample Ru	Ŋ													
	CITIZENS OF THE Y TO THE INTERPLANE JUST STEP ONTO TH AND ADMIRE THE GE OF OUR MOST MODER JUST HIT THE ENTE	TA E OM N	R M ME A	Y O' Ti	A VI RI	RNCS	T G T	G W P.A.	AI AI	LK NI	E W VI	R: A: N( W	Z . Z		
	+++++++++++ +++++++++++ ++++++++++++ ++++														
	OOOOOHHH!!	==	=	==	-	=									
5)		==	:= := :=	== ==	==	=	*			.1					
1,7							>> >>	>> >>	>: >: >: >:	>> >>	>	>: >:	>		kS.
	00000ннн г				80										
									:		•			:	:
		::	-				-		-	-	-		_	:	•

00000НН!1

GURGEUUS	STUFFI			
		 	•	

GORGEOUS STUFF!

GORGEOUS STUFF!

BEAUTIFUL!!

Break

### ☐ Program Listing

10 REM INTERPLANETARY ART GALLERY

100 PRINT "CITIZENS OF THE YEAR 2090, WELCOME"

110 PRINT "TO THE INTERPLANETARY ART GALLERY."

120 PRINT "JUST STEP ONTO THE MOVING WALKWAY"

130 PRINT "AND ADMIRE THE GEOMETRIC PAINTINGS"

140 PRINT "OF OUR MOST MODERN ARTISTS. NOW"

```
150, PRINT "JUST HIT THE ENTER KEY TO BEGIN":
160 INPUT H$
170 FOR W=1 TO 25
180 PRINT
190 NEXT W
200 K=1
210 A = INT(RND(1) * 30) + 1
220 B=INT(RND(1) \star (30-A))+8
230 K=K+1
240 IF K>8 THEN K=1
250 ON K GOTO 260,280,300,320,340,360,380,400
260 A$="."
270 GOTO 410
280 A$="/"
290 GOTO 410
300 A$="-"
310 GOTO 410
320 A$="+"
330 GOTO 410
340 A$="="
350 GOTO 410
360 A$=">"
370 GOTO 410
380 A$=":"
390 GOTO 410
400 A$="@"
410 FOR H=1 TO B/2
420 PRINT TAB(A);
430 FOR T=1 TO B+1
440 PRINT AS;
450 NEXT T
460 NEXT H
470 FOR S=1 TO 5
480 PRINT
490 NEXT S
500 ON (RND(1) *9)+1 GOTO 510,530,550,570,210,210,210,210,210
510 PRINT TAB(5); "BEAUTIFUL(1"
520 GOTO 580
530 PRINT TAB(5); "OOOOOHHH!!"
540 GOTO 580
550 PRINT TAB(5); GORGEOUS STUFF!
560 GOTO 580
570 PRINT TAB(5); "FABULOUS!!"
580 FOR Q=1 TO 5
590 PRINT
600 NEXT Q
610 GOTO 210
```

## ☐ If You Have . . .

#### APPLE II

Add: 455 PRINT

Change: 150 PRINT "JUST HIT THE RETURN KEY TO BEGIN";

#### **ATARI**

Add: 95 DIM H\$(1), A\$(1)

455 PRINT

625 FOR AA=1 TO IDX

635 PRINT " ";: NEXT AA: RETURN

Change: 150 PRINT "JUST HIT THE RETURN KEY TO BEGIN";

210 A=INT(RND(0) \*28)+1

220 B=INT(RND(0)  $\star$  (28-A))+8

420 IDX=A: GOSUB 625

500 ON (RND(0) \*9) +1 GOTO 510,530,550,570,210,210,

210,210,210

510 IDX=5:GOSUB 625:PRINT "BEAUTIFUL!!"

530 IDX=5:GOSUB 625:PRINT "OOOOOHHH!!"

550 IDX=5:GOSUB 625:PRINT "GORGEOUS STUFF!"

570 IDX=5:GOSUB 625:PRINT "FABULOUS!!"

#### **COMMODORE 64**

Add: 455 PRINT

Change: 150 PRINT "JUST HIT THE ENTER KEY TO BEGIN";

210 A=INT(RND(0) \*30) +1

220 B=INT(RND(0) \*(30-A))+8

500 ON (RND(0)\*9)+1 GOTO 510,530,550,570,210,210,

210,210,210

#### **COMMODORE VIC-20**

Add: 455 PRINT

Change: 150 PRINT "JUST HIT THE ENTER KEY TO BEGIN"

210 A=INT(RND(0) \*15)+1

220 B=INT(RND(0) \*(15-A))+5

500 ON (RND(0) \*9)+1 GOTO 510,530,550,570,210,210, 210,210,210

#### **TEXAS INSTRUMENTS 99/4A**

Add: 245 K=1

Change: 210 A=INT(RND\*30)+1

220 B=INT(RND\*(27-A))+1 240 IF K<=8 THEN 250

500 ON INT(RND\*9)+1 GOTO 510,530,550,570,210,210,

210,210,210

#### TRS-80 COLOR COMPUTER

Add: 455 PRINT

Change: 210 A=RND (23)

220 B=RND(23-A)+7

500 ON RND(9) GOTO 510,530,550,570,210,210,210,

210,210

# Bust the Scales

You may not have a weight problem here on Earth. But on another planet, it could be a different story. Here's one way to gain or lose weight fast!

## ☐ Sample Run

HOW MANY POUNDS DO YOU WEIGH? 120

- 1 THE SUN
- 2 MERCURY
- 3 VENUS
- 4 THE MOON
- 5 MARS
- 6 JUPITER
- 7 SATURN
- 8 URANUS
- 9 NEPTUNE
- 10 PLUTO

TO FIND OUT HOW MUCH YOU WOULD WEIGH IF YOU LIVED IN ANOTHER PART OF THE SOLAR SYSTEM, CHOOSE A NUMBER? 2

ON EARTH, YOU WEIGH 120 POUNDS. BUT ON MERCURY YOU'D WEIGH 33.6 POUNDS.

WANT TO CHECK ANOTHER PLACE? YES

TO FIND OUT HOW MUCH YOU WOULD WEIGH IF YOU LIVED IN ANOTHER PART OF THE SOLAR SYSTEM, CHOOSE A NUMBER? 6

ON EARTH, YOU WEIGH 120 POUNDS. BUT ON JUPITER YOU'D WEIGH 312 POUNDS.

WANT TO CHECK ANOTHER PLACE? YES

TO FIND OUT HOW MUCH YOU WOULD WEIGH IF YOU LIVED IN ANOTHER PART OF THE SOLAR SYSTEM, CHOOSE A NUMBER? 10

ON EARTH, YOU WEIGH 120 POUNDS. BUT ON PLUTO YOU'D WEIGH 16.8 POUNDS.

WANT TO CHECK ANOTHER PLACE? NO

## □ Program Listing

```
10 REM BUST THE SCALES
100 PRINT
110 PRINT "HOW MANY POUNDS DO YOU WEIGH";
120 INPUT Y
130 IF Y<30 THEN 110
140 PRINT
150 FOR J=1 TO 10
160 READ A$
170 READ X$
180 PRINT TAB(5); J; TAB(10); A$
190 NEXT J
200 RESTORE
210 PRINT
220 PRINT "TO FIND OUT HOW MUCH YOU WOULD"
230 PRINT "WEIGH IF YOU LIVED IN ANOTHER"
240 PRINT "PART OF THE SOLAR SYSTEM,"
250 PRINT "CHOOSE A NUMBER";
260 INPUT L
270 IF L<1 OR L>10 THEN 250
280 FOR I=1 TO L
290 READ P$
300 READ F$
310 NEXT I
320 PRINT
330 RESTORE
340 G=VAL(F$)
350 GOSUB 450
360 PRINT "ON EARTH, YOU WEIGH ";Y;" POUNDS.
370 PRINT "ON "; P$; " YOU'D WEIGH "; G*Y; " POUNDS. "
380 GOSUB 450
390 PRINT
400 PRINT "WANT TO CHECK ANOTHER PLACE";
410 INPUT Y$
420 IF MID$(Y$,1,1) <>"N" THEN 140
430 END
440 REM --- LINE SUBROUTINE
450 PRINT
460 FOR L=1 TO 39
470 PRINT "=";
480 NEXT L
490 PRINT
500 PRINT
```

510 RETURN

520 DATA THE SUN, 28, MERCURY, .28, VENUS, .85

530 DATA THE MOON, .16, MARS, .38, JUPITER, 2.6

540 DATA SATURN, 1.2, URANUS, 1.1, NEPTUNE, 1.4

550 DATA PLUTO,.14

## If You Have.

APPLEII No Changes Required

#### **ATARI**

Add: 95 DIM A\$(8), X\$(4), P\$(8), F\$(4), Y\$(1)

565 FOR AA=1 TO IDX

575 PRINT " ";:NEXT AA:RETURN

Change: 180 IDX=4:GOSUB 565:PRINT J;:GOSUB 565:PRINT A\$

420 IF Y\$(1,1) <>"N" THEN 140

460 FOR L=1 TO 37

#### COMMODORE 64 No Changes Required

#### **COMMODORE VIC-20**

Change: 110 PRINT "HOW MANY POUNDS DO YOU WEIGH"

250 PRINT "CHOOSE A NUMBER"
400 PRINT "WANT TO CHECK ANOTHER PLACE"
460 FOR L=1 TO 21

#### TEXAS INSTRUMENTS 99/4A

Change: 270 IF (L<1)+(L>10) THEN 250

420 IF SEG\$(Y\$,1,1) <> "N" THEN 140

460 FOR L=1 TO 28

#### TRS-80 COLOR COMPUTER

Change: 460 FOR L=1 TO 31

## **First Bank of Pluto**

How does a bank on far-off Pluto stay in business? By offering interest rates that are truly astronomical! The bank will pay you terrific interest rates — every year, every month, or even every day! Make a deposit — you can double your marsmarks in no time!

## ☐ Sample Run

HI! I'M YOUR FRIENDLY COMPUTER BANKING TERMINAL, ATTACHED DIRECTLY TO THE FIRST INTERSTELLAR BANK OF PLUTO. I CAN DEPOSIT YOUR MONEY VIA SPACELINK AND CALCULATE YOUR INTEREST, TAKING INTO ACCOUNT THE AMOUNT, THE INTEREST RATE, AND COMPOUNDING.

HOW MUCH WOULD YOU LIKE TO SAVE? \$20000

#### IS THAT IN:

ä	1	PESETAS		2	RUBLES
	3	<b>PIASTERS</b>		4	POUNDS
	5 ′	BAHTS		6	DINARS
	7	RUPEES	100	8	DRACHMAS
	9	RIALS		10	KRONA
	11	WON		12	ZLOTYS
	13	FRANCS		14	<b>ESCUDOS</b>
	15	LEKS		16	LIRA
	17	YEN		18	QUETZALS
	19	KWACHA		20	GUILDER
	21	CRUZEIROS	-	22	SCHILLINGS
	23	KOPECKS		24	MARKS
	25	<b>FLORINS</b>		26	STIVERS
	27	GUARANI		28	RANDS
	29	RUPEES		30	PATACAS
	31	PESOS	734	32	DOLLARS
	33	MARSMARKS		34	SATURNSKINS
	35	VENUSBUCKS		36	PLUTOPENCE

#### CHOOSE A NUMBER? 35

OKAY. NOW ENTER THE INTEREST RATE--

FROM 2 TO 50%? 50%

HOW OFTEN SHOULD WE PAY YOU 50% INTEREST ON YOUR VENUSBUCKS?

- 1 EVERY DAY
- 2 EVERY WEEK
- 3 EVERY MONTH
- 4 EVERY YEAR

#### CHOOSE A NUMBER? 1

HOW MANY DAYS DO YOU WANT TO KEEP YOUR 20000 VENUSBUCKS TIED UP? 7

- 30000 VENUSBUCKS
- 45000 VENUSBUCKS
- 3 67500 VENUSBUCKS
- 101250 VENUSBUCKS
- 151875 VENUSBUCKS
- **227812.5 VENUSBUCKS**
- **341718.8 VENUSBUCKS**

AFTER 7 DAYS YOU WOULD HAVE 341718.8 VENUSBUCKS.

WANT TO TRY AGAIN? NO

## ☐ Program Listing

```
10 REM FIRST BANK OF PLUTO
```

- 100 PRINT "HI! I'M YOUR FRIENDLY COMPUTER BANKING"
- 110 PRINT "TERMINAL, ATTACHED DIRECTLY TO THE"
- 120 PRINT "FIRST INTERSTELLAR BANK OF PLUTO. I"
- 130 PRINT "CAN DEPOSIT YOUR MONEY VIA SPACELINK"
- 140 PRINT "AND CALCULATE YOUR INTEREST, TAKING"
- 150 PRINT "INTO ACCOUNT THE AMOUNT, THE INTEREST"
- 160 PRINT "RATE, AND COMPOUNDING."
- 170 K=0
- 180 PRINT
- 190 PRINT "HOW MUCH WOULD YOU LIKE TO SAVE";
- 200 INPUT A\$
- 210 IF A\$="" THEN 190
- 220 IF MID\$(A\$,1,1) = "\$" THEN A\$=MID\$(A\$,2,LEN(A\$)-1)
- 230 N=VAL(A\$)
- 240 IF N>1 THEN 290 250 PRINT
- 260 PRINT 1??? NOT MUCH POINT IN THAT!
- 270 PRINT "TRY AGAIN, PLEASE..."
- 280 GOTO 190
- 290 IF N<1 THEN 170
- 300 PRINT
- 310 PRINT "IS THAT IN:"
- 320 PRINT
- 330 FOR M=1 TO 18
- 340 K=K+1
- 350 GOSUB 1070
- 360 PRINT TAB(2); K; TAB(6); M\$;
- 370 K=K+1
- 380 GOSUB 1070
- 390 PRINT TAB(20); K; TAB(24); M\$
- 400 NEXT M
- 410 PRINT

```
420 PRINT "CHOOSE A NUMBER";
430 INPUT C$
440 IF C$="" THEN 420
450 C=VAL(C$)
460 IF C<1 OR C>36 THEN 420
470 K=C
480 GOSUB 1070
490 Z$=M$
500 PRINT
510 PRINT "OKAY. NOW ENTER THE INTEREST RATE--"
520 PRINT "FROM 2 TO 50%";
530 INPUT R$
540 IF R$="" THEN 510
550 IF MID$(R$, LEN(R$), 1) = {}^{n} * THEN R$=MID$(R$, 1, LEN(R$) - 1)
560 R=VAL(R$)
570 IF R<2 OR R>50 THEN 510
580 PRINT
590 PRINT "HOW OFTEN SHOULD WE PAY YOU"
600 PRINT R;"% INTEREST ON YOUR "; Z$; "?"
610 PRINT
620 FOR M=1 TO 36
630 READ M$
640 NEXT M
650 FOR L=1 TO 4
660 READ P$
670 PRINT TAB(5); L; TAB(9); "EVERY "; MID$(P$,1,LEN(P$)-1)
680 NEXT L
690 RESTORE
700 PRINT
710 PRINT "CHOOSE A NUMBER";
720 INPUT FS
730 F=VAL(F$)
740 IF F<1 OR F>4 THEN 710
750 L=F
760 K=36
770 GOSUB 1070
780 PRINT
790 PRINT "HOW MANY "; P$; " DO YOU WANT TO"
800 PRINT "KEEP YOUR ";N;" ";Z$;" TIED UP";
810 INPUT T$
820 T=VAL(T$)
830 IF T<1 THEN 790
840 PRINT
850 FOR E=1 TO T
860 N=N*(1+(R/100))
870 S=INT(N*100+.5)/100
880 PRINT TAB(5); E; TAB(10); S; " "; Z$
890 IF J<19 THEN 950
900 J=0
910 PRINT
920 PRINT "HIT THE ENTER KEY TO CONTINUE";
930 INPUT X$
940 PRINT
950 J=J+1
960 NEXT E
```

```
970 PRINT
980 PRINT "AFTER ";T;" ";P$;" YOU WOULD HAVE"
990 PRINT S; " "; Z$; "."
1000 PRINT
1010 PRINT "WANT TO TRY AGAIN";
1020 INPUT YS
1030 IF MID$(Y$,1,1) = "N" THEN END
1040 J=0
1050 GOTO 170
1060 REM --- READ SUBROUTINE ---
1070 FOR A=1 TO K
1080 READ M$
1090 NEXT A
1100 FOR B=1 TO L
1110 READ P$
1120 NEXT B
1130 RESTORE
1140 RETURN
1150 DATA PESETAS, RUBLES, PIASTERS, POUNDS
1160 DATA BAHTS, DINARS, RUPEES, DRACHMAS
1170 DATA RIALS, KRONA, WON, ZLOTYS
1180 DATA FRANCS, ESCUDOS, LEKS, LIRA, YEN
1190 DATA QUETZALS, KWACHA, GUILDER, CRUZEIROS
1200 DATA SCHILLINGS, KOPECKS, MARKS
1210 DATA FLORINS, STIVERS, GUARANI, RANDS
1220 DATA RUPEES, PATACAS, PESOS, DOLLARS
1230 DATA MARSMARKS, SATURNSKINS, VENUSBUCKS, PLUTOPENCE
1240 DATA DAYS, WEEKS, MONTHS, YEARS
```

### ☐ If You Have...

#### APPLE II

Change: 920 PRINT "HIT THE RETURN KEY TO CONTINUE";

#### **ATARI**

```
Add: 95 DIM A$(7),M$(11),C$(5),Z$(11),R$(3)
96 DIM P$(11),F$(5),T$(5),Y$(3),X$(1)
355 IX=3:IF K>9 THEN IX=2
375 LT=LEN(M$):LX=2
385 IX=15-LT
675 GOSUB 1255:PRINT_" EVERY ";P$(1,LEN(P$)-1)
1255 FOR AA=1 TO IDX
1265 PRINT " ";:NEXT AA:RETURN

Change: 220 IF A$(1,1)="$" THEN A$=A$(2,LEN(A$)-1)
360 IDX=2:GOSUB 1255:PRINT K;:IDX=IX:GOSUB 1255:
PRINT M$;
390 IDX=IX:GOSUB 1255:PRINT K;:IDX=LX:GOSUB 1255:
PRINT M$
(continued)
```

550 IF R\$(LEN(R\$),LEN(R\$)) = "%" THEN R\$=
 R\$(1,LEN(R\$)-1)
670 IDX=4:GOSUB 1255:PRINT L;
880 IDX=4:GOSUB 1255:PRINT E;:GOSUB 1255:PRINT S;
 ";2\$
920 PRINT "HIT THE RETURN KEY TO CONTINUE";
1030 IF Y\$(1,1) = "N" THEN END

#### **COMMODORE 64**

Change: 920 PRINT "HIT THE RETURN KEY TO CONTINUE";

#### **COMMODORE VIC-20**

Add: 395 IF K<>12 AND K<>24 THEN 400

396 PRINT "HIT THE RETURN KEY TO SEE MORE"

397 INPUT MOR\$

Change: 190 PRINT "HOW MUCH WOULD YOU LIKE TO SAVE"

360 PRINT TAB(2); K; TAB(6); M\$

390 PRINT TAB(2); K; TAB(6); M\$

420 PRINT "CHOOSE A NUMBER"

520 PRINT "FROM 2 TO 50%"

710 PRINT "CHOOSE A NUMBER"

800 PRINT "KEEP YOUR ";N; " ";Z\$; " TIED UP"

880 PRINT TAB(1); E; TAB(6); S; " ": 2\$

890 PRINT IF J<14 THEN 950

920 PRINT "HIT THE RETURN KEY TO CONTINUE"

1010 PRINT "WANT TO TRY AGAIN"

#### **TEXAS INSTRUMENTS 99/4A**

Add: 225 A=SEG\$(A\$,2,LEN(A\$)-1)

555 R\$=SEG\$(R\$,1,LEN(R\$)-1)

1035 END

Change: 220 IF SEG\$(A\$,1,1)<>"\$" THEN 230

360 PRINT K; TAB(5); M\$;

390 PRINT TAB(14); K; TAB(18); M\$

460 IF (C<1)+(C>36) THEN 420

550 IF SEG\$(R\$,LEN(R\$),1) <> "%" THEN 560

570 IF (R<2)+(R>50) THEN 510

670 PRINT TAB(5); L; TAB(9); "EVERY ";

SEG\$(P\$,1, LEN(P\$)-1)

740 IF (F<1)+(F>4) THEN 710

1030 IF SEG\$(Y\$,1,1) <>"N" THEN 1035

#### TRS-80 COLOR COMPUTER

Add: 395 IF K<>12 THEN 400

396 PRINT "HIT RETURN TO SEE MORE"

397 INPUT XS

Change: 390 PRINT TAB(16); K; TAB(20); M\$

# Planetary Birthday Party\_\_\_\_

How many birthdays have you had here on Earth? On some planets you'd have had a lot more — but on others you'd still be waiting for your first birthday party! Run this program and find out which planets use up the most birthday candles!

## ☐ Sample Run

WANT TO KNOW HOW MANY YEARS OLD YOU WOULD BE ON OTHER PLANETS? I CAN FIGURE IT OUT FOR YOU.

HOW MANY YEARS OLD ARE YOU? 12

AND HOW MANY MONTHS  $(\emptyset-11)$ ? 4

ON EARTH, YOU ARE ROUGHLY 4504 DAYS OLD-OR 12.33 YEARS OLD. NOW LET'S FIND OUT HOW OLD YOU'D BE ON THESE PLANETS.

- 1 MERCURY
- 2 VENUS
- 3 EARTH
- 4 MARS
- 5 JUPITER
- 6 SATURN
- 7 URANUS
- 8 NEPTUNE
- 9 PLUTO

SELECT A PLANET BY NUMBER? 9

ON PLUTO YOU'D BE .05 YEARS OLD.

WANT TO TRY ANOTHER PLANET? YES

SELECT A PLANET BY NUMBER? 1

ON MERCURY YOU'D BE 51.19 YEARS OLD.

WANT TO TRY ANOTHER PLANET? YES

SELECT A PLANET BY NUMBER? 5

ON JUPITER YOU'D BE 1.04 YEARS OLD.

WANT TO TRY ANOTHER PLANET? NO

## □ Program Listing

```
10 REM PLANETARY BIRTHDAY PARTY
100 PRINT "WANT TO KNOW HOW MANY YEARS OLD YOU"
110 PRINT "WOULD BE ON OTHER PLANETS?
120 PRINT "FIGURE IT OUT FOR YOU."
130 PRINT
140 PRINT "FIRST, I NEED TO KNOW YOUR EARTH AGE--"
150 PRINT "HOW MANY YEARS AND MONTHS OLD YOU ARE."
160 PRINT
170 PRINT "HOW MANY YEARS OLD ARE YOU";
180 INPUT Y
190 IF Y<1 OR Y>100 THEN 170
200 PRINT
210 PRINT "AND HOW MANY MONTHS (0-11)";
220 INPUT M
230 IF M<0 OR M>11 THEN 210
240 PRINT
250 T=(Y*365.26)+(M*30.44)
260 PRINT
270 PRINT "ON EARTH, YOU ARE ROUGHLY "; INT(T); " DAYS"
280 PRINT "OLD--OR "; INT(T/365.26*100+.5)/100; " YEARS OLD."
290 PRINT "NOW LET'S FIND OUT HOW OLD YOU'D BE"
300 PRINT "ON THESE PLANETS."
310 PRINT
320 FOR B=1 TO 9
330 Z=B
340 GOSUB 610
350 PRINT TAB(5);B;TAB(15);P$
360 NEXT B
370 PRINT
380 PRINT "SELECT A PLANET BY NUMBER";
390 INPUT N
400 IF N<1 OR N>9 THEN 380
410 PRINT
420 Z=N
430 GOSUB 610
440 GOSUB 530
450 PRINT "ON "; P$; " YOU'D BE ";
460 PRINT INT(T/D(N) *100+.5)/100; " YEARS OLD."
470 GOSUB 530
```

```
480 PRINT "WANT TO TRY ANOTHER PLANET";
490 INPUT Y$
500 IF MID$(Y$,1,1) = "N" THEN END
510 GOTO 310
520 REM --- LINE SUBROUTINE ---
530 PRINT
540 FOR L=1 TO 38
550 PRINT "=";
560 NEXT L
570 PRINT
580 PRINT
590 RETURN
600 REM --- READING SUBROUTINE ---
610 FOR C=1 TO 9
620 READ D(C)
630 NEXT C
640 FOR A=1 TO Z
650 READ P$
660 NEXT A
670 RESTORE
680 RETURN
690 DATA 88,224.7,365.26
700 DATA 687,4332.7,10759.7
710 DATA 30686.3,60192,90740.6
720 DATA MERCURY, VENUS, EARTH
730 DATA MARS, JUPITER, SATURN
740 DATA URANUS, NEPTUNE, PLUTO
```

## $\square$ If You Have . . .

**APPLEII** No Changes Required

#### **ATARI**

Add: 95 DIM D(9),Y\$(3),P\$(7)
625 D(C)=DC
755 FOR AA=1 TO IDX
765 PRINT "";:NEXT AA:RETURN

Change: 350 IDX=5:GOSUB 755:PRINT B;:IDX=8:GOSUB 755:
PRINT P\$
500 IF Y\$(1,1)="N" THEN END
540 FOR L=1 TO 36
620 READ DC

**COMMODORE 64** No Changes Required

#### **COMMODORE VIC-20**

Change: 170 PRINT "HOW MANY YEARS OLD ARE YOU"

210 PRINT "AND HOW MANY MONTHS (0-11)"

350 PRINT TAB(5);B;TAB(10);P\$

380 PRINT "SELECT A PLANET BY NUMBER"
480 PRINT "WANT TO TRY ANOTHER PLANET"

540 FOR L=1 TO 21

#### **TEXAS INSTRUMENTS 99/4A**

Add: 505 END

Change: 190 IF (Y<1)+(Y>100) THEN 170

230 IF (M<0)+(M>11) THEN 210 400 IF (N<1)+(N>9) THEN 380

500 IF SEG\$(Y\$,1,1)<>"N" THEN 510

540 FOR L=1 TO 28

#### TRS-80 COLOR COMPUTER

Change: 270 PRINT "ON EARTH, YOU ARE ROUGHLY"; INT(T)

280 PRINT "DAYS OLD--OR "; INT(T/365.26\*100+.5)/100;

"YEARS OLD."

290 PRINT "NOW LET'S FIND OUT HOW OLD"

300 PRINT "YOU'D BE ON THESE PLANETS"

540 FOR L=1 TO 31

# Beautiful Stars

Want to see stars? They're right here on your computer screen!

## ☐ Sample Run

WE'RE OUT HERE IN THE COUNTRY, FAR AWAY FROM HARSH CITY LIGHTS. THE SKY IS A DEEP VELVETY BLACK. YOU CAN SEE THOUSANDS AND THOUSANDS OF STARS, EVEN THE VERY FAINT ONES SHIMMERING IN THE MILKY WAY. SIT BACK AND GET READY TO ENJOY THE BEAUTIFUL DISPLAY, THEN HIT "ENTER"?

THE STARS ARE ALL OUT TONIGHT

AND ISN'T IT A BEAUTIFUL SIGHT?

#### HOW MANY PEOPLE NEAR THOSE STARS

ARE STARING UP IN THE SKY AT US

AND SAYING HOW BEAUTIFUL WE LOOK?

THE STARS ARE ALL OUT TONIGHT

Break

## □ Program Listing

```
10 REM BEAUTIFUL STARS
100 PRINT "WE'RE OUT HERE IN THE COUNTRY, FAR"
110 PRINT "AWAY FROM HARSH CITY LIGHTS.
120 PRINT "SKY IS A DEEP VELVETY BLACK.
130 PRINT "CAN SEE THOUSANDS AND THOUSANDS OF"
140 PRINT "STARS, EVEN THE VERY FAINT ONES"
150 PRINT "SHIMMERING IN THE MILKY WAY. SIT"
160 PRINT "BACK AND GET READY TO ENJOY THE"
170 PRINT "BEAUTIFUL DISPLAY, THEN HIT ";
180 PRINT CHR$(34); "ENTER"; CHR$(34);
190 INPUT W$
200 PRINT CHR$(12)
210 A = INT(RND(1) * 40)
220 PRINT TAB(A); "*"
23Ø OLDA=A
240 IF A+OLDA>77 THEN 260
250 GOTO 210
260 S=S+1
270 IF S>5 THEN S=1
280 ON S GOTO 290,310,330,350,370
290 PRINT TAB(6); "THE STARS ARE ALL OUT TONIGHT"
300 GOTO 210
310 PRINT TAB(4); "AND ISN'T IT A BEAUTIFUL SIGHT?"
320 GOTO 210
330 PRINT TAB(4); "HOW MANY PEOPLE NEAR THOSE STARS"
340 GOTO 210
350 PRINT TAB(5); "ARE STARING UP IN THE SKY AT US"
360 GOTO 210
370 PRINT TAB(4); "AND SAYING HOW BEAUTIFUL WE LOOK?"
380 GOTO 210
```

## $\square$ If You Have . . .

#### **APPLEII**

Change: 180 PRINT CHR\$(34); "RETURN"; CHR\$(34); 200 HOME

#### **ATARI**

Add: 95 DIM w\$(1) 395 FOR AA=1 TO IDX

405 PRINT " "; :NEXT AA: RETURN

Change: 180 PRINT CHR\$ (34); "RETURN"; CHR\$ (34);

200 PRINT CHR\$(125) 210 A=INT(RND(0)\*40)

220 IDX=A:GOSUB 395:PRINT "\*"

290 IDX=5:GOSUB 395:PRINT "THE STARS ARE OUT TONIGHT"

310 IDX=3:GOSUB 395:PRINT "AND ISN'T IT A BEAUTIFUL SIGHT?"

330 IDX=3:GOSUB 395:PRINT "HOW MANY PEOPLE NEAR THOSE STARS"

350 IDX=4:GOSUB 395:PRINT "ARE STARING UP IN THE SKY AT US"

370 IDX=3:GOSUB 395:PRINT "AND SAYING HOW BEAUTIFUL WE LOOK?"

#### **COMMODORE 64**

Change: 180 PRINT CHR\$(34); "RETURN"; CHR\$(34);

200 PRINT CHR\$(147) 210 A=INT(RND(0)\*40)

#### **COMMODORE VIC-20**

Change: 180 PRINT CHR\$(34); "RETURN"; CHR\$(34)

200 PRINT CHR\$(147) 210 A=INT(RND(0)\*40)

#### **TEXAS INSTRUMENTS 99/4A**

Add: 275 S=1

Change: 200 CALL CLEAR,

210 A=INT(RND\*40)

270 IF S<=5 THEN 280

#### TRS-80 COLOR COMPUTER

Add: 215 FOR I=1 TO 100:NEXT I

Change: 200 CLS

210 A=RND(40)

# Jupiter Stock \_\_Exchange\_\_

Want to be an interplanetary trader? With this program, you can make a fortune — or lose one! The Jupiter stock market awaits your investment.

## Sample Run

YOU'RE STUCK ON JUPITER, SO YOU MAY AS WELL MAKE SOME MONEY WHILE YOU'RE HERE. THE JUPITER STOCK EXCHANGE WILL LET YOU BUY AND SELL STOCKS FOR FROM 2 TO 52 WEEKS. HOW MANY WEEKS WOULD YOU LIKE TO TRY? 8

#### IT IS NOW WEEK 1

YOUR STOCKS ARE NOW WORTH \$ 0 YOUR BANK BALANCE IS \$ 1000 YOU DON'T OWN ANY STOCKS RIGHT NOW.

#### \*\*\*\*\*THE AVAILABLE STOCKS ARE\*\*\*\*

ID	#	NAME	PI	RICE
1		MARTIAN GOLDMINES	\$	17
2		SATURN SILVER	\$	12
3		PLUTO DIAMONDS LTD	\$	47
4		LUNAR LASER	\$	102
5		RIGEL ROCKETS CO	\$	102
6		PHOTON TORPEDO INC	\$	152
7		VENUS SPACE SUITS	\$	2
8	22	NEPTUNE TRAVEL	\$	202

WILL YOU SELL, BUY, OR QUIT (S,B,Q)? B

WHICH ONE DO YOU WANT TO BUY--ENTER THE ID NUMBER? 7

VENUS SPACE SUITS SELLS FOR \$2 HOW MANY SHARES DO YOU WANT TO BUY? 450

#### IT IS NOW WEEK 2

THE MARKET WENT UP BY 6 POINTS! YOUR STOCKS ARE NOW WORTH \$3600 YOUR BANK BALANCE IS \$100 YOU OWN THE FOLLOWING STOCKS:

#### 450 SHARES VENUS SPACE SUITS (#7)

#### \*\*\*\*\*THE AVAILABLE STOCKS ARE\*\*\*\*

ID	#.	NAME	-	PI	RICE
1		MARTIAN GOLDMINES	Į <sup>2</sup>	\$	23
2		SATURN SILVER	-	\$	18
3		PLUTO DIAMONDS LTD		\$	53
4		LUNAR LASER		\$	108
5		RIGEL ROCKETS CO		\$	108
6	7	PHOTON TORPEDO INC		\$	158
7		VENUS SPACE SUITS		\$	8
8		NEPTUNE TRAVEL		\$	208

WILL YOU SELL, BUY, OR QUIT (S,B,Q)? S

YOU OWN THE FOLLOWING STOCKS:

450 SHARES VENUS SPACE SUITS (#7)

WHICH ONE DO YOU WANT TO SELL--ENTER THE ID NUMBER? 7

YOU OWN 450 SHARES OF VENUS SPACE SUITS HOW MANY SHARES DO YOU WANT TO SELL? 440

[There's more — when you get to the end of the program, you'll see something like this:]

#### IT IS NOW WEEK 8

THE MARKET WENT DOWN BY 9 POINTS! YOUR STOCKS ARE NOW WORTH \$830 YOUR BANK BALANCE IS \$2846 YOU OWN THE FOLLOWING STOCKS:

- 5 SHARES SATURN SILVER (#2)
- 7 SHARES LUNAR LASER (#4)
- 4 SHARES VENUS SPACE SUITS (#7)

#### \*\*\*\*\*THE AVAILABLE STOCKS ARE\*\*\*\*

ID #	NAME	PRICE
1	MARTIAN GOLDMINES	\$ 20
2	SATURN SILVER	\$ 15
3	PLUTO DIAMONDS LTD	\$ 50
4	LUNAR LASER	\$ 105
5	RIGEL ROCKETS CO	\$ 105
6	PROTON TORPEDO INC	\$ 155
7	VENUS SPACE SUITS	\$ 5
8	NEPTUNE TRAVEL	\$ 205

WILL YOU SELL, BUY, OR QUIT (S,B,Q)? S

YOU OWN THE FOLLOWING STOCKS:

- 5 SHARES SATURN SILVER (#2)
- 7 SHARES LUNAR LASER (#4)
- 4 SHARES VENUS SPACE SUITS (#7)

WHICH ONE DO YOU WANT TO SELL--ENTER THE ID NUMBER? 4

YOU OWN 7 SHARES OF LUNAR LASER HOW MANY SHARES DO YOU WANT TO SELL? 7

YOU STARTED OUT WITH \$1000, AND AFTER 9 WEEKS YOU ARE NOW WORTH \$4411--

WHAT A TERRIFIC INTERSTELLAR WHEELER-DEALER YOU ARE!

## □ Program Listing

- 10 REM JUPITER STOCK EXCHANGE
- 100 DIM S(8),P(8)
- 110 BK=1000
- 120 FOR G=1 TO 8
- 130 P(G)=G\*5\*INT(RND(1)\*6)+2
- 140 NEXT G
- 150 PRINT "YOU'RE STUCK ON JUPITER, SO YOU MAY"
- 160 PRINT "AS WELL MAKE SOME MONEY WHILE YOU'RE"
- 170 PRINT "HERE. THE JUPITER STOCK EXCHANGE"
- 180 PRINT "WILL LET YOU BUY AND SELL STOCKS FOR"
- 190 PRINT "FROM 2 TO 52 WEEKS. HOW MANY WEEKS"
- 200 PRINT "WOULD YOU LIKE TO TRY";
- 210 INPUT WK
- 220 IF WK<2 OR WK>52 THEN 150
- 230 RANDOMIZE WK
- 240 PRINT
- 250 REM --- MAIN LOOP ---
- 260 PRINT
- 270 NW=NW+1
- 280 IF NW=WK+1 THEN 1360
- 290 GOSUB 1760
- 300 PRINT TAB(11); "IT IS NOW WEEK "; NW
- 310 GOSUB 1760
- 320 IF NW=1 THEN 520
- 330 UDR=INT(RND(1)\*2)
- 340 IF UDR=1 THEN 380
- 350 UDR=-1
- 360 UD\$="DOWN"
- 370 GOTO 390

```
380 UD$="UP"
390 AMT=INT(RND(1)*10)
400 IF AMT>0 THEN 430
410 PRINT "THE MARKET IS UNCHANGED."
420 GOTO 470
430 I F AMT>1 THEN 460
440 PRINT "THE MARKET WENT "; UD$; " BY "; AMT; " POINT!"
450 GOTO 470
460 PRINT "THE MARKET WENT "; UD$; " BY "; AMT; " POINTS!"
470 FOR F=1 TO 8
480 P(F) = P(F) + AMT * UDR
490 IF P(F) < 2 THEN P(F) = 1
500 NEXT F
510 TW=0
520 FOR J=1 TO 8
530 W=S(J)*P(J)
540 TW=TW+W
550 NEXT J
560 PRINT "YOUR STOCKS ARE NOW WORTH $"; TW
570 PRINT "YOUR BANK BALANCE IS $"; BK
580 GOSUB 1510
590 PRINT "*****THE AVAILABLE STOCKS ARE*****
600 PRINT
610 PRINT TAB(1); "ID #"; TAB(8); "----NAME----;
    TAB (30); "PRICE"
620 FOR Z=1 TO 8
630 GOSUB 1260
640 PRINT TAB(1); Z; TAB(8); SN$; TAB(30); "$"; P(Z)
650 NEXT 2
660 PRINT
670 PRINT "WILL YOU SELL, BUY, OR QUIT (S,B,Q)";
680 INPUT DC$
690 IF DC$="Q" THEN 1360
700 IF DC$="S" THEN 730
710 IF DC$="B" THEN 1020
720 GOTO 670
730 REM --- SELLING MODULE ---
740 PRINT
750 GOSUB 1510
760 PRINT "WHICH ONE DO YOU WANT TO SELL--"
770 PRINT "ENTER THE ID NUMBER";
780 INPUT R$
790 R=VAL(R$)
800 IF R<1 OR R>8 THEN 760
810 IF S(R)>0 THEN 880
820 Z=R
830 GOSUB 1260
840 PRINT
850 PRINT "YOU DON'T OWN "; SN$; "!"
860 PRINT "TRY AGAIN."
870 GOTO 740
880 Z=R
890 GOSUB 1260
900 PRINT
910 PRINT "YOU OWN "; S(R); " SHARES OF "; SN$
```

```
920 PRINT "HOW MANY SHARES DO YOU WANT TO SELL";
930 INPUT NS
940 IF NS<=S(R) THEN 980
950 PRINT
960 PRINT "YOU DON'T OWN "; NS; " SHARES. TRY AGAIN!"
970 GOTO 740
980 S(R)=S(R)-NS
990 BK=BK+NS*P(R)
1000 GOTO 260
1010 REM --- BUYING MODULE ---
1 020 PRINT
1030 PRINT "WHICH ONE DO YOU WANT TO BUY--"
1040 PRINT "ENTER THE ID NUMBER";
1050 INPUT H$
1060 H=VAL(H$)
1070 IF H<1 OR H>8 THEN 1030
1080 Z=H
1090 GOSUB 1260
1100 PRINT
1110 PRINT SN$;" SELLS FOR $ ";P(H)
1120 PRINT "HOW MANY SHARES DO YOU WANT TO BUY";
1130 INPUT U$
1140 U=VAL(U$)
1150 IF U<1 THEN 1120
1160 IF U*P(H) <= BK THEN 1210
1170 PRINT "SORRY. YOU DON'T HAVE ENOUGH MONEY."
1180 PRINT "TRY AGAIN!"
1190 PRINT
1200 GOTO 590
1210 BK=BK-U*P(H)
1220 S(H)=S(H)+U
1230 GOTO 260
1240 END
1250 REM --- READING SUBROUTINE -
1260 IF DONE=1 THEN RESTORE
1270 FOR D=1 TO Z
1280 READ SN$
1290 NEXT D
1300 DONE=1
1310 RETURN
1320 DATA MARTIAN GOLDMINES, SATURN SILVER, PLUTO DIAMONDS
     LTD
1330 DATA LUNAR LASER, RIGEL ROCKETS CO, PHOTON TORPEDO INC
1340 DATA VENUS SPACE SUITS, NEPTUNE TRAVEL
1350 REM --- ENDING MODULE ---
1360 PRINT
1370 PRINT "YOU STARTED OUT WITH $ 1000, AND AFTER"
1380 PRINT NW;" WEEKS YOU ARE NOW WORTH $";TW+BK;"--"
1390 PRINT
1400 IF TW+BK<>1000 THEN 1430
1410 PRINT "A LOT OF WORK FOR NOT MUCH REWARD!"
1420 END
1430 IF TW+BK>1000 THEN 1460
1440 PRINT "TOUGH LUCK. WHY DON'T YOU TRY AGAIN?"
1450 END
                                                  (continued)
```

```
1460 PRINT "WHAT A TERRIFIC INTERSTELLAR WHEELER-"
1470 PRINT "DEALER YOU ARE!"
1480 PRINT
1490 END
1500 REM --- PORTFOLIO ---
1510 TEST=0
1520 FOR A=1 TO 8
1530 IF S(A) = 0 THEN 1550
1540 TEST=1
1550 NEXT A
1560 IF TEST=1 THEN 1600
1570 PRINT "YOU DON'T OWN ANY STOCKS RIGHT NOW."
1580 PRINT
1590 GOTO 590
1600 PRINT "YOU OWN THE FOLLOWING STOCKS:"
1610 PRINT
1620 FOR B=1 TO 8
1630 IF S(B) = \emptyset THEN 1720
1640 IF S(B)>1 THEN 1690
1650 Z=B
1660 GOSUB 1260
1670 PRINT TAB(1); S(B); " SHARE "; SN$; " (# "; B; ")"
1680 GOTO 1720
1690 Z=B
1700 GOSUB 1260
1710 PRINT TAB(1); S(B); " SHARES "; SN$; " (# "; B; ") "
1720 NEXT B
1730 PRINT
1740 RETURN
1750 REM --- LINE SUBROUTINE --
1760 PRINT
1770 FOR L=1 TO 39
1780 PRINT "-";
1790 NEXT L
1800 PRINT
1810 PRINT
1820 RETURN
```

## $\square$ If You Have .

#### APPLE II

Delete: 230

Add: 145 PRINT

Change: 1260 IF DNE=1 THEN RESTORE

1300 DNE=1

#### **ATARI**

Delete: 230

```
Add: 95 DIM UD$(4),SN$(22),DC$(1),R$(1),H$(1),U$(5)
        102 FOR A=1 TO 8:S(A)=0:NEXT A
        105 FOR A=1 TO 8:P(A)=0:NEXT A
        645 IDX=(25-LEN(SN$)):GOSUB 1835:PRINT "$";P(Z)
        1835 FOR AA=1 TO IDX
        1845 PRINT " ";:NEXT AA:RETURN
Change: 130 P(G) = G*5*INT(RND(0)*6)+2
        300 IDX=10:GOSUB 1835:PRINT "IT IS NOW WEEK "; NW
        330 UDR=INT(RND(\emptyset) *2)
        390 AMT=INT(RND(\emptyset) *1\emptyset)
        610 PRINT "ID # ":"
                                 ": "----";
              PRICE'
        640 PRINT " "; Z;: IDX=5: GOSUB 1835: PRINT SN$;
        1670 PRINT " "; S(B); " SHARE "; SN$; " (# "; B; ")"
        1710 PRINT " ";S(B);" SHARES ";SN$;" (# ";B;")"
        1770 FOR L=1 TO 37
```

#### **COMMODORE 64**

Delete: 230

Change: 130 P(G) = G\*5\*INT(RND(0)\*6)+2

330 UDR=INT(RND(0) \*2) 390 AMT=INT(RND(0) \*10)

1260 IF DNE=1 THEN RESTORE

1300 DNE=1

970 GOTO 750 1000 GOTO 270

#### **COMMODORE VIC-20**

Delete: 230,240,260,660,730,740,840,900,950,1010,1100, 1190,1250,1360,1390,1610,1760,1800

Change: 130 P(G) = G\*5\*INT(RND(0)\*6)+2200 PRINT "WOULD YOU LIKE TO TRY" 280 IF NW=WK+1 THEN 1370 290 GOSUB 1770 300 PRINT TAB(2); "IT IS NOW WEEK "; NW 310 GOSUB 1770 330 UDR=INT(RND( $\emptyset$ )\*2) 390 AMT=INT(RND( $\emptyset$ ) \*10) 590 PRINT "\*THE AVAILABLE STOCKS ARE\*" 610 PRINT "ID #"; TAB(3); "---NAME---"; TAB(16); "PRICE" 640 PRINT Z; TAB(3); SN\$; TAB(16); "\$"; P(Z) 670 PRINT "WILL YOU SELL, BUY, OR QUIT (S,B,Q)" 690 IF DCS="O" THEN 1370 700 IF DC\$="S" THEN 750 770 PRINT "ENTER THE ID NUMBER" 870 GOTO 750 920 PRINT "HOW MANY SHARES DO YOU WANT TO SELL"

1040 PRINT "ENTER THE ID NUMBER"

ZZZ Z ZZZZ Z Z ZZZ ZZZZZ Z ZZZZ Z ZZZZ Z Z Z Z ZZZZZ Z Z ZZZ Z zzZ ZZZ ZZZ Z ZZZZZ ZZZZ Z ZZZ Z Z z Z z Z ZZZ BLASTOFF!

BLASTOFF! BLASTOFF! BLASTOFF! BLASTOFF! BLASTOFF! BLASTOFF! BLASTOFF! BLASTOFF! FOR ANOTHER COUNTDOWN, ENTER "C" TO PUT JUST ONE NUMBER ON THE SCREEN, ENTER THAT NUMBER (0-10)YOU'RE NOW USING CHARACTER: Z AS YOUR BUILDING BLOCK--TO SWITCH THE CHARACTER THAT MAKES UP THE NUMBERS, ENTER "S" TO QUIT, ENTER "Q" WHAT IS YOUR CHOICE? S WHICH LETTER OR NUMBER WOULD YOU LIKE TO USE? G -----FOR ANOTHER COUNTDOWN, ENTER "C" TO PUT JUST ONE NUMBER ON THE SCREEN, ENTER THAT NUMBER  $(0-1\emptyset)$ YOU'RE NOW USING CHARACTER: G AS YOUR BUILDING BLOCK--TO SWITCH THE CHARACTER THAT MAKES UP THE NUMBERS, ENTER "S" TO QUIT, ENTER "O" \_\_\_\_\_\_ WHAT IS YOUR CHOICE? 5 **GGGGG** G **GGGG GGGG** FOR ANOTHER COUNTDOWN, ENTER "C" TO PUT JUST ONE NUMBER ON THE SCREEN, ENTER THAT NUMBER  $(\emptyset-1\emptyset)$ YOU'RE NOW USING CHARACTER: G AS YOUR

YOU'RE NOW USING CHARACTER: G AS YOUR BUILDING BLOCK-TO SWITCH THE CHARACTER THAT MAKES UP THE NUMBERS, ENTER "S"

TO QUIT, ENTER "Q"

WHAT IS YOUR CHOICE? Q

## $\square$ Program Listing

```
10 REM SUPER COUNTDOWN
100 S$=CHR$(32)
110 GOSUB 1490
120 PRINT
130 PRINT TAB(8); "THE ROCKET IS ON THE PAD..."
140 PRINT
150 PRINT TAB(8); "HIT THE ENTER KEY FOR"
160 PRINT TAB(8); "THE FINAL COUNTDOWN";
170 INPUT D$
180 PRINT CHR$(12)
190 FOR A=10 TO 0 STEP -1
200 PRINT CHR$(7)
210 ON A+1 GOSUB 1260,630,700,770,840,910,980,1050,1120,
    1190,1340
220 FOR B=1 TO 350
230 NEXT B
240 PRINT CHR$(12)
250 NEXT A
260 FOR C=1 TO 25
270 PRINT
280 NEXT C
290 FOR D=1 TO 25
300 PRINT TAB(10); "BLASTOFF1"
310 NEXT D
320 FOR E=1 TO 25
330 PRINT
340 NEXT E
350 PRINT CHR$(12)
360 GOSUB 1410
370 PRINT "FOR ANOTHER COUNTDOWN, ENTER";
380 PRINT CHR$(34); "C"; CHR$(34)
390 PRINT
400 PRINT "TO PUT JUST ONE NUMBER ON THE SCREEN,"
410 PRINT "ENTER THAT NUMBER (0-10)"
420 PRINT
430 PRINT "YOU'RE NOW USING CHARACTER: "; P$; " AS YOUR"
440 PRINT "BUILDING BLOCK--TO SWITCH THE CHARACTER"
450 PRINT "THAT MAKES UP THE NUMBERS, ENTER "; CHR$(34);
    "S"; CHR $(34)
460 PRINT
470 PRINT "TO QUIT, ENTER "; CHR$(34); "Q"; CHR$(34)
480 GOSUB 1410
490 PRINT TAB(8); "WHAT IS YOUR CHOICE";
500 INPUT F$
510 IF F$="" THEN 400
520 IF F$="C" THEN 120
530 IF F$="Q" THEN END
540 IF F$<>"S" THEN 570
550 GOSUB 1490
560 GOTO 350
570 F=VAL(F$)
580 IF F=10 THEN 600
```

```
590 IF ASC(F$) < 48 OR ASC(F$) > 57 THEN 350
600 PRINT CHR$(12)
610 ON F+1 GOSUB 1260,630,700,770,840,910,980,1050,
    1120,1190,1340
620 GOTO 360
630 REM -- 1 --
640 PRINT SS: PS
650 PRINT SS:PS
660 PRINT S$; P$
670 PRINT S$;P$
680 PRINT SS:PS
690 RETURN
700 REM -- 2 --
710 PRINT S$; S$; P$; P$; P$; S$
720 PRINT S$;P$;S$;S$;S$;P$
730 PRINT S$;S$;S$;P$;S$
740 PRINT S$; S$; P$; S$; S$; S$
750 PRINT S$;P$;P$;P$;P$;P$
760 RETURN
770 REM -- 3
780 PRINT S$;S$;P$;P$;P$;S$
790 PRINT S$:P$:S$:S$:S$:P$
800 PRINT S$;S$;S$;P$;P$;S$
810 PRINT S$;P$;S$;S$;S$;P$
820 PRINT S$;S$;P$;P$;P$;S$
830 RETURN
840 REM -- 4
850 PRINT S$;P$;S$;S$;S$;P$
860 PRINT S$;P$;S$;S$;S$;P$
870 PRINT S$;P$;P$;P$;P$
880 PRINT S$:S$:S$:S$:P$
890 PRINT S$;S$;S$;S$;P$
900 RETURN
910 REM -- 5
920 PRINT S$;P$;P$;P$;P$
930 PRINT S$;P$;S$;S$;S$;S$
940 PRINT S$;P$;P$;P$;S$
950 PRINT S$;S$;S$;S$;P$
960 PRINT S$;P$;P$;P$;S$
970 RETURN
980 REM -- 6 --
990 PRINT S$; S$; P$; P$; P$; S$
1000 PRINT S$;P$;S$;S$;S$;S$
1010 PRINT S$;P$;P$;P$;P$;S$
1020 PRINT S$;P$;S$;S$;S$;P$
1030 PRINT S$;S$;P$;P$;P$;S$
1040 RETURN
1050 REM -- 7
1060 PRINT S$;P$;P$;P$;P$
1070 PRINT S$;S$;S$;S$;P$
1080 PRINT S$;S$;S$;S$;P$;S$
1090 PRINT S$; S$; S$; S$; P$; S$
1100 PRINT S$;S$;S$;S$;P$;S$
1110 RETURN
```

1120 REM -- 8

```
1130 PRINT S$;S$;P$;P$;P$;S$
1140 PRINT S$; P$; S$; S$; P$
1150 PRINT S$;S$;P$;P$;P$;S$
1160 PRINT S$;P$;S$;S$;S$;P$
1170 PRINT S$;S$;P$;P$;P$;S$
1180 RETURN
1190 REM -- 9
1200 PRINT S$;S$;P$;P$;P$;S$
1210 PRINT S$;P$;S$;S$;S$;P$
1220 PRINT S$;S$;P$;P$;P$;P$
1230 PRINT S$; S$; S$; S$; P$
1240 PRINT SS:SS:PS:PS:PS:SS
1250 RETURN
1260 REM -- 0
1270 PRINT S$;S$;P$;P$;P$;S$
1280 PRINT S$; P$; S$; S$; S$; P$
1290 PRINT S$;P$;S$;S$;S$;P$
1300 PRINT S$; P$; S$; S$; P$
1310 PRINT S$; S$; P$; P$; P$; S$
1320 RETURN
1330 REM -- 10 --
1340 PRINT P$; S$; S$; P$; P$; P$; S$
1350 PRINT P$;S$;P$;S$;S$;S$;P$
1360 PRINT P$; S$; P$; S$; S$; S$; P$
1370 PRINT P$; S$; P$; S$; S$; S$; P$
1380 PRINT P$; S$; S$; P$; P$; P$; S$
1390 RETURN
1400 REM --- LINE DRAWING SUBROUTINE
1410 PRINT
1420 FOR L=1 TO 38
1430 PRINT "=";
1440 NEXT L
1450 PRINT
1460 PRINT
1470 RETURN
1480 REM --- BUILDING BLOCK SUBROUTINE
1490 PRINT CHR$(12)
1500 GOSUB 1410
1510 PRINT TAB(2); "YOU CAN CHOOSE ANY LETTER (A-Z) OR"
1520 PRINT TAB(2); "NUMBER (0-9) AS A BUILDING BLOCK"
1530 PRINT TAB(2); "TO CREATE THE BIG NUMBERS USED IN"
1540 PRINT TAB(2); "THIS PROGRAM."
1550 GOSUB 1410
1560 PRINT TAB(8); "WHICH LETTER OR NUMBER"
1570 PRINT TAB(8); "WOULD YOU LIKE TO USE";
1580 INPUT P$
1590 IF P$="" THEN 1490
1600 P=MID(P,1,1)
1610 IF ASC(P$) < 48 OR ASC(P$) > 90 THEN 1490
1620 PRINT CHR$(12)
1630 RETURN
```

## $\square$ If You Have . . .

#### APPLE II

```
Change: 150 PRINT TAB(8); "HIT THE RETURN KEY FOR"
180 HOME
240 HOME
350 HOME
600 HOME
1490 HOME
1620 HOME
```

Add: 95 DIM S\$(1),D\$(1),P\$(1),F\$(1)

#### **ATARI**

```
1645 FOR AA=1 TO IDX
        1655 PRINT " ";: NEXT AA: RETURN
Change: 130 IDX=7:GOSUB 1645:PRINT "THE ROCKET IS ON THE
            PAD...
       150 GOSUB 1645: PRINT "HIT THE RETURN KEY FOR"
       160 GOSUB 1645: PRINT "THE FINAL COUNTDOWN";
       180 PRINT CHR$(125)
       200 PRINT CHR$ (253)
       240 PRINT CHR$(125)
        300 IDX=9:GOSUB 1645:PRINT "BLASTOFF!"
        350 PRINT CHR$(125)
        490 IDX=7:GOSUB 1645:PRINT "WHAT IS YOUR CHOICE";
       600 PRINT CHR$(125)
       1420 FOR L=1 TO 37
        1490 PRINT CHR$(125)
        1510 PRINT " "; "YOU CAN CHOOSE ANY LETTER (A-Z) OR"
        1520 PRINT " "; "NUMBER (1-9) AS A BUILDING BLOCK"
        1530 PRINT "; "TO CREATE THE BIG NUMBERS USED IN"
        1540 PRINT " "; "THIS PROGRAM."
        1560 IDX=7:GOSUB 1645:PRINT "WHICH LETTER OR NUMBER"
        1570 GOSUB 1645: PRINT "WOULD YOU LIKE TO USE";
        1600 P=P(1,1)
        1620 PRINT CHR$(125)
```

#### **COMMODORE 64**

```
Change: 150 PRINT TAB(8); "HIT THE RETURN KEY FOR"
180 PRINT CHR$(147)
240 PRINT CHR$(147)
350 PRINT CHR$(147)
600 PRINT CHR$(147)
1490 PRINT CHR$(147)
1620 PRINT CHR$(147)
```

#### **COMMODORE VIC-20**

Change: 150 PRINT TAB(8); "HIT THE RETURN KEY FOR"
160 PRINT TAB(8); "THE FINAL COUNTDOWN"
180 PRINT CHR\$(147)
240 PRINT CHR\$(147)
350 PRINT CHR\$(147)
490 PRINT TAB(8); "WHAT IS YOUR CHOICE"
600 PRINT CHR\$(147)
1420 FOR L=1 TO 21
1490 PRINT CHR\$(147)
1570 PRINT TAB(8); "WOULD YOU LIKE TO USE"
1620 PRINT CHR\$(147)

#### **TEXAS INSTRUMENTS 99/4A**

Add: 214 FOR SP=1 TO 6 215 PRINT 216 NEXT SP 535 END

Change: 180 CALL CLEAR
200 CALL SOUND(100,440,3)
240 CALL CLEAR
350 CALL CLEAR
530 IF F\$<>"Q" THEN 540
590 IF (ASC(F\$)<48)+(ASC(F\$)>57) THEN 350
600 CALL CLEAR
1420 FOR L=1 TO 28
1490 CALL CLEAR
1600 P\$=SEG\$(Y\$,1,1)
1610 IF (ASC(P\$)<48)+(ASC(P\$)>90) THEN 1490
1620 CALL CLEAR

#### TRS-80 COLOR COMPUTER

Delete: 390,420,460,1460

Add: 615 PRINT "HIT ENTER TO CONTINUE" 616 INPUT X\$

Change: 180 CLS 200 SOUND 173,3 240 CLS 350 CLS 600 CLS 1410 REM

1420 FOR L=1 TO 31

1490 CLS 1620 CLS

# Far, Far Away

When you're flying through space, you can't stop to get a map at a gas station — but you don't need to! When you want to find out how far a planet is from the sun, you simply consult your onboard computer. It's just like the one you have at home!

## ☐ Sample Run

- 1 MERCURY
- 2 VENUS
- 3 EARTH
- 4 MARS
- 5 JUPITER
- 6 SATURN
- 7 URANUS
- 8 NEPTUNE
- 9 PLUTO

WHICH PLANET IS 93 MILLION MILES FROM THE SUN? 3

RIGHT!

WHICH PLANET IS 142 MILLION MILES FROM THE SUN? 4

RIGHT!

WHICH PLANET IS 3675 MILLION MILES FROM THE SUN? 1

SORRY--THE ANSWER IS NOT MERCURY!

[There's more — when you get to the end of the program, you'll see something like this:]

WHICH PLANET IS 484 MILLION MILES FROM THE SUN? 5

RIGHT!

IT TOOK YOU 12 TRIES TO GET THEM ALL--WANT TO TRY AGAIN? NO

## ☐ Program Listing

```
10 REM FAR, FAR AWAY
100 \text{ DIM } M(9), A(9)
110 K=1
120 T=T+1
130 GOSUB 430
140 PRINT
150 PRINT "WHICH PLANET IS "; M(K); " MILLION MILES"
160 PRINT "FROM THE SUN";
170 INPUT Q
180 PRINT
190 IF Q<1 OR Q>9 THEN 150
200 IF Q<>A(K) THEN 260
210 PRINT "RIGHT!"
220 PRINT
230 K=K+1
240 IF K=10 THEN 340
250 GOTO 120
260 PRINT "SORRY--THE ANSWER IS NOT ";
270 FOR C=1 TO Q
280 READ P$
290 NEXT C
300 RESTORE
310 PRINT P$;"!"
320 PRINT
330 GOTO 120
340 PRINT
350 PRINT "IT TOOK YOU ";T;" TRIES TO GET THEM ALL--"
360 PRINT "WANT TO TRY AGAIN";
370 INPUT Y$
380 IF MID$(Y$,1,1) = "N" THEN END
390 T=0
400 PRINT
410 GOTO 110
420 REM --- READING SUBROUTINE ---
430 FOR A=1 TO 9
440 READ P$
450 PRINT TAB(5); A; TAB(10); P$
460 NEXT A
470 FOR B=1 TO 9
480 READ M(B), A(B)
490 NEXT B
500 RESTORE
510 RETURN
520 DATA MERCURY, VENUS, EARTH
530 DATA MARS, JUPITER, SATURN
540 DATA URANUS, NEPTUNE, PLUTO
550 DATA 93,3,142,4,3675,9
560 DATA 1784,7,67,2,887,6
570 DATA 36,1,2795,8,484,5
```

## ☐ If You Have . . .

#### **APPLE II** No Changes Required

#### **ATARI**

Add: 95 DIM P\$(7),Y\$(3)

485 M(B)=MB:A(B)=AB 585 FOR AA=1 TO IDX

595 PRINT " ";:NEXT AA:RETURN

Change: 380 IF Y\$(1,1)="N" THEN END

450 IDX=4:GOSUB 585:PRINT A;:GOSUB 585:PRINT P\$

480 READ MB, AB

### **COMMODORE 64** No Changes Required

#### **COMMODORE VIC-20**

Change: 160 PRINT "FROM THE SUN"

360 PRINT "WANT TO TRY AGAIN"

#### TEXAS INSTRUMENTS 99/4A

Add: 385 END

Change: 190 IF (Q<1)+(Q>9) THEN 150

380 IF SEG\$(Y\$,1,1)<>"N" THEN 390

430 FOR AA=1 TO 9

450 PRINT TAB(5); AA; TAB(10); P\$

460 NEXT AA

TRS-80 COLOR COMPUTER No Changes Required

# Shape Invaders

On the planet of polygons, you have to use your computer to stay alive. The computer usually works just fine — it's the human who sometimes has a screw loose. Even on the planet Earth, you may find this program handy when you need to figure the area of a polygon.

## ☐ Sample Run

MATHEMATICAL TRAVEL AGENCY

TOURS TO STRANGE PLACES OUR SPECIALTY

TODAY'S CHARTER FLIGHT TO THE PLANET
OF POLYGONS LEAVES IN...ONE MINUTE...
YOU'D BETTER HURRY--AND DON'T FORGET

YOUR TOOTHBRUSH! HIT THE ENTER KEY
TO HAVE YOUR PASSPORT STAMPED?

PLANET OF POLYGONS AIRPORT PASSPORT OFFICE

APPROVAL GRANTED FOR TEMPORARY VISIT

WARNING TO ALL FOREIGN VISITORS--BE ALERT FOR MYSTERIOUS SHAPE MONSTERS THAT HAVE BEEN EATING TRAVELERS LIKE CANDY. THE ONLY WAY TO AVOID BEING TURNED INTO LUNCH FOR ONE IS TO FIGURE OUT THE MONSTER'S AREA. BE CAREFUL!

HIT THE ENTER KEY TO CONTINUE?

ONE RULE--DON'T MIX MEASUREMENTS!
IF YOU ENTER FEET FOR ONE SIDE OF
A RECTANGLE, DON'T USE INCHES FOR
THE OTHER SIDE!

HERE COMES A MONSTER! IT'S A STRANGE SHAPE, BUT YOU CAN'T TELL EXACTLY WHICH FROM HERE. GUESS WHICH SHAPE THE MONSTER IS. YOUR COMPUTER WILL DO THE REST!

- 1 TRIANGLE
- 2 SQUARE
- 3 TRAPEZOID
- 4 RECTANGLE
- 5 REGULAR PENTAGON
- 6 REGULAR HEXAGON
- 7 REGULAR OCTAGON
- 8 CIRCLE

WHICH SHAPE IS THE MONSTER? 6

ENTER THE LENGTH OF ONE SIDE? 560 MILES

THE AREA OF A REGULAR HEXAGON WITH EACH SIDE MEASURING 560 MILES IS 814732.8 MILES

YOU DIDN'T GUESS THE SHAPE, BUT AT LEAST THE CALCULATION WAS RIGHT. TRY ONE MORE TIME BEFORE YOU BECOME LUNCH.

- 1 TRIANGLE
- 2 SQUARE
- 3 TRAPEZOID
- 4 RECTANGLE
- 5 REGULAR PENTAGON
- 6 REGULAR HEXAGON
- 7 REGULAR OCTAGON
- 8 CIRCLE

WHICH SHAPE IS THE MONSTER? 3

ENTER THE HEIGHT OF THE TRAPEZOID? 500 YARDS ENTER THE LENGTH OF THE LONGER PARALLEL SIDE? 900 YARDS ENTER THE LENGTH OF THE SHORTER OF THE SHORTER PARALLEL SIDE? 350 YARDS

THE AREA OF A TRAPEZOID OF HEIGHT 500 YARDS WITH THE LONGER PARALLEL SIDE MEASURING 900 YARDS AND THE SHORTER PARALLEL SIDE MEASURING 350 YARDS IS 312500 YARDS

TERRIFIC!! YOU GUESSED THE SHAPE OF THE MONSTER AND CALCULATED ITS AREA PROPERLY!!

DO YOU WANT TO TRY ANOTHER? NO

## $\square$ Program Listing

```
10 REM SHAPE INVADERS
100 PRINT CHR$(12)
110 GOSUB 1950
120 PRINT TAB(6); "MATHEMATICAL TRAVEL AGENCY"
130 GOSUB 1950
140 PRINT "TOURS TO STRANGE PLACES OUR SPACIALTY"
150 GOSUB 1950
160 PRINT "TODAY'S CHARTER FLIGHT TO THE PLANET"
170 PRINT "OF POLYGONS LEAVES IN...ONE MINUTE...
180 PRINT "YOU'D BETTER HURRY--AND DON'T FORGET"
190 PRINT "YOUR TOOTHBRUSH! HIT THE ENTER KEY"
200 PRINT "TO HAVE YOUR PASSPORT STAMPED";
210 INPUT W$
220 PRINT CHR$(12)
230 GOSUB 1950
240 PRINT
250 PRINT TAB(9); "PLANET OF POLYGONS"
260 PRINT TAB(7); "AIRPORT PASSPORT OFFICE"
270 PRINT
280 GOSUB 1950
290 PRINT " APPROVAL GRANTED FOR TEMPORARY VISIT"
300 GOSUB 1950
310 PRINT
320 PRINT
330 PRINT "WARNING TO ALL FOREIGN VISITORS--BE"
340 PRINT "ALERT FOR MYSTERIOUS SHAPE MONSTERS"
350 PRINT "THAT HAVE BEEN EATING TRAVELERS LIKE"
                  THE ONLY WAY TO AVOID BEING"
360 PRINT "CANDY.
370 PRINT "TURNED INTO LUNCH FOR ONE IS TO FIGURE"
380 PRINT "OUT THE MONSTER'S AREA. BE CAREFUL!"
390 PRINT
400 PRINT "HIT THE ENTER KEY TO CONTINUE";
410 INPUT W$
420 PRINT CHR$(12)
430 PRINT
440 PRINT "ONE RULE--DON'T MIX MEASUREMENTS!"
450 PRINT "IF YOU ENTER FEET FOR ONE SIDE OF"
460 PRINT "A RECTANGLE, DON'T USE INCHES FOR"
470 PRINT "THE OTHER SIDE!"
480 Z=INT(RND(1)*8)+1 --
490 PRINT
500 PRINT "HERE COMES A MONSTER! IT'S A STRANGE"
510 PRINT "SHAPE, BUT YOU CAN'T TELL EXACTLY"
520 PRINT "WHICH FROM HERE. GUESS WHICH SHAPE"
530 PRINT "THE MONSTER IS. YOUR COMPUTER WILL"
540 PRINT "DO THE REST!"
550 PRINT
560 PRINT TAB(2); "1"; TAB(7); "TRIANGLE"
570 PRINT TAB(2); "2"; TAB(7); "SQUARE"
580 PRINT TAB(2); "3"; TAB(7); "TRAPEZOID"
590 PRINT TAB(2); "4"; TAB(7); "RECTANGLE"
600 PRINT TAB(2); "5"; TAB(7); "REGULAR PENTAGON"
```

```
610 PRINT TAB(2); "6"; TAB(7); "REGULAR HEXAGON"
620 PRINT TAB(2); "7"; TAB(7); "REGULAR OCTAGON"
630 PRINT TAB(2); "8"; TAB(7); "CIRCLE"
640 PRINT
650 PRINT "WHICH SHAPE IS THE MONSTER";
660 INPUT T
670 PRINT
680 ON T GOTO 700,860,920,1160,1320,1380,1440,1500
690 GOTO 650
700 PRINT "ENTER THE LENGTH OF THE BASE";
    INPUT C$
710
720 GOSUB 1850
730 B=C
740 MS=DS
750 IF B<=0 THEN 700
760 PRINT "ENTER THE HEIGHT":
770 INPUT C$
780 GOSUB 1850
790 H=C
800 IF H<=0 THEN 760
810 PRINT
          "THE AREA OF A TRIANGLE WITH A BASE"
820 PRINT
830 PRINT "OF ";B;" ";M$;" AND A HEIGHT OF ";H;" ";M$
840 PRINT "IS "; B*H/2;" "; M$
850 GOTO 1590
860 GOSUB 1780
870 IF S<=0 THEN 860
880 PRINT
890 PRINT "THE AREA OF A SQUARE WITH EACH SIDE"
900 PRINT "MEASURING ";S;" ";M$;" IS ";S*S;" ";M$
910 GOTO 1 590
920 PRINT "ENTER THE HEIGHT OF THE TRAPEZOID":
930 INPUT CS
940 GOSUB 1850
950 H=C
960 M$=D$
970 IF H<=0 THEN 920
980 PRINT "ENTER THE LENGTH OF THE LONGER PARALLEL"
990 PRINT "SIDE":
1000 INPUT C$
1010 GOSUB 1850
1020 L=C
1030 IF L<=0 THEN 980
1040 PRINT "ENTER THE LENGTH OF THE SHORTER"
1050 PRINT "OF THE SHORTER PARALLEL SIDE";
1060 INPUT CS
1070 GOSUB 1850
1080 S=C
1090 IF S<=0 THEN 1040
1100 PRINT
1110 PRINT "THE AREA OF A TRAPEZOID OF HEIGHT"; H; "; M$
1120 PRINT "WITH THE LONGER PARALLEL SIDE MEASURING"
1130 PRINT L;" "; M$;" AND THE SHORTER PARALLEL SIDE"
1140 PRINT "MEASURING ";S;" ";M$;" IS ";(H*(L+S))/2;" ";
     MS
                                                  (continued)
```

```
1150 GOTO 1590
1160 PRINT "ENTER THE LENGTH OF THE BASE";
1170 INPUT C$
1180 GOSUB 1850
1190 B=C
1200 MS=DS
1210 IF B<=0 THEN 1160
1220 PRINT "ENTER THE HEIGHT":
1230 INPUT C$
1240 GOSUB 1850
1250 H=C
1260 IF H<=0 THEN 1220
1270 PRINT
1280 PRINT "THE AREA OF A RECTANGLE WITH A BASE"
1290 PRINT "MEASURING ";B;" ";M$; " AND A HEIGHT"
1300 PRINT "OF ";H;" ";M$;" ";"IS ";B*H;" ":M$
1310 GOTO 1590
1320 GOSUB 1780
1330 IF S<=0 THEN 1320
1340 PRINT
1350 PRINT "THE AREA OF A REGULAR PENTAGON WITH"
1360 PRINT "EACH SIDE MEASURING ";S;" ";M$;" IS ";
     1.72*S*S;" ";M$
1370 GOTO 1590
1380 GOSUB 1780
1390 IF S<=0 THEN 1380
1400 PRINT
1410 PRINT "THE AREA OF A REGULAR HEXAGON WITH"
1420 PRINT "EACH SIDE MEASURING ";S;" ";M$;" IS ";
     2.598*S*S;" ";M$
1430 GOTO 1590
1440 GOSUB 1780
1450 IF S<=0 THEN 1440
1460 PRINT
1470 PRINT "THE AREA OF A REGULAR OCTAGON WITH"
1480 PRINT "EACH SIDE MEASURING ";S;" ";M$;" IS ";
     4.828*S*S;" "; M$
1490 GOTO 1590
1500 PRINT "ENTER THE LENGTH OF THE RADIUS";
1510 INPUT C$
1520 GOSUB 1850
1530 R=C
1540 M$=D$
1550 IF R<=0 THEN 1500
1560 PRINT
1570 PRINT "THE AREA OF A CIRCLE WITH A RADIUS OF"
1580 PRINT R;" "; M$; " IS "; 3.1416*R*R; " "; M$
1590 PRINT
1600 IF T=2 THEN 1670
1610 GOSUB 1950
1620 PRINT "YOU DIDN'T GUESS THE SHAPE, BUT AT"
1630 PRINT "LEAST THE CALCULATION WAS RIGHT. TRY"
1640 PRINT "ONE MORE TIME BEFORE YOU BECOME LUNCH."
1650 GOSUB 1950
1660 GOTO 550
```

```
1670 GOSUB 1950
1680 PRINT "TERRIFIC!! YOU GUESSED THE SHAPE OF"
1690 PRINT "THE MONSTER AND CALCULATED ITS AREA"
1700 PRINT "PROPERLY!!"
1710 GOSUB 1950
1720 PRINT
1730 PRINT "DO YOU WANT TO TRY ANOTHER";
1740 INPUT Y$
1750 IF MID$(Y$,1,1) <>"N" THEN 480
1760 END
1770 REM --- SIDE ENTERING SUBROUTINE
1780 PRINT "ENTER THE LENGTH OF ONE SIDE";
1790 INPUT C$
1800 GOSUB 1850
1810 S=C
1820 M$=D$
1830 RETURN
1840 REM --- TESTING SUBROUTINE ---
1850 D$=""
1860 FOR Z=1 TO LEN(C$)
1870 IF ASC(MID$(C$,Z,1))>57 THEN 1910
1880 NEXT Z
1890 C=VAL(C$)
1900 GOTO 1930
1910 C=VAL(MID$(C$,1,Z))
1920 D=MID(C$, Z, LEN(C$) -Z+1)
1930 RETURN
1940 REM --- LINE SUBROUTINE
1950 FOR V=1 TO 38
1960 PRINT "=";
1970 NEXT V
1980 PRINT
1990 RETURN
```

## $\square$ If You Have . .

#### APPLEII

Change: 100 HOME
190 PRINT "YOUR TOOTHBRUSH! HIT THE RETURN KEY"
220 HOME
400 PRINT "HIT THE RETURN KEY TO CONTINUE";
420 HOME

#### **ATARI**

Add: 95 DIM W\$(1),M\$(7),D\$(7),C\$(2),Y\$(3) 1995 FOR AA=1 TO IDX 1997 PRINT " ";:NEXT AA:RETURN

Change: 100 PRINT CHR\$(125) (continued)

```
190 PRINT "YOUR TOOTHBRUSH! HIT THE RETURN KEY"
220 PRINT CHR$(125)
250 IDX=8:GOSUB 1995:PRINT "PLANET OF POLYGONS"
260 IDX=6:GOSUB 1995:PRINT "AIRPORT PASSPORT
    OFFICE"
400 PRINT "HIT THE RETURN KEY TO CONTINUE";
420 PRINT CHR$ (125)
480 Z=INT(RND(\emptyset) *8)+1
560 PRINT " ";"1";:IDX=4:GOSUB 1995:PRINT
     "TRIANGLE"
570 PRINT " ";"2";:GOSUB 1995:PRINT "SQUARE"
580 PRINT " "; "3"; :GOSUB 1995:PRINT "TRAPEZOID"
590 PRINT " "; "4"; :GOSUB 1995: PRINT "RECTANGLE"
600 PRINT " "; "5"; :GOSUB 1995: PRINT "REGULAR
    PENTAGON"
610 PRINT " "; "6"; :GOSUB 1995:PRINT "REGULAR
HEXAGON"
620 PRINT " ";"7";:GOSUB 1995:PRINT "REGULAR
     OCTAGON"
630 PRINT " "; "8"; : GOSUB 1995: PRINT "CIRCLE"
1750 IF Y$(1,1) <>"N" THEN 480
1870 IF ASC(C$(Z,Z))>57 THEN 1910
1910 C=VAL(C$(1,Z))
1920 D$=C$(Z,LEN(C$))
```

#### **COMMODORE 64**

Add:

```
Change: 100 PRINT CHR$(147)
190 PRINT "YOUR TOOTHBRUSH! HIT THE RETURN KEY"
220 PRINT CHR$(147)
400 PRINT "HIT THE RETURN KEY TO CONTINUE";
420 PRINT CHR$(147)
```

552 PRINT "HIT ENTER KEY TO CONTINUE"

#### COMMODORE VIC-20 Program Will Not Run on Unexpanded VIC-20

#### **TEXAS INSTRUMENTS 99/4A**

1950 FOR V=1 TO 37

```
554 INPUT DU$
556 CALL CLEAR

Change: 100 CALL CLEAR
220 CALL CLEAR
420 CALL CLEAR
480 Z=INT(RND*8)+1
1750 IF SEG$(Y$,1,1)<>"N" THEN 480
1870 IF ASC(SEG$(C$,Z,1))>57 THEN 1910
1910 C=VAL(SEG$(C$,1,Z))
1920 D$=SEG$(C$,Z,LEN(C$)-Z+1)
1950 FOR V=1 TO 28
```

#### TRS-80 COLOR COMPUTER

Delete: 390,490

Add: 325 INPUT X\$

497 PRINT "HIT THE ENTER KEY TO CONTINUE" 498 INPUT X\$

498 INPUT X\$
499 PRINT

Change: 100 CLS

220 CLS

320 PRINT "HIT THE ENTER KEY TO CONTINUE"

420 CLS

460 PRINT "A RECTANGLE, DON'T USE INCHES"

470 PRINT "FOR THE OTHER SIDE"

480 Z=RND(8)

1130 PRINT L;" "; M\$; " AND THE SHORTER PARALLEL"

1140 PRINT "SIDE MEASURING "; S; " "; M\$; " IS ";

(H\*(L+S))/2;" ";M\$

1950 FOR V=1 TO 31

# Star Bright

If you're going to use a star to help guide your spacecraft home, you'd better pick a bright one. You'll find all the brightest stars right here.

## ☐ Sample Run

THERE ARE APPROXIMATELY 100 BILLION STARS IN OUR OWN GALAXY, THE MILKY WAY. CAN YOU RANK THE 15 THAT ARE BRIGHTEST TO OBSERVERS ON EARTH?

HOW MANY STARS DO YOU THINK YOU CAN SEE AT NIGHT? 100000

#### COUNTING THE STARS...

BILLION ... 6.666667 13.33333 BILLION 20 BILLION 26.66667 BILLION 33.33334 BILLION BILLION BILLION 46.66667 53.33334 BILLION BILLION ... 66.66667 BILLION 73.33334 BILLION BILLION ... 86.66667 BILLION BILLION . 93.33334 100 BILLION ...

#### NOW

#### RANKING THE STARS

STAR	#		RANK	
· .			£	
POLLUX	1			
CAPELLA	2			
SIRIUS	₁3 <sup>□</sup>			is.
RIGEL	4			
ALDEBARAN	5			
ACHERNAR	6			
BETA CENTAURI	7			22
PROCYON	8			
ARCTURUS	9		94	
VEGA	10		c	
CANOPUS	11			
ALTAIR	12			
ALPHA CENTAURI	13			÷÷
SPICA	14			
BETELGEUSE	15	60		

TRY TO ARRANGE THESE 15 STARS IN THE ORDER OF THEIR BRIGHTNESS.

WHICH STAR RANKS # 1 IN BRIGHTNESS? 3 RIGHT! SIRIUS IS RANKED # 1 HIT THE ENTER KEY TO CONTINUE?

TRY TO ARRANGE THESE 15 STARS IN THE ORDER OF THEIR BRIGHTNESS.

WHICH STAR RANKS # 2 IN BRIGHTNESS? 11 RIGHT! CANOPUS IS RANKED # 2 HIT THE ENTER KEY TO CONTINUE

TRY TO ARRANGE THESE 15 STARS IN THE ORDER OF THEIR BRIGHTNESS.

WHICH STAR RANKS # 3 IN BRIGHTNESS? 7 NO, BETA CENTAURI IS NOT RANKED # 3 HIT THE ENTER KEY TO CONTINUE?

[There's more — when you get to the end of the program, you'll see something like this:]

TRY TO ARRANGE THESE 15 STARS IN THE ORDER OF THEIR BRIGHTNESS.

WHICH STAR RANKS # 15 IN BRIGHTNESS? 1 RIGHT! POLLUX IS RANKED # 15 HIT THE ENTER KEY TO CONTINUE?

CONGRATULATIONS!! YOU GOT THEM ALL!!
AND IT TOOK YOU ONLY 24 TRIES!!

WANT TO TRY AGAIN? NO

## ☐ Program Listing

- 10 REM STAR BRIGHT
- 100 PRINT
- 110 PRINT "THERE ARE APPROXIMATELY 100 BILLION"
- 120 PRINT "STARS IN OUR OWN GALAXY, THE MILKY WAY."
- 130 PRINT "CAN YOU RANK THE 15 THAT ARE BRIGHTEST"
- 140 PRINT "TO OBSERVERS ON EARTH?"
- 150 PRINT
- 160 PRINT "HOW MANY STARS DO YOU THINK YOU CAN SEE"
- 170 PRINT "AT NIGHT";
- 180 INPUT D
- 190 IF D<1 THEN 160
- 200 IF D>30000 THEN D=500
- 210 RANDOMIZE D
- 220 Q\$="SRATSEHTGNIKNAR"
- 230 PRINT
- 240 DIM RD(15),N(15,14)

(continued)

```
250 F≖0
260 R=0
270 PRINT TAB(13); "COUNTING THE STARS...
280 PRINT
290 FOR D=1 TO 15
300 \text{ RD}(D) = INT((RND(1)*15)+1)
310 FOR E=1 TO D-1
320 IF RD(D)=RD(E) THEN 300
330 NEXT E
340 PRINT TAB(12); E*6.666667; "BILLION ..."
350 NEXT D
360 PRINT CHR$(12)
370 GOSUB 900
380 R=R+1
390 IF R=16 THEN 800
400 PRINT CHR$(12)
                                       -- RANK ---
410 PRINT TAB(6); "---- STAR ----
420 F=F+1
430 PRINT
440 GOSUB 1080
450 PRINT
460 PRINT "TRY TO ARRANGE THESE 15 STARS IN"
470 PRINT "THE ORDER OF THEIR BRIGHTNESS."
480 PRINT
490 PRINT "WHICH STAR RANKS #":R:" IN BRIGHTNESS";
500 INPUT G
510 IF G>0 AND G<16 THEN 570
520 PRINT
530 PRINT "YOUR CHOICES ARE ONLY 1 THROUGH 15!"
540 PRINT "HIT THE ENTER KEY TO CONTINUE";
550 INPUT X$
560 GOTO 400
570 IF RD(G)=R THEN 690
580 PRINT "NO, ";
590 FOR M=1 TO 14
600 IF N(RD(G), M) <> 32 THEN 620
610 PRINT CHR$(32):
620 IF N(RD(G),M) <65 THEN 640
630 PRINT CHR$(N(RD(G),M));
640 NEXT M
650 PRINT " IS NOT RANKED #";R
660 PRINT "HIT THE ENTER KEY TO CONTINUE";
670 INPUT X$
680 GOTO 400
69Ø PRINT "RIGHT! ";
700 FOR W=1 TO 14
710 IF N(RD(G),W)<>32 THEN 730
720 PRINT CHR$(32);
730 IF N(RD(G), W) < 65 THEN 750
740 PRINT CHR(N(RD(G),W));
750 NEXT W
760 PRINT " IS RANKED #";R
770 PRINT "HIT THE ENTER KEY TO CONTINUE";
780 INPUT D$
790 GOTO 380
```

```
800 PRINT
810 PRINT "CONGRATULATIONS!! YOU GOT THEM ALL!
820 PRINT "AND IT TOOK YOU ONLY ";F;" TRIES!!"
830 PRINT
840 PRINT "WANT TO TRY AGAIN";
850 INPUT YS
860 IF MID$(Y$,1,1) = "N" THEN END
870 RESTORE
880 GOTO 250
890 REM --- DECIPHER SUBROUTINE --
900 FOR U=1 TO 8
910 PRINT
920 NEXT U
930 PRINT TAB(18); "NOW"
940 PRINT
950 PRINT TAB(12);
960 FOR A=1 TO 15
970 PRINT MID$(Q$,16-A,1);
980 IF A=7 OR A=10 THEN PRINT CHR$(32);
990 FOR B=1 TO 14
1000 READ C
1010 IF C=32 THEN 1030
1020 IF C<65 THEN 1040
1030 N(A,B) = C
1040 NEXT B
1050 NEXT A
1060 RETURN
1070R EM --- DISPLAY SUBROUTINE
1080 FOR J=1 TO 15
1090 PRINT TAB(6);
1100 FOR K=1 TO 14
1110 PRINT CHR$(N(RD(J),K));
1120 NEXT K
1130 PRINT CHR$(32);
1140 IF RD(J) = > R THEN 1170
1150 PRINT J; TAB(27); RD(J)
1160 GOTO 1180
1170 PRINT J
1180 NEXT J
1190 RETURN
1200 DATA 52,43,56,61,47,42,57,59,83,73,82,73,85,83
1210 DATA 48,41,52,47,61,45,53,67,65,78,79,80,85,83
1220 DATA 65,76,80,72,65,32,67,69,78,84,65,85,82,73
1230 DATA 63,57,43,54,52,61,45,53,63,51,86,69,71,65
1240 DATA 57,61,43,47,53,52,63,67,65,80,69,76,76,65
1250 DATA 63,59,62,45,49,52,65,82,67,84,85,82,85,83
1260 DATA 57,61,43,47,52,55,61,53,60,82,73,71,69,76
1270 DATA 60,63,57,51,43,45,44,80,82,79,67,89,79,78
1280 DATA 59,45,61,55,41,44,65,67,72,69,82,78,65,82
1290 DATA 60,66,69,84,65,32,67,69,78,84,65,85,82,73
1300 DATA 57,53,55,61,45,47,45,53,65,76,84,65,73,82
1310 DATA 60,57,52,48,66,69,84,69,76,71,69,85,83,69
1320 DATA 58,53,55,47,61,65,76,68,69,66,65,82,65,78
1330 DATA 60,62,45,56,60,45,43,61,53,83,80,73,67,65
1340 DATA 54,46,48,46,54,61,55,44,80,79,76,76,85,88
```

## $\square$ If You Have $\ldots$

#### APPLEII

Delete: 210

Add; 305 IF D=1 THEN E=1:GOTO 340

Change: 360 HOME

400 HOME

540 PRINT "HIT THE RETURN KEY TO CONTINUE"; 660 PRINT "HIT THE RETURN KEY TO CONTINUE"; 770 PRINT "HIT THE RETURN KEY TO CONTINUE";

1150 PRINT TAB(21); J; TAB(27); RD(J)

1170 PRINT TAB(21);J

#### **ATARI**

Delete: 210

Add: 95 DIM X\$(1),Q\$(17),Y\$(3),D\$(1)

245 FOR A=1 TO 15:  $RD(A) = \emptyset$ 

246 FOR B=1 TO 14:N(A,B)=0:NEXT B:NEXT A

315 IF D=1 THEN 330

335 AE=-1E-06:AX=INT(D/3):AE=AE\*AX

1145 DX=6:IF J>9 THEN DX=5

1355 FOR AA=1 TO IDX

1365 PRINT " ";: NEXT AA: RETURN

Change: 270 IDX=12:GOSUB 1355:PRINT COUNTING THE STARS..."

300 RD(D)=INT((RND(0) \*15)+1)

340 IDX=11:GOSUB 1355:PRINT E\*6.666667+AE;"

BILLION...

360 PRINT CHR\$(125)

400 PRINT CHR\$(125)

410 IDX=5:GOSUB 1355:PRINT "--- STAR ---- #

RANK ----

540 PRINT "HIT THE RETURN KEY TO CONTINUE";

660 PRINT "HIT THE RETURN KEY TO CONTINUE";

770 PRINT "HIT THE RETURN KEY TO CONTINUE";

860 IF Y\$(1,1) = "N" THEN END

930 IDX=17:GOSUB 1355:PRINT "NOW"

950 IDX=11:GOSUB 1355

970 PRINT Q\$(16-A,16-A);

1090 IDX=5:GOSUB 1355

1140 IF RD(J) >= R THEN 1170

1150 PRINT J;:IDX=DX:GOSUB 1355:PRINT RD(J)

#### COMMODORE 64

Delete: 210

Add: 305 IF D=1 THEN 350

Change: 300 RD(D) = INT((RND(0)\*15)+1)

360 PRINT CHR\$(147)

```
400 PRINT CHR$(147)
540 PRINT "HIT THE RETURN KEY TO CONTINUE";
660 PRINT "HIT THE RETURN KEY TO CONTINUE";
770 PRINT "HIT THE RETURN KEY TO CONTINUE";
1150 PRINT TAB(21); J; TAB(27); RD(J)
1170 PRINT TAB(21); J
```

#### **COMMODORE VIC-20** Program Will Not Run On Unexpanded VIC-20

#### **TEXAS INSTRUMENTS 99/4A**

```
Delete: 1090
  Add: 205 D=500
        865 END
        985 PRINT CHR$(32);
Change: 200 IF D<=30000 THEN 220
        210 RANDOMIZE
        270 PRINT TAB(3); "COUNTING THE STARS..."
        300 \text{ RD}(D) = INT((RND*15)+1)
        340 PRINT TAB(5); E*6.666667; BILLION...
        360 CALL CLEAR
        400 CALL CLEAR
        460 PRINT "ARRANGE THESE 15 STARS IN"
        510 IF (G>0)*(G<16) THEN 570
        860 IF SEG$(Y$,1,1) <> "N" THEN 870
        930 PRINT TAB(12); "NOW"
        950 PRINT TAB(6);
        970 PRINT SEG$(Q$,16-A,1);
        980 IF(A<>7) *(A<>10) THEN 990
        1140 IF RD(J) >= R THEN 1170
        1150 PRINT J; TAB(24); RD(J)
```

#### TRS-80 COLOR COMPUTER

```
Delete: 210,450,460,470,480
   Add: 305 IF D=1 THEN 350
        373 CLS
        374 PRINT "TRY TO ARRANGE THESE 15 STARS IN"
        375 PRINT "ORDER OF THEIR BRIGHTNESS."
        376 PRINT
        377 PRINT "HIT THE ENTER KEY TO CONTINUE"
        378 INPUT X$
Change: 300 \text{ RD}(D) = \text{RND}(15)
        340 PRINT TAB(8); E*6.666667; BILLION..."
        360 CLS
        400 CLS
        410 PRINT TAB(6); "---- STAR ---- # - RANK -"
        490 PRINT "WHICH IS #";R;" IN BRIGHTNESS";
        1150 PRINT TAB(21); J; TAB(27); RD(J)
        1170 PRINT TAB(21);J
```

# Escape from the Asteroid of Doom

You've crash-landed on a distant asteroid. You're searching for a secret treasure. Can you find the treasure and leave the cave alive?

Others have tried — and failed.

This adventure may be your most difficult challenge ever. Making a map may help you stay alive. But be warned: The caves on this asteroid are not all on the same level. Because of twisting corridors, you might have to go south from Room 7 to Room 8 — and the same direction from Room 8 to Room 7!

## ☐ Sample Run

YOU CAN PLAY THIS GAME IN 3 VERSIONS:

1--HARD

2--MEDIUM

3--EASY

WHICH ONE DO YOU WANT ? 3

YOU ARE A FAMOUS CAPTAIN OF THE FASTEST STARSHIP IN THE GALAXY. YOU HAVE JUST CRASH-LANDED ON A DISTANT ASTEROID, AND MOST OF YOUR CREW HAS NOT SURVIVED THE IMPACT. AS ONE OF THEM TAKES HIS FINAL BREATH, HE REVEALS THAT HE PROGRAMMED THE SHIP'S COMPUTER TO CRASH-LAND THE SHIP ON THIS BARREN SPACE ROCK SO THAT HE COULD RETRIEVE A VAST TREASURE HIDDEN DEEP INSIDE A SERIES OF TUNNELS AND CHAMBERS BENEATH THE SURFACE. BEFORE HE CAN SAY WHERE IT IS, HE TELLS YOU THAT ONCE YOU ENTER THE MAZE YOU HAVE TO FIND A SPECIAL PLATINUM KEY--OR YOU'LL NEVER BE ABLE TO ESCAPE ALIVE.

(HIT THE ENTER KEY FOR MORE)?

THEN HE PRESSES A BUTTON AT HIS FEET AND TRANSPORTS YOU TO THE FIRST ROOM INSIDE THE LABYRINTH OF TUNNELS. YOU HAVE TO FIND THE KEY TO ESCAPE--AND ALSO TRY TO LOCATE THE TREASURE BEFORE ANY OF THE TERRIBLE CREATURES AND HAZARDS INSIDE THE TUNNELS ENDS YOUR SEARCH.

ONE MORE THING--SOMEWBERE IN
THE MAZE OF TUNNELS IS A MAGIC
WORD WRITTEN BY THE ONLY OTHER
EARTHMAN TO VISIT THESE CAVES.
(HE DIED HORRIBLY BECAUSE HE
FORGOT IT.) THE WORD CHANGES
FROM TIME TO TIME, SO BE VERY
CAREFUL! A PROWLING MONSTER
WILL SMOTHER YOU IN SLIME UNLESS
YOU REMEMBER IT!! FINALLY, AT
THE END, YOU MUST GET TO ROOM 1
TO LEAVE THE CAVE. GOOD LUCK!
HOW MANY CREW MEMBERS WERE LOST? 4

\$

-- NOTHING BUT YOUR WITS --

----> YOU ARE IN ROOM 1 <----WHICH WAY DO YOU WANT TO GO? W

----> YOU ARE IN ROOM 17 <----WHICH WAY DO YOU WANT TO GO? N
SORRY, THERE IS NO DOOR HERE.
YOU CAN'T GO THIS WAY.

----> YOU ARE IN ROOM 17 <----WHICH WAY DO YOU WANT TO GO? W

#### 

YOU SEE A WORD WRITTEN IN BLOOD ON THE WALL: IPYBESU REMEMBER THIS WORD--IT MAY JUST SAVE YOUR LIFE!!!

----> YOU ARE IN ROOM 12 <----WHICH WAY DO YOU WANT TO GO? E

(continued)

WHICH WAY DO YOU WANT TO GO? N
HIDEOUS GIGANTIC SPIDERS BAR THE WAY.
YOU CAN'T GO THIS WAY.

----> YOU ARE IN ROOM 13 <----WHICH WAY DO YOU WANT TO GO? E

WHICH WAY DO YOU WANT TO GO? E
THIS DOOR HAS SUDDENLY CLOSED.
YOU CAN'T GO THIS WAY.

----> YOU ARE IN ROOM 14 <----WHICH WAY DO YOU WANT TO GO? N

THE TERRIBLE SLIME MONSTER IS IN THE ROOM WITH YOU. YOU HAVE ONE CHANCE TO USE THE MAGIC WORD TO SAVE YOUR LIFE--OTHERWISE YOU'LL BE DEAD IN TWO SECONDS! QUICK! WHAT IS IT? IPYBESU THIS TIME YOU REMEMBERED. LUCKY YOU!

#### 

----> YOU ARE IN ROOM 8 <----WHICH WAY DO YOU WANT TO GO? N

----> YOU ARE IN ROOM 4 <----WHICH WAY DO YOU WANT TO GO? E

----> YOU ARE IN ROOM 5 <---\*\* YOU'VE FOUND ALL THE TREASURE!!! \*\*
(NOW TRY TO LOCATE THE KEY TO ESCAPE)

WHICH WAY DO YOU WANT TO GO? N T HIS DOOR HAS SUDDENLY CLOSED. YOU CAN'T GO THIS WAY.

----> YOU ARE IN ROOM 5 <----WHICH WAY DO YOU WANT TO GO? W

----> YOU ARE IN ROOM 4 <----WHICH WAY DO YOU WANT TO GO? W

\*\* YOU'VE FOUND THE KEY TO ESCAPE!! \*\*

\*\*\*\*\*\*\*\*\*\*\* YOU DID IT!!! \*\*\*\*\*\*\*\*

YOU FOUND THE TREASURE—AND ALSO FOUND

THE PLATINUM KEY TO LET YOU ESCAPE !!!

NOW JUST RETURN TO ROOM 1 TO GET BACK!

WHICH WAY DO YOU WANT TO GO? S

----> YOU ARE IN ROOM 7 <----WHICH WAY DO YOU WANT TO GO? S

SORRY, THERE IS NO DOOR HERE.
YOU CAN'T GO THIS WAY.

----> YOU ARE IN ROOM 7 <----WHICH WAY DO YOU WANT TO GO? W

THIS DOOR HAS SUDDENLY CLOSED.
YOU CAN'T GO THIS WAY.

WHICH WAY DO YOU WANT TO GO? E
HISSING SCUM CREATURES BLOCK THE WAY,
BUT YOU FREEZE AND THEY IGNORE YOU.

#### 

YOU SEE A WORD WRITTEN IN BLOOD ON THE WALL: QZLFNUW REMEMBER THIS WORD--IT MAY JUST SAVE YOUR LIFE!!!

#### 

----> YOU ARE IN ROOM 7 <----WHICH WAY DO YOU WANT TO GO? N

----> YOU ARE IN ROOM 3 <----WHICH WAY DO YOU WANT TO GO? N
YOU ALMOST WALK INTO A CLOUD OF POISON.
YOU CAN'T GO THIS WAY.

WHICH WAY DO YOU WANT TO GO? E

----> YOU ARE IN ROOM 4 <----WHICH WAY DO YOU WANT TO GO? N

WHICH WAY DO YOU WANT TO GO? N
AHEAD ARE SHEETS OF UNPASSABLE FLAME.
YOU CAN'T GO THIS WAY.

WHICH WAY DO YOU WANT TO GO? W

YOU DID IT!!!! YOU FOUND THE TREASURE AND THE KEY TO ESCAPE--AND YOU MADE IT BACK TO ROOM 1 TO GET OUT OF THE MAZE!

## □ Program Listing

```
10 REM ESCAPE FROM THE ASTEROID OF DOOM
100 PRINT "YOU CAN PLAY THIS GAME IN 3 VERSIONS:"
110 PRINT
120 PRINT TAB(5); "1--HARD"
130 "PRINT TAB(5); "2--MEDIUM"
140 PRINT TAB(5); "3--EASY"
150 PRINT
160 PRINT "WHICH ONE DO YOU WANT ";
170 INPUT DI
180 IF DI<1 THEN 160
190 IF DI>3 THEN 160
200 FOR A=1 TO 25
210 PRINT
220 NEXT A
230 PRINT "YOU ARE A FAMOUS CAPTAIN OF THE"
240 PRINT "FASTEST STARSHIP IN THE GALAXY.
250 PRINT "YOU HAVE JUST CRASH-LANDED ON"
260 PRINT "A DISTANT ASTEROID, AND MOST OF"
270 PRINT "YOUR CREW HAS NOT SURVIVED THE"
280 PRINT "IMPACT. AS ONE OF THEM TAKES"
290 PRINT "HIS FINAL BREATH, HE REVEALS"
300 PRINT "THAT HE PROGRAMMED THE SHIP'S"
310 PRINT "COMPUTER TO CRASH-LAND THE SHIP"
320 PRINT "ON THIS BARREN SPACE ROCK SO"
330 PRINT "THAT HE COULD RETRIEVE A VAST"
340 PRINT "TREASURE HIDDEN DEEP INSIDE A"
350 PRINT "SERIES OF TUNNELS AND CHAMBERS"
360 PRINT "BENEATH THE SURFACE.
                                  BEFORE HE"
370 PRINT "CAN SAY WHERE IT IS, HE TELLS"
380 PRINT "YOU THAT ONCE YOU ENTER THE"
390 PRINT "MAZE YOU HAVE TO FIND A SPECIAL"
400 PRINT "PLATINUM KEY--OR YOU'LL NEVER"
410 PRINT "BE ABLE TO ESCAPE ALIVE."
420 FOR A=1 TO 3
430 PRINT
440 NEXT A
-450 PRINT "(HIT THE ENTER KEY FOR MORE)";
460 INPUT X$
470 FOR A=1 TO 16
480 PRINT
490 NEXT A
500 PRINT "THEN HE PRESSES A BUTTON AT HIS"
510 PRINT "FEET AND TRANSPORTS YOU TO THE"
520 PRINT "FIRST ROOM INSIDE THE LABYRINTH"
530 PRINT "OF TUNNELS. YOU HAVE TO FIND"
540 PRINT "THE KEY TO ESCAPE--AND ALSO TRY"
550 PRINT "TO LOCATE THE TREASURE BEFORE"
560 PRINT "ANY OF THE TERRIBLE CREATURES"
 570 PRINT "AND HAZARDS INSIDE THE TUNNELS
580 PRINT "ENDS YOUR SEARCH."
590 PRINT
600 PRINT "ONE MORE THING--SOMEWHERE IN"
```

```
610 PRINT "THE MAZE OF TUNNELS IS A MAGIC"
620 PRINT "WORD WRITTEN BY THE ONLY OTHER"
630 PRINT "EARTHMAN TO VISIT THESE CAVES."
640 PRINT "(HE DIED HORRIBLY BECAUSE HE"
650 PRINT "FORGOT IT.) THE WORD CHANGES"
660 PRINT "FROM TIME TO TIME, SO BE VERY"
670 PRINT "CAREFUL! A PROWLING MONSTER"
680 PRINT "WILL SMOTHER YOU IN SLIME UNLESS"
690 PRINT "YOU REMEMBER IT!! FINALLY, AT"
700 PRINT "THE END, YOU MUST GET TO ROOM 1"
710 PRINT "TO LEAVE THE CAVE. GOOD LUCK!"
720 Q$=CHR$(34)
730 PRINT "HOW MANY CREW MEMBERS WERE LOST";
740 INPUT LOST
750 IF LOST<1 THEN 730
760 RANDOMIZE (LOST)
770 FOR A=1 TO 17
780 PRINT
790 NEXT A
800 GOSUB 3380
810 PRINT
820 FIND=0
-830 TR=INT(RND(1) *16)+3
840 MG=INT(RND(1) *16) +3
850 IF MG=TR THEN 840
860 REM -----
870 ROOM=1
880 GOSUB 4010
890 GOSUB 2980
900 IF I$="E" THEN 970
910 IF I$="W" THEN 2320
920 IF I $= "N" THEN GOSUB 2500
930 IF I$="S" THEN GOSUB 2500
940 GOSUB 3260
950 GOTO 890
960 REM -----
970 ROOM=2
980 GOSUB 2980
990 IF I$="E" THEN GOSUB 2500
1000 IF I$="W" THEN 870
1010 IF I $= "N" THEN GOSUB 2500
1020 IF I$="S" THEN 1150
1030 GOSUB 3260,
1040 GOTO 980
1050 REM -----
1060 ROOM=3
1070 GOSUB 2980
1080 IF I$="E" THEN 1150
1090 IF I$="W" THEN GOSUB 2500
1100 IF I $= "N" THEN GOSUB 2500
1110 IF I$="S" THEN 1420
1120 GOSUB 3260
1130 GOTO 1070
1140 REM -----
1150 ROOM=4
```

```
1160 GOSUB 2980
1170 IF I $="E" THEN 1240"
1180 IF I$="W" THEN 1060
1190 IF I$="N" THEN 970
1200 IF IS="S" THEN 1510
1210 GOSUB 3260
1220 GOTO 1160
1230 REM ---
1240 ROOM=5
1250 GOSUB 2980
1260 IF I $= "E" THEN 1330
1270 IF I $="W" THEN 1150
1280 IF I $="N" THEN GOSUB 2500
1290 IF I$="S" THEN 1600
1300 GOSUB 3260
1310 GOTO 1250
1320 REM -----
1330 ROOM=6
1340 GOSUB 2980
1350 IF I$="E" THEN GOSUB 2500
1360 IF I $="W" THEN 1240
1370 IF I $="N" THEN GOSUB 2500
1380 IF I $= "S" THEN 1780
1390 GOSUB 3260
1400 GOTO 1340
1410 REM -----
1420 ROOM=7
1430 GOSUB 2980
1440 IF IS="E" THEN GOSUB 2500
1450 IF I$="W" THEN GOSUB 2500
1460 IF I$="N" THEN 1060
1470 IF I $= "S" THEN GOSUB 2500
1480 GOSUB 3260
1490 GOTO 1430
1500 REM ------
1510 ROOM=8
1520 GOSUB 2980
1530 IF I$="E" THEN GOSUB 2500
1540 IF I$="W" THEN GOSUB 2500
1550 IF I$="N" THEN 1150
1560 IF I$="S" THEN 2050
1570 GOSUB 3260
1580 GOTO 1520
1590 REM -----
1600 ROOM=9
1610 GOSUB 2980
1620 IF I$="E" THEN GOSUB 2500
1630 IF I $= "W" THEN GOSUB 2500
1640 IF IS="N" THEN 1240
1650 IF I$="S" THEN 2140
1660 GOSUB 3260
1670 GOTO 1610
1680 REM ------
1690 ROOM=10
1700 GOSUB 2980
```

```
1710 IF I$="E" THEN 1780
1720 IF I$="W" THEN GOSUB 2500
1730 IF I$="N" THEN GOSUB 2500
1740 IF I$="S" THEN GOSUB 2500
1750 GOSUB 3260
1760 GOTO 1700
1770 REM -----
1780 ROOM=11
1790 GOSUB 2980
1800 IF I$="E" THEN 2410
1810 IF I$="W" THEN 1690
1820 IF IS="N" THEN 1330
1830 IF I$="S" THEN 2230
1840 GOSUB 3260
1850 GOTO 1790
1860 REM -----
1870 ROOM=12
1880 GOSUB 2980
1890 IF I$="E" THEN 1960
1900 IF I$="W" THEN GOSUB 2500
1910 IF I$="N" THEN GOSUB 2500
1920 IF I$="S" THEN 2320
1930 GOSUB 3260
1940 GOTO 1880
1950 REM -----
1960 ROOM=13
1970 GOSUB 2980
1980 IF I$="E" THEN 2050
1990 IF I$="W" THEN 1870
2000 IF I$="N" THEN GOSUB 2 500
2010 IF I$="S" THEN GOSUB 2500
2020 GOSUB 3260
2030 GOTO 1970
2040 REM ----
2050 ROOM=14
2060 GOSUB 2980
2070 IF I$="E" THEN GOSUB 2500
2080 IF I$="W" THEN 1960
2090 IF I$="N" THEN 1510
2100 IF I$="S" THEN GOSUB 2500
2110 GOSUB 3260
2120 GOTO 2060
2130 REM -----
2140 ROOM=15
2150 GOSUB 2980
2160 IF I$="E" THEN 2230
2170 IF I$="W" THEN GOSUB 2500
2180 IF I$="N" THEN 1600
2190 IF IS="S" THEN 2410
2200 GOSUB 3260
2210 GOTO 2150
2220 REM ----
2230 ROOM=16
2240 GOSUB 2980
2250 IF I$="E" THEN GOSUB 2500
```

(continued)

```
2260 IF IS="W" THEN 2140
  2270 IF I$="N" THEN 1780
  2280 IF I$="S" THEN 2410
  2290 GOSUB 3260
  2300 GOTO 2240
  2310 REM ---
  2320 ROOM=17
  2330 GOSUB 2980
  2340 IF I$="E" THEN 2410
  2350 IF I$="W" THEN 1870
  2360 IF I$="N" THEN GOSUB 2500
  2370 IF I$="S" THEN 870
  2380 GOSUB 3260
  2390 GOTO 2330
  2400 REM ----
  2410 ROOM=18
  2420 GOSUB 2980
  2430 IF I$="E" THEN GOSUB 2500
  2440 IF I$="W" THEN 2320
  2450 'IF I $= "N" THEN 2140
  2460 IF I$="S" THEN 1780
  2470 GOSUB 3260
  2480 GOTO 2420
  2490 REM ----- DEAD END SUBROUTINE -
  2500 ON DI GOTO 2510,2520,2530
-2510 ON INT(RND(1) *5) +1 GOTO 2540,2570,2660,2690,2720
-2520 ON INT(RND(1) *6) +1 GOTO 2540,2600,2630,2660,2690,2720
-2530 ON INT (RND(1) *7) +1 GOTO 2540,2570,2600,2630,2660,2690,
       2720
  2540 PRINT "SORRY, THERE IS NO DOOR HERE."
  2550 PRINT "YOU CAN'T GO THIS WAY."
  2560 RETURN
  2570 PRINT "THIS WHOLE WALL IS OOZING ACID."
  2580 PRINT "YOU CAN'T GO THIS WAY."
 2590 RETURN
  2600 PRINT "THIS DOOR HAS SUDDENLY CLOSED."
  2610 PRINT "YOU CAN'T GO THIS WAY."
  2620 RETURN
  2630 PRINT "HIDEOUS GIGANTIC SPIDERS BAR THE WAY."
  2640 PRINT "YOU CAN'T GO THIS WAY."
  2650 RETURN
  2660 PRINT "AHEAD ARE SHEETS OF UNPASSABLE FLAME."
  2670 PRINT "YOU CAN'T GO THIS WAY."
  2680 RETURN
  2690 PRINT "YOU ALMOST WALK INTO A CLOUD OF POISON."
  2700 PRINT "YOU CAN'T GO THIS WAY."
  2710 RETURN
2720 ON INT(RND(1) *4)+1 GOTO 2730,2790,2850,2910
  2730 PRINT "THE WALLS START TO CRUMBLE AROUND YOU,"
=2740 ON INT(RND(1) *6)+1 GOTO 2750,2750,2750,2750,2750,2770
  2750 PRINT "BUT YOU NARROWLY ESCAPE."
  2760 RETURN
  2770 PRINT "AND CRUSH YOU INTO DUST.
                                         THE END."
  2780 END
  2790 PRINT "YOU FALL INTO A SNAKE-FILLED PIT"
```

```
-2800 ON INT(RND(1) *6) +1 GOTO 2810,2810,2810,2810,2810,2830
 2810 PRINT "BUT MANAGE TO PULL YOURSELF TO SAFETY."
 2820 RETURN
 2830 PRINT "AND ARE SAVAGELY BITTEN.
                                        THE END."
 2840 END
 2850 PRINT "HISSING SCUM CREATURES BLOCK THE WAY,"
-2860 ON INT(RND(1)*6)+1 GOTO 2870,2870,2870,2870,2870,2890
 2870 PRINT "BUT YOU FREEZE AND THEY IGNORE YOU."
 2880 RETURN
 2890 PRINT "AND TURN YOU INTO JELLY.
                                        THE END."
 2900 END
 2910 PRINT "A TORNADO OF BROKEN GLASS BLOWS AT YOU,"
-2920 ON INT(RND(1)*6)+1 GOTO 2930,2930,2930,2930,2930,2930,2950
 2930 PRINT "BUT YOU DUCK SAFELY BEHIND FALLEN ROCK."
 2940 RETURN
 2950 PRINT "AND SLASHES YOU TO SHREDS. THE END"
 2960 END
 2970 REM ----- KEY SUBROUTINE -----
 2980 PRINT
 2990 GOSUB 3620
 3000 PRINT " -----> YOU ARE IN ROOM "; ROOM; " <-----
 3010 IF ROOM<>TR THEN 3090
 3020 IF F1=1 THEN 3170
 3030 PRINT "** YOU'VE FOUND ALL THE TREASURE!!! **"
 3040 GOSUB 3550
 3050 Fl=1
 3060 IF F2=1 THEN 3200
 3070 PRINT "(NOW TRY TO LOCATE THE KEY TO ESCAPE)"
 3080 PRINT
 3090 IF ROOM<>MG THEN 3170
 3100 IF F2=1 THEN 3170
 3110 PRINT "** YOU'VE FOUND THE KEY TO ESCAPE!! **"
 3120 GOSUB 3550
 3130 F2=1
 3140 IF F1=1 THEN 3200
 3150 PRINT "(NOW TRY TO LOCATE THE VAST TREASURE)"
 3160 PRINT
 3170 PRINT "WHICH WAY DO YOU WANT TO GO";
 3180 INPUT I$
 3190 RETURN
 3200 PRINT "******** YOU DID IT!!! ********
 3210 PRINT "YOU FOUND THE TREASURE--AND ALSO FOUND"
 3220 PRINT "THE PLATINUM KEY TO LET YOU ESCAPE !!!"
 3230 PRINT "NOW JUST RETURN TO ROOM 1 TO GET BACK!"
 3240 GOTO 3160
 3250 REM ---- DIRECTION SUBROUTINE ----
 3260 IF I$="N" OR I$="E" OR I$="W" OR I$="S" THEN 3530
 3270 PRINT CHR$(7)
 3280 K=K+1
 3290 IF K<5 THEN 3360
 3300 PRINT "YOU DIDN'T LISTEN, DID YOU?"
 3310 PRINT "YOU ARE NOW FALLING INTO A HORRIBLE"
 3320 PRINT "STINKING SWAMP CRAWLING WITH DEADLY"
 3330 PRINT "CREATURES THAT WILL DEVOUR YOU IN"
```

3340 PRINT "JUST UNDER 20 SECONDS. BYE-BYE!"

(continued)

```
3350 END
 3360 PRINT "YOU JUST WALKED INTO A CORNER AND"
 3370 PRINT "BANGED YOUR HEAD ON A SLIMY ROCK."
 3380 PRINT
 3390 PRINT "*";
 3400 FOR L=1 TO 34
 3410 PRINT "=";
 3420 NEXT L
 3430 PRINT "*"
 3440 PRINT ": THE ONLY DIRECTIONS YOU CAN GIVE :"
 3450 PRINT ": ARE: ";Q$;"N";Q$;", ";Q$;"E";Q$;", ";Q$;"S";
~3460 PRINT Q$;", OR ";Q$;"W";Q$;TAB(36);":"
 3470 PRINT "*";
 3480 FOR L=1 TO 34
 3490 PRINT "=":
 3500 NEXT L
 3510 PRINT
 3520 PRINT
 3530 RETURN
 3540 REM ---- BEEP SUBROUTINE ---
 3550 FOR B=1 TO 5
 3560 PRINT CHR$(7);
 3570 FOR LAG=1 TO 600
 3580 NEXT LAG
 3590 NEXT B
 3600 RETURN
 3610 REM --- SLIME MONSTER SUBROUTINE --
 3620 IF SLIME=1 THEN 3750
 3630 IF ROOM<7 THEN 3900
 3640 MAGICS=""
 3650 FOR S=1 TO 7
-3660 MAGIC = MAGIC + CHR + (INT(RND(1) + 26) + 65)
 3670 NEXT S
 3680 GOSUB 3920
 3690 PRINT "YOU SEE A WORD WRITTEN IN BLOOD ON THE"
 3700 PRINT "WALL: "; MAGIC$; " REMEMBER THIS WORD--"
 3710 PRINT "IT MAY JUST SAVE YOUR LIFE!!!"
 3720 GOSUB 3920
 3730 SLIME=1
 3740 GOTO 3900
-3750 IF INT(RND(1) * 7) <>1 THEN 3900
 3760 GOSUB 3920
 3770 PRINT "THE TERRIBLE SLIME MONSTER IS IN THE"
 3780 PRINT "ROOM WITH YOU.
                            YOU HAVE ONE CHANCE"
 3790 PRINT "TO USE THE MAGIC WORD TO SAVE YOUR"
 3800 PRINT "LIFE--OTHERWISE YOU'LL BE DEAD IN TWO"
 3810 PRINT "SECONDS! QUICK! WHAT IS IT";
 3820 INPUT TRY$
 3830 IF TRY$=MAGIC$ THEN 3870
 3840 PRINT "NO!! YOU'RE WRONG! THE MAGIC WORD"
 385Ø PRINT "WAS "; MAGIC$;". AARRRRGGGHHHH!!!!"
 3860 END
 3870 PRINT "THIS TIME YOU REMEMBERED. LUCKY YOU!"
 3880 SLIME=0
 3890 GOSUB 3920
```

```
3900 RETURN
 3910 REM --- LINE DRAWING SUBROUTINE
 3920 PRINT
-3930 N=(RND(1)*26)+64
 3940 FOR E=1 TO 38
 3950 PRINT CHR$(N);
 3960 NEXT E
 3970 PRINT
 3980 PRINT
 3990 RETURN
 4000 REM --- CHECK FOR WIN SUBROUTINE ---
 4010 IF F1<>1 OR F2<>1 THEN 4120
 4020 PRINT
 4040 PRINT
 4050 PRINT "YOU DID IT!!!! YOU FOUND THE TREASURE"
 4060 PRINT "AND THE KEY TO ESCAPE--AND YOU MADE IT"
 4070 PRINT "BACK TO ROOM 1 TO GET OUT OF THE MAZE!"
 4080 PRINT
 4100 GOSUB 3550
 4110 END
 4120 PRINT
 4130 PRINT "$$$$$$$$$$$$ INVENTORY $$$$$$$$$$
 4140 PRINT
 4150 IF F2=1 THEN PRINT TAB(9); "-- KEY TO ESCAPE -- "
 4160 IF F1=1 THEN PRINT TAB(7); "-- FABULOUS TREASURE --"
 4170 IF F1=0 AND F2=0 THEN PRINT TAB(5);"-- NOTHING BUT
      YOUR WITS --"
 4180 PRINT
 4190 PRINT "$$$$$$$$$$$$$$$$$$$$$$$$$$$$
 4200 PRINT
 4210 RETURN
```

### If You Have.

#### APPLEII

Delete: 760

Change: 450 PRINT "(HIT THE RETURN KEY FOR MORE)";

#### **ATARI**

Delete: 760

Add: 95 DIM X\$(1),Q\$(1),I\$(1),MAGIC\$(7),TRY\$(7)4175 PRINT "--NOTHING BUT YOUR WITS--" 4225 FOR AA=1 TO IDX 4230 PRINT " ";:NEXT AA:RETURN (continued)

```
Change: 120 IDX=4:GOSUB 4225:PRINT "1--HARD"
        130 GOSUB 4225: PRINT "2--MEDIUM"
        140 GOSUB 4225: PRINT "3--EASY"
        450 PRINT "(HIT THE RETURN KEY FOR MORE)";
        830 TR=INT(RND(0) *16) +3
        840 MG=INT(RND(0) *16) +3
        2510 ON INT(RND(\emptyset) *5)+1 GOTO 2540,2570,2660,2690,
              2720
        2520 ON INT(RND(0) *6)+1 GOTO 2540,2600,2630,2660,
              2690,2720
        2530 ON INT(RND(0) *7) +1 GOTO 2540,2570,2600,2630,
              2660,2690,2720
        2720 ON INT(RND(0) *4)+1 GOTO 2730,2790,2850,2910
        2740 ON INT(RND(0) \star 6)+1 GOTO 2750,2750,2750,
             2750,2770
        2800 ON INT(RND(0) \star6) +1 GOTO 2810,2810,2810,
             2810,2830
        2860 ON INT(RND(0) \star6) +1 GOTO 2870,2870,2870,
              2870,2890
        2920 ON INT(RND(0) \pm 6) +1 GOTO 2930,2930,2930,
             2930,2950
        3270 PRINT CHR$(253)
        3460 PRINT Q$; ", OR ";Q$; "W";Q$;:IDX=8:GOSUB 4225:
              PRINT ":"
        3560 PRINT CHR$ (253)
        3660 MAGIC\$(S) = CHR\$(INT(RND(0) * 26) + 65)
        3750 IF INT(RND(0) \star7) <>1 THEN 3900
        3930 N = (RND(0) * 26) + 64
        3940 FOR E=1 TO 37
        4150 IF F2=1 THEN IDX=8:GOSUB 4225:PRINT "--KEY
              TO ESCAPE--"
        4160 IF F1=1 THEN IDX=6:GOSUB 4225:PRINT
              "--FABULOUS TREASURE--"
         4170 IF F1=0 AND F2=0 THEN IDX=4:GOSUB 4225
```

#### **COMMODORE 64**

Delete: 760

Change: 450 PRINT "(HIT THE RETURN KEY FOR MORE)";
830 TR=INT(RND(0)\*16)+3
840 MG=INT(RND(0)\*16)+3
2510 ON INT(RND(0)\*5)+1 GOTO 2540,2570,2660,2690,
2720
2520 ON INT(RND(0)\*6)+1 GOTO 2540,2600,2630,2660,
2690,2720
2530 ON INT(RND(0)\*7)+1 GOTO 2540,2570,2600,2630,
2660,2690,2720
2720 ON INT(RND(0)\*4)+1 GOTO 2730,2790,2850,2910
2740 ON INT(RND(0)\*6)+1 GOTO 2750,2750,2750,2750,
2750,2770
2800 ON INT(RND(0)\*6)+1 GOTO 2810,2810,2810,2810,
2810,2830

```
2860 ON INT(RND(0)*6)+1 GOTO 2870,2870,2870,2870,

2870,2890
2920 ON INT(RND(0)*6)+1 GOTO 2930,2930,2930,2930,

2930,2950
3460 PRINT Q$;", OR ";Q$;"W";Q$;TAB(27);":"
3660 MAGIC$=CHR$(INT(RND(0)*26)+65)
3750 IF INT(RND(0)*7)<>1 THEN 3900
3930 N=(RND(0)*26)+64
```

#### COMMODORE VIC-20 Program Will Not Run on Unexpanded VIC-20

#### TEXAS INSTRUMENTS 99/4A

```
Add: 315 PRINT "(HIT THE ENTER KEY FOR MORE)"
        316 INPUT Y$
        605 PRINT "(HIT THE ENTER KEY FOR MORE)"
        606 INPUT Y$
        935 GOSUB 2500
        1025 GOSUB 2500
        1115 GOSUB 2500
        1295 GOSUB 2500
        1385 GOSUB 2500
        1475 GOSUB 2500
        1565 GOSUB 2500
        1655 GOSUB 2500
        1745 GOSUB 2500
        1925 GOSUB 2500
        2015 GOSUB 2500
        2105 GOSUB 2500
        2195 GOSUB 2500
2285 GOSUB 2500
        2375 GOSUB 2500
        2465 GOSUB 2500
        4155 PRINT TAB(9);"--KEY TO ESCAPE--"
        4165 PRINT TAB(7); "--FABULOUS TREASURE--"
        4173 GOTO 4180
        4175 PRINT TAB(5); "--NOTHING BUT YOUR WITS--"
Change: 830 \text{ TR} = \text{INT}(\text{RND} * 16) + 3
        840 MG=INT (RND*16)+3
        920 IF I$="N" THEN 935
        930 IF I$<>"S" THEN 940
        990 IF I$="E" THEN 1025
        1010 IF I$="N" THEN 1025
        1090 IF I$="W" THEN 1115
        1100 IF I$="N" THEN 1115
        1280 IF I$="N" THEN 1295
        1350 IF I$="E" THEN 1385
        1370 IF I$="N" THEN 1385
        1440 IF I$="E" THEN 1475
        1450 IF I$="W" THEN 1475
        1470 IF I$<>"S" THEN 1480
        1530 IF I$="E" THEN 1565
                                                     (continued)
```

```
1540 IF I $= "W" THEN 1565
1620 IF I$="E" THEN 1655
1630 IF I$="W" THEN 1655
     IF I$="W" THEN 1745
1720
1730 IF I$="N" THEN 1745
1740 IF I$<>"S" THEN 1750
1900 IF I$="W" THEN 1925
1910 IF I$="N" THEN 1925
2000 IF I$="N" THEN 2015
        I$<>"S" THEN 2020
2010 IF
        I$="E" THEN 2105
2070
     IF
2100 IF I$<>"S" THEN 2110
2170 IF I$="W" THEN 2195
2250 IF I$="E" THEN 2285
2360 IF I$="N" THEN 2375
2430 IF IS="E" THEN 2465
2510 ON INT(RND*5)+1 GOTO 2540,2570,2660,2690,2720
2520 ON INT(RND*6)+1 GOTO 2540,2600,2630,2660,2690,
     2720
2530 ON INT(RND*7)+1 GOTO 2540,2570,2600,2630,2660,
     2690,2720
2720 ON INT(RND*4)+1 GOTO 2730,2790,2850,2910
2740 ON INT(RND*6)+1 GOTO 2750,2750,2750,2750,
     2770
2800 ON INT(RND*6)+1 GOTO 2810,2810,2810,2810,2810,
     2830
2860 ON INT(RND*6)+1 GOTO 2870,2870,2870,2870,2870,
     2890
2920 ON INT(RND*6)+1 GOTO 2930,2930,2930,2930,2930,
     2950
3000 PRINT "-> YOU ARE IN ROOM";ROOM; " <-"
3200 PRINT "****** YOU DID IT ******
3260 IF (I\$="N")+(I\$="E")+(I\$="W")+(I\$="S") THEN
     3530
3270 CALL SOUND(100,440,2)
3400 FOR L=1 TO 26
3480 FOR L=1 TO 26
3560 CALL SOUND (100,440,2)
3570 FOR LAG=1 TO 100
3660 MAGIC$=MAGIC$&CHR$(INT(RND*26)+65)
3750 IF INT(RND*7) <>1 THEN 3900
3930 N=(RND*26)+64
3940 FOR E=1 TO 28
4010 IF (F1<>1)+(F2<>1) THEN 4120
4130 PRINT "SSSSSSS INVENTORY $$$$$$$
4150 IF F2<>1 THEN 4160
4160 IF F1<>1 THEN 4170
4170 IF (F1=0)*(F2=0) THEN 4175
```

#### TR8-80 COLOR COMPUTER

Delete: 110,150,210,420,430,440,450,460,470,480,490,760,770,790,810,860,960,1050,1140,1230,1320,1410,1500,1590,1600,1610,1620,1630,1640,1650,1660,1670,1680,

```
3080,3160,3250,3380,3470,3480,3490,3500,3510,3540,
        3610,3680,3720,3760,3890,3910,3920,3930,3940,3950,
        3960,3970,3980,3990,4000,4030,4040,4120,4140,4180,
        4190
   Add: 355 PRINT "HIT ENTER FOR MORE"
        356 INPUT X$
        595 INPUT X$
Change: 350 PRINT "SERIES OF TUNNELS AND CHAMBERS BENEATH
            THE SURFACE."
        360 PRINT "BEFORE HE";
        590 PRINT "HIT ENTER FOR MORE"
        780 CLS
        800 GOSUB 3390
        830 TR = RND(16) + 2
        840 TR=RND(16)+2
        1290 IF I$="S" THEN 2140
        2180 IF I$="N" THEN 1240
        2510 ON RND(5) GOTO 2540,2570,2660,2690,2720
        2520 ON RND(6) GOTO 2540,2600,2630,2660,2690,2720
        2530 ON RND(7) GOTO 2540,2570,2600,2630,2660,2690,
             2720
        2720 ON RND(4) GOTO 2730,2790,2850,2910
        2740 ON RND(6) GOTO 2750,2750,2750,2750,2750,2770
        2800 ON RND(6) GOTO 2810,2810,2810,2810,2810,2830
        2860 ON RND(6) GOTO 2870,2870,2870,2870,2870,2890
        2920 ON RND(6) GOTO 2930,2930,2930,2930,2930
        2980 REM
        3000 PRINT "YOU ARE IN ROOM "; ROOM;
       3200 PRINT "YOU DID IT!!"
        3270 SOUND 173,3
        3400 FOR L=1 TO 28
        3560 SOUND 173,3
        3660 MAGIC$=MAGIC$+CHR$(RND(26)+64)
       3750 IF RND(7)<>1 THEN 3900
        4010 IF F1<>1 OR F2<>2 THEN 4130
```

4130 PRINT "INVENTORY"

1770,1860,1950,2040,2130,2220,2310,2400,2490,2970,

## **MORE FUN! LESS TYPING!**

If you just finished typing in all the programs in this book and your pet gerbil ate the disk or cassette you saved them on...

If you're already enjoying the shorter programs in this book, but your fingers tremble at the thought of typing in the longer ones...

If your secretary does all your typing, but she's on a year-long vacation in Moose Jaw, Saskatchewan...

Or if you'd just like to have all these programs handy on a disk or cassette that'll run on your machine...

You're in luck!

You can get every single program in this book (along with extra bonus programs not included in this book) on ready-to-run cassettes or disks, for only \$19.95 per book. You can also get tapes or disks for any of the other exciting books in this series.

(As a special bonus, order any three, and we'll throw in the fourth one absolutely free—so you can enjoy the entire set—and we'll pay all postage and handling!)

Programs are available in one format only (disk or cassette) for each specific computer, so please make sure you have the proper equipment before ordering. Remember to check off BOTH the set of programs you want AND the type of machine you have.

Hard/Soft Inc., PO Box 1277, Riverdale, NY 10471

Yes! Please send me the ready-to-run programs I've checked off below.

1. Che	ck off which format you want:	2. Check off which programs you are
	ARI cassette tapes	ordering:
	PLE II/II PLUS/IIE disks	☐ Computer Craziness (\$19.95)
	MMODORE 64 disks	☐ Computer Monsters (\$19.95)
	MMODORE VIC-20 disks	☐ Computer Olympics (\$19.95)
` 🔲 IBI	M PC/PC ir disks	☐ Computer Space Adventures (\$19.95)
	XAS INŠTRUMENTS 99/4A	☐ ALL FOUR BOOKS (SPECIAL
cass	sette tapes	PRICE \$59.85)
	S-80 COLOR COMPUTER	
cas	sette tapes	C
**	N I	
3	N. 11	4
	NT your name and address, and fill in a specificable sales taxes!)	in the amount you are enclosing:
ån S	I am enclosing \$19.95 PLUS \$2.00 cable sales tax for each. Total: \$	postage and handling and appli-
	a	*
	SPECIAL BONUS: Check this be	ox, enclose \$59.85, and we'll send
- 38 - 16	you all FOUR sets of programs for to postage! This is a savings of over \$2	
		<b>U</b> )
	PRINT your name	
	220	74
	PRINT your address	
	225	11 3
	City State_	Zip
	Make checks payable to Hard/Soft, In	ic. Allow 2-8 weeks for delivery.



## LET YOUR COMPUTER TRKE YOU TO THE FARTHEST REACHES OF THE GALAXY!

COMPUTER SPACE ADVENTURES is jam-packed with exciting adventure, math, simulation, and fantasy games—along with programs that will test your knowledge of the universe. After escaping from nasty space creatures and distant caverns, you can watch your imaginary bank account grow at astronomical rates or try your reading skills on the Pluto Stock Exchange.

Dozens of easy-to-type-and-run programs will let you help a broken navigational computer get you safely back to Earth, create dazzling unworldly graphics, or just settle back and watch the stars go by.

With COMPLITER SPACE ADVENTURES you'll spend hours learning about the vast distances in space, the brightness of the stars, and your weight on other planets. Once you sample the fun-filled pages you simply won't want to come back down to Earth!

Other books by these authors:

COMPUTER CRAZINESS COMPUTER OLYMPICS COMPUTER MONSTERS



